

Designing Programs with Class

Sam Tobin-Hochstadt, David Van Horn and Benjamin Lerner

April 7, 2014

This book introduces the fundamental elements of class-based program design.

The book is also available in PDF form [here](#).

Contents

Preface

This book is a **draft** textbook to accompany Fundamentals of Computer Science II (Honors): Introduction to Class-Based Program Design at Northeastern University.¹ It began its life as a series of course notes from the Spring 2011 incarnation of the class, evolved during Spring 2012, and is now being revised throughout the Spring 2013 semester.

The “book” is very much a work in progress, which means there are large omissions and numerous errors. Moreover, it’s certain to change as the course progresses. Your patience, feedback, and bug reports are greatly appreciated. The book and its accompanying software are maintained on Github at the following URL:

<https://github.com/dvanhorn/dpc>

¹<http://www.ccs.neu.edu/course/cs2510h/>

Design Recipes

It puts a lot of emphasis on something called The Design Recipe, which sounded hokey to me at first, but when I was shown what The Design Recipe was, I realized, even as someone who's been programming for 35 years, that rather than code a solution to something complex in say, an hour, after 15-20 minutes, when I thought I was a third done, I was actually and suddenly completely done, sometimes almost feeling like magic. It basically puts a high emphasis on thinking and designing first, writing test cases very early, picking the right program structure, and then filling in the blanks and expanding them in a few places. Following the Design Recipe, not only was I often "suddenly done," but because my tests were written before I started coding the meat of a solution, I had (and continue to have) high confidence that my program works and will continue to work (or alert me quickly to new errors) should I make any "enhancements" down the road.

— Geoffrey S. Knauth, 2011

The main focus of this book is the *design process* that leads from problem statements to well-organized solutions, oriented around the concept of *objects*. We make extensive use of explicit design guidelines formulated as a number *program design recipes* as described in *How to Design Programs*. The most general form of the recipe for designing programs is given in figure 1.

1. Problem Analysis & Data Definition
2. Contract, Purpose & Effect Statements, Header
3. Examples
4. Template
5. Code
6. Tests

Figure 1: The Design Recipe for Programs

The objective of the design recipe is to provide a *technique* that guides programmers—novices and professionals alike—with a systematic approach to problem solving. By reasoning systematically, programmers can eliminate the incidental complexities of writing software and instead focus their creative energy on what is essentially difficult about a particular problem. A great programmer is first a master technician; they continue to hone and internalize their skills with each day of programming. As the chef Jacques Pépin said, ² speaking of another field that values the idea of recipes, to master the technique, you have no choice: “you have to repeat, repeat, repeat, repeat until it becomes part of yourself.”

²*New York Times*, “There’s the Wrong Way and Jacques Pépin’s Way” October 18, 2011.

The Choice of Language and Environment

Another lesson we should have learned from the recent past is that the development of “richer” or “more powerful” programming languages was a mistake in the sense that these baroque monstrosities, these conglomerations of idiosyncrasies, are really unmanageable, both mechanically and mentally. I see a great future for very systematic and very modest programming languages.

— Edsger W. Dijkstra “The Humble Programmer”, Turing Award Lecture, 1972

In support of this book, we have developed a series of modest programming languages that emphasize the principles of object-oriented design. We use these languages throughout the first part of the book. They are deliberately not industrial strength programming languages, complete with full-featured libraries, convenience mechanisms, and the usual idiosyncrasies that accompany “real” languages. Instead we have designed a progression of simple and consistent languages which embody the common core of modern object-oriented languages. The idea is that by focusing on the conceptual basis of object-oriented programming, students can apply their design knowledge regardless of whatever linguistic context they happen to find themselves in down the road.

In the second part, we transition to Java, a widely used industrial language that has been developed over more than 15 years. The move to Java allows us to explore the application of the principles introduced in the first portion of the course in the context of a practical language.

As of this draft of the book, the course software has been developed and tested with version 5.2 of Racket. To install the course software, you will first need to install Racket:

<http://racket-lang.org/>

Once installed, launch DrRacket, the development environment that ships with the Racket system. Click on the File|Install PLT File, and then enter the URL:

`http://www.ccs.neu.edu/course/cs2510h/class-system-latest.plt`

After the PLT file has been installed, select Language|Choose Language... and select the “Use the language declared in the source” option. You can now write programs in any of the languages included in the course software by writing `#lang class/N` as the first line of a file, where *N* is a number in 0, 1, 2, etc. To find out more about the languages, use Help Desk and search for `class/0` to get started.

The Parts of the Book

Acknowledgments

We are grateful to Matthias Felleisen, our TAs: Daniel Brown (2011), Asumu Takikawa (2012), and Nicholas Labich(2013), our tutors: Alex Lee, Nikko Patten, Jim Shargo, Trevor Sontag (2011), Spencer Florence, Sarah Laplante, Ryan Plessner (2012), Becca MacKenzie, and Kathleen Mullins (2013), and the Northeastern students we have had the privilege of teaching in 2011 and 2012.

Part I

Basic Design with Objects

Chapter 1

Objects = Data + Function

One of the key concepts behind so-called *object-oriented programming* (OOP) is the notion of an *object*. An object is a new kind of value that can, as a first cut, be understood as a pairing together of two familiar concepts: data and function.

- **An object is like a structure** in that it has a fixed number of fields, thus an object (again, like a structure) can represent compound data. But unlike a structure, an object contains not just data, but *functionality* too;
- **An object is like a (set of) function(s)** in that it has behavior—it *computes*; it is not just inert data.

This suggests that objects are a natural fit for well-designed programs since good programs are organized around data definitions and functions that operate over such data. An object, in essence, packages these two things together into a single programming apparatus. This has two important consequences:

1. **You already know how to design programs oriented around objects.**

Since objects are just the combination of two familiar concepts that you already use to design programs, you already know how to design programs around objects, even if you never heard the term “object” before. In short, the better you are at programming with functions, the better you will be at programming with objects.

2. **Objects enable new kinds of abstraction and composition.**

Although the combination of data and function may seem simple, objects enable new forms of abstraction and composition. That is, objects open up new approaches to the construction of computations. By studying these new approaches, we can distill new design principles. Because we understand objects

are just the combination of data and function, we can understand how all of these principles apply in the familiar context of programming with functions. In short, the better you are at programming with objects, the better you will be at programming with functions.

In this chapter, we will explore the basic concepts of objects by revisiting a familiar program, first organized around data and functions and then again organized around objects.

1.1 Functional rocket

In this section, let's develop a simple program that animates the lift-off of a rocket.

The animation will be carried out by using the `big-bang` system of the `2htdp/universe` library. For an animation, `big-bang` requires settling on a representation of *world states* and two functions: one that renders a world state as an image, and one that consumes a world state and produce the subsequent world state.

Generically speaking, to make an animation we must design a program of the form:

```
(big-bang <world0>           ; World
      (on-tick <tick>)       ; World -> World
      (to-draw <draw>))      ; World -> Scene
```

where `World` is a data definition for world states, `<tick>` is an expression whose value is a `World -> World` function that computes successive worlds and `<draw>` is an expression whose value is a `World -> Scene` function that renders a world state as an image.

For the purposes of a simple animation, the world state can consist of just the rocket:

```
;; A World is a Rocket.
```

The only relevant piece of information that we need to keep track of to represent a rocket lifting off is its height. That leads us to using a single number to represent rockets. Since rockets only go up in our simple model, we can use non-negative numbers. We'll interpret a non-negative number as meaning the distance between the ground and the (base of the) rocket measured in *astronomical units* (AU):

```
;; A Rocket is a non-negative Number.
;; Interp: distance from the ground to base of rocket in AU.
```

This dictates that we need to develop two functions that consume Rockets:

```
;; next : Rocket -> Rocket
;; Compute next position of the rocket after one tick of time.

;; render : Rocket -> Scene
;; Render the rocket as a scene.
```

Let's take them each in turn.

1.1.1 The `next` function

For `next`, in order to compute the next position of a rocket we need to settle on the amount of elapsed time a call to `next` embodies and how fast the rocket rises per unit of time. For both, we define constants:

```
(define CLOCK-SPEED 1/30) ; SEC/TICK
(define ROCKET-SPEED 1)   ; AU/SEC
```

The `CLOCK-SPEED` is the rate at which the clock ticks, given in seconds per tick, and `ROCKET-SPEED` is the rate at which the rocket lifts off, given in AU per second. We use these two constants to define a third, computed, constant that gives change in the rocket's distance from the ground per clock tick:

```
(define DELTA (* CLOCK-SPEED ROCKET-SPEED)) ; AU/TICK
```

We can now give examples of how `next` should work. We are careful to write test-cases in terms of the defined constants so that if we revise them later our tests will still be correct:


```
(check-expect (next 10) (+ 10 DELTA))
```

Now that we have developed a purpose statement, contract, and example, we can write the code, which is made clear from the example:


```
;; next : Rocket -> Rocket
;; Compute next position of the rocket after one tick of time.
(check-expect (next 10) (+ 10 DELTA))
(define (next r)
  (+ r DELTA))
```

1.1.2 The `render` function

The purpose of `render` is visualize a rocket a scene. Remember that rockets are represented by the distance between the ground and their base, so a rocket at height `0` should sitting at the bottom of a scene. We want it to look something like:



To do so we need to settle on the size of the scene and the look of the rocket. Again, we define constants for this. We use the 2http/image library for constructing images.



```
(define ROCKET ) ; Use rocket key to insert the rocket here.
(define WIDTH 100) ; PX
(define HEIGHT 200) ; PX
(define MT-SCENE (empty-scene WIDTH HEIGHT))
```

You can copy and paste the rocket image from this program, or you can access the image as follows:

```
> (bitmap class/0/rocket.png)
```



Since we may want to draw rockets on scenes other than the `MT-SCENE`, let's develop a helper function:

```
;; draw-on : Rocket Scene -> Scene
;; Draw rocket on to scene.
(define (draw-on r scn) ...)
```

allowing us to define `render` simply as:

```
;; render : Rocket -> Scene
;; Render the rocket as a scene.
(define (render r)
  (draw-on r MT-SCENE))
```

Recall that a rocket is represented by the distance from the ground to its **base**. On the other hand, the `2htdp/image` library works in terms of *pixels* (PX) and *graphics coordinates*. We need `draw-on` to establish the mapping between AU. For simplicity, we assume 1 PX equals 1 AU. Using `overlay/align/offset`, the `draw-on` function places the rocket on the scene on the center, bottom of the scene, offset vertically by the height of the rocket:

```
;; draw-on : Rocket Scene -> Scene
;; Draw rocket on to scene.
(define (draw-on r scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 r)
    scn))
```

1.1.3 Lift off

With these functions in place, let's launch a rocket:

```
;; Lift off!
(big-bang 0
  (tick-rate CLOCK-SPEED)
  (on-tick next)
  (to-draw render))
```

Our complete BSL program is:

```
(require 2htdp/image)
(require 2htdp/universe)
```

```

; A World is a Rocket.

; A Rocket is a non-negative Number.
; Interp: distance from the ground to base of rocket in AU.

```

```

(define CLOCK-SPEED 1/30) ; SEC/TICK
(define ROCKET-SPEED 1)   ; AU/SEC
(define DELTA (* CLOCK-SPEED ROCKET-SPEED)) ; AU/TICK

```



```

(define ROCKET ) ; Use rocket key to insert the rocket here.
(define WIDTH 100) ; PX
(define HEIGHT 200) ; PX
(define MT-SCENE (empty-scene WIDTH HEIGHT))

```

```

; next : Rocket -> Rocket
; Compute next position of the rocket after one tick of time.
(check-expect (next 10) (+ 10 DELTA))
(define (next r)
  (+ r DELTA))

```

```

; render : Rocket -> Scene
; Render the rocket as a scene.
(define (render r)
  (draw-on r MT-SCENE))

```

```

; draw-on : Rocket Scene -> Scene
; Draw rocket on to scene.
(check-expect (draw-on 0 (empty-scene 100 100))
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 1
    (empty-scene 100 100)))

```

```

(define (draw-on r scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 r)
    scn))

```

```

; Lift off!
(big-bang 0
  (tick-rate CLOCK-SPEED)
  (on-tick next))

```



```
(to-draw render))
```

1.2 Object-oriented rocket

Now let's redevelop this program only instead of using data and functions, we'll use objects.

You'll notice that there are two significant components to the rocket program. There is the *data*, which in this case is a number representing the distance the rocket has traveled, and the *functions* that operate over that class of data, in this case `next` and `render`.

This should be old-hat programming by now. But in this book, we are going to explore a new programming paradigm that is based on *objects*. As a first approximation, you can think of an *object* as the coupling together of the two significant components of our program (data and functions) into a single entity: an object.

Since we are learning a new programming language, you will no longer be using BSL and friends. Instead, select Language|Choose Language... in DrRacket, then select the "Use the language declared in the source" option and add the following to the top of your program:

```
#lang class/0
```

The constants of the rocket program remain the same, so our new program still includes a set of constant definitions:

```
(define CLOCK-SPEED 1/30) ; SEC/TICK
(define ROCKET-SPEED 1)   ; AU/SEC
(define DELTA (* CLOCK-SPEED ROCKET-SPEED)) ; AU/TICK
```



```
(define ROCKET ) ; Use rocket key to insert the rocket here.
(define WIDTH 100) ; PX
(define HEIGHT 200) ; PX
(define MT-SCENE (empty-scene WIDTH HEIGHT))
```

A set of objects is defined by a *class*, which determines the number and name of fields and the name and meaning of each behavior that every object in the set contains. By analogy, while an object is like a structure, a class definition is like a structure definition.

1.2.1 A class of rockets

The way to define a class is with `define-class`:

```
(define-class rocket%
  (fields dist))
```

This declares a new class of values, namely `rocket%` objects. (By convention, we will use the `%` suffix for the name of classes.) For the moment, `rocket%` objects consist only of data: they have one *field*, the `dist` between the rocket and the ground.

Like a structure definition, this class definition defines a new kind of data, but it does not make any particular instance of that data. To make a new instance of a particular class, i.e. an object, you use the `new` syntax, which takes a class name and expressions that produce a value for each field of the new object. Since a `rocket%` has one field, `new` takes the shape:

```
> (new rocket% 7)
(new rocket% 7)
```

This creates a `rocket%` representing a rocket with height 7.

In order to access the data, we can invoke the `dist` accessor method. Methods are like functions for objects and they are called by using the `send` form like so:

```
> (send (new rocket% 7) dist)
7
```

This suggests that we can now re-write the data definition for Rockets:

```
;; A Rocket is a (new rocket% NonNegativeNumber)
;; Interp: distance from the ground to base of rocket in AU.
```

1.2.2 The `next` and `render` methods

To add functionality to our class, we define *methods* using the `define` form. In this case, we want to add two methods `next` and `render`:

```
;; A Rocket is a (new rocket% NonNegativeNumber)
;; Interp: distance from the ground to base of rocket in AU.
(define-class rocket%
  (fields dist)
```

```
;; next : ...
(define (next ...) ...)

;; render : ...
(define (render ...) ...)
```

We will return to the contracts and code, but now that we've seen how to define methods, let's look at how to *apply* them in order to actually compute something. To call a defined method, we again use the `send` form, which takes an object, a method name, and any arguments to the method:

```
(send (new rocket% 7) next ...)
```

This will call the `next` method of the object created with `(new rocket% 7)`. This is analogous to applying the `next` function to `7` in the section 2.1 section. The elided code `(...)` is where we would write additional inputs to the method, but it's not clear what further inputs are needed, so now let's turn to the contract and method headers for `next` and `render`.

When we designed the functional analogues of these methods, the functions took as input the rocket on which they operated, i.e. they had headers like:

```
;; next : Rocket -> Rocket
;; Compute next position of the rocket after one tick of time.

;; render : Rocket -> Scene
;; Render the rocket as a scene.
```

But in an object, the data and functions are packaged together. Consequently, the method does not need to take the world input; that data is already a part of the object and the values of the fields are accessible using accessors. In other words, methods have an implicit input that does not show up in their header—it is the object that has called the method. That value, since it is not available as an explicit parameter of the method, is made available through the `this` variable. We likewise revise the purpose statements to reflect the fact “the rocket” is the object calling the method, so we instead write “this rocket”, emphasizing that `this` refers to a rocket.

That leads us to the following method headers:

rocket%

```
;; next : -> Rocket
;; Compute next position of this rocket after one tick of time.
(define (next) ...)
```

```
;; render : -> Scene
;; Render this rocket as a scene.
(define (render) ...)
```

The `rocket%` box is our way of saying that this code should live in the `rocket%` class.

Since we now have contracts and have seen how to invoke methods, we can now formulate test cases:

rocket%

```
;; next : -> Rocket
;; Compute next position of this rocket after one tick of time.
(check-expect (send (new rocket% 10) next)
               (new rocket% (+ 10 DELTA)))
(define (next) ...)

;; render : -> Scene
(check-expect (send (new rocket% 0) render)
               (overlay/align/offset "center" "bottom"
                                     ROCKET
                                     0 1
                                     MT-SCENE))
(define (render) ...)
```

Finally, we can write the code from our methods:

rocket%

```
(define (next)
  (new rocket% (+ (send this dist) DELTA)))

(define (render)
  (send this draw-on MT-SCENE))
```

Just as in the functional design, we choose to defer to a helper to draw a rocket on to the empty scene, which we develop as the following method:


rocket%

```
;; draw-on : Scene -> Scene
;; Draw this rocket on to scene.
(define (draw-on scn)
  (overlay/align/offset "center" "bottom"
                        ROCKET
                        0 (add1 (send this dist))
                        scn))
```

At this point, we can construct `rocket%` objects and invoke methods.

Examples:

```
> (new rocket% 7)
(new rocket% 7)
> (send (new rocket% 7) next)
(new rocket% 211/30)
> (send (new rocket% 80) render)
```



1.2.3 A big-bang oriented to objects

It's now fairly easy to construct a program using `rocket%` objects that is of the generic form of a **big-bang** animation:

```
(big-bang <world0>           ; World
  (on-tick <tick>)           ; World -> World
  (to-draw <draw>))          ; World -> Scene
```

We can again define a world as a rocket:

```
;; A World is a Rocket.
```

We now need to construct `Rocket -> Rocket` and `Rocket -> Scene` functions—but the work of these functions is already taken care of by the `next` and `render` methods. Thus we construct simple functions that call the appropriate method on the given rocket:

```
(require 2htdp/universe)
(big-bang (new rocket% 0)
  (on-tick (λ (r) (send r next)))
  (to-draw (λ (r) (send r render)))))
```

This creates the desired animation, but something should stick out about the above code. The `big-bang` system works by giving a piece of data (a number, a position, an image, an object, etc.), and a set of functions that operate on that kind of data. That sounds a lot like... *an object*! It's almost as if the interface for `big-bang` were designed for, but had to fake, objects.

Now that we have objects proper, we can use a new `big-bang` system has an interface more suited to objects. To import this OO-style `big-bang`, add the following to the top of your program:

```
(require class/universe)
```

In the functional setting, we had to explicitly give a piece of data representing the state of the initial world and list which functions should be used for each event in the system. In other words, we had to give both data and functions to the `big-bang` system. In an object-oriented system, the data and functions are already packaged together, and thus the `big-bang` form takes a single argument: an object that both represents the initial world and implements the methods needed to handle system events such as `to-draw` and `on-tick`.

So to launch our rocket, we simply do the following:

```
(big-bang (new rocket% 0))
```

In order to handle events, we need to add the methods `on-tick` and `to-draw` to `rocket%`:

```
;; on-tick : -> World
;; Tick this world
(define (on-tick) ...)

;; to-draw : -> Scene
;; Draw this world
(define (to-draw) ...)
```

These methods, for the moment, are synonymous with `next` and `render`, so their code is simple:

```
rocket%
```

```
rocket%
```

```
(define (on-tick) (send this next))
(define (to-draw) (send this render))
```


Our complete program is:


```
#lang class/0
(require 2htdp/image)
(require class/universe)

; A World is a Rocket.

; A Rocket is a (new rocket% NonNegativeNumber).
; Interp: distance from the ground to base of rocket in AU.

(define CLOCK-SPEED 1/30) ; SEC/TICK
(define ROCKET-SPEED 1)    ; AU/SEC
(define DELTA (* CLOCK-SPEED ROCKET-SPEED)) ; AU/TICK



(define ROCKET ) ; Use rocket key to insert the rocket here.
(define WIDTH 100) ; PX
(define HEIGHT 200) ; PX
(define MT-SCENE (empty-scene WIDTH HEIGHT))

(define-class rocket%
  (fields dist)

  ; next : -> Rocket
  ; Compute next position of this rocket after one tick of time.
  (check-expect (send (new rocket% 10) next)
    (new rocket% (+ 10 DELTA)))
  (define (next)
    (new rocket% (+ (send this dist) DELTA)))

  ; render : -> Scene
  ; Render this rocket as a scene.
  (check-expect (send (new rocket% 0) render)
    (overlay/align/offset "center" "bottom"
      ROCKET
      0 1
      MT-SCENE))

  (define (render)
    (send this draw-on MT-SCENE))

  ; draw-on : Scene -> Scene
```

```

; Draw this rocket on to scene.
(define (draw-on scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 (send this dist))
    scn))

; on-tick : -> World
; Tick this world
(define (on-tick) (send this next))

; to-draw : -> Scene
; Draw this world
(define (to-draw) (send this render)))

; Lift off!
(big-bang (new rocket% 0))

```

You've now seen the basics of how to write programs with objects.

1.3 A Brief History of Objects

Objects are an old programming concept that first appeared in the late 1950s and early 1960s just across the Charles river at MIT in the AI group that was developing Lisp. Simula 67, a language developed in Norway as a successor to Simula I, introduced the notion of classes. In the 1970s, Smalltalk was developed at Xerox PARC by Alan Kay and others. Smalltalk and Lisp and their descendants have influenced each other ever since. Object-oriented programming became one of the predominant programming styles in the 1990s. This coincided with the rise of graphical user interfaces (GUIs), which objects model well. The use of object and classes to organize interactive, graphical programs continues today with libraries such as the Cocoa framework for Mac OS X.

1.4 Exercises

1.4.1 Complex, with class

For this exercise, you will develop a class-based representation of complex numbers, which are used in several fields, including: engineering, electromagnetism, quantum physics, applied mathematics, and chaos theory.

A *complex number* is a number consisting of a *real part* and an *imaginary part*. It can be written in the mathematical notation $a+bi$, where a and b are real numbers, and i is the standard imaginary unit with the property $i^2 = -1$.

You can read more about the sophisticated number system of Racket in the section ??? section on section ???.

Complex numbers are so useful, it turns out they are included in the set of numeric values that Racket supports. The Racket notation for writing down complex numbers is $5+3i$, where this number has a real part of 5 and an imaginary part of 3; $4-2i$ has a real part of 4 and imaginary part of -2. (Notice that complex numbers *generalize* the real numbers since any real number can be expressed as a complex number with an imaginary part of 0.) Arithmetic operations on complex numbers work as they should, so for example, you can add, subtract, multiply, and divide complex numbers. (One thing you can't do is *order* the complex numbers, so $<$ and friends work only on real numbers.)

Examples:

```
; Verify the imaginary unit property.
> (sqr (sqrt -1))
-1
> (sqr 0+1i)
-1
; Arithmetic on complex numbers.
> (+ 2+3i 4+5i)
6+8i
> (- 2+3i 4+5i)
-2-2i
> (* 2+3i 4+5i)
-7+22i
> (/ 2+3i 4+5i)
23/41+2/41i
; Complex numbers can't be ordered.
> (< 1+2i 2+3i)
<: contract violation
  expected: real?
  given: 1+2i
  argument position: 1st
  other arguments....:
    2+3i
; Real numbers are complex numbers with an imaginary part of 0,
; so you can perform arithmetic with them as well.
> (+ 2+3i 2)
4+3i
> (- 2+3i 2)
0+3i
> (* 2+3i 2)
4+6i
```

```

> (/ 2+3i 2)
1+3/2i
> (magnitude 3+4i)
5

```

Supposing your language was impoverished and didn't support complex numbers, you should be able to build them yourself since complex numbers are easily represented as a pair of real numbers—the real and imaginary parts.

Design a structure-based data representation for Complex values. Design the functions `=?`, `plus`, `minus`, `times`, `div`, `sq`, `mag`, and `sqrt`. Finally, design a utility function `to-number` which can convert Complex values into the appropriate Racket complex number. Only the code and tests for `to-number` should use Racket's complex (non-real) numbers and arithmetic since the point is to build these things for yourself. However, you can use Racket to double-check your understanding of complex arithmetic.

For mathematical definitions of complex number operations, see the Wikipedia entries on complex numbers and the square root of a complex number.

```

> (define c-1 (make-cpx -1 0))

> (define c0+0 (make-cpx 0 0))

> (define c2+3 (make-cpx 2 3))

> (define c4+5 (make-cpx 4 5))

> (=? c0+0 c0+0)
#t
> (=? c0+0 c2+3)
#f
> (=? (plus c2+3 c4+5)
      (make-cpx 6 8))
#t

```

Develop a class-based data representation for Complex values. Add accessor methods for extracting the `real` and `imag` parts. Develop the methods `=?`, `plus`, `minus`, `times`, `div`, `sq`, `mag`, `sqrt` and `to-number`.

Examples:

```

; Some example Complex values.
> (define c-1 (new complex% -1 0))

> (define c0+0 (new complex% 0 0))

```

```

> (define c2+3 (new complex% 2 3))

> (define c4+5 (new complex% 4 5))

; Verify the imaginary unit property.
> (send c-1 mag)
1
> (send c-1 sqroot)
(new complex% 0 1)
> (send (send (send c-1 sqroot) sq) =? c-1)
#t
> (send (send (new complex% 0 1) sq) =? c-1)
#t
; Arithmetic on complex numbers.
> (send c0+0 =? c0+0)
#t
> (send c0+0 =? c2+3)
#f
> (send (send c2+3 plus c4+5) =?
      (new complex% 6 8))
#t
> (send (send c2+3 minus c4+5) =?
      (new complex% -2 -2))
#t
> (send (send c2+3 times c4+5) =?
      (new complex% -7 22))
#t
> (send (send c2+3 div c4+5) =?
      (new complex% 23/41 2/41))
#t
> (send (new complex% 3 4) mag)
5

```

1.4.2 Circles

For this exercise, you will develop a structure-based representation of circles and functions that operate on circles, and then develop a class-based representation of circles.

A *circle* has a radius and color. They also have a position, which is given by the coordinates of the center of the circle (using the graphics coordinates system).

1. The `circ` structure and functions.

Design a structure-based data representation for `Circle` values.

Design the functions `=?`, `area`, `move-to`, `move-by`, `stretch`, `draw-on`, `to-image`, `within?`, `overlap?`, and `change-color`.

Here are a few examples to give you some ideas of how the functions should work (note you don't necessarily need to use the same structure design as used here).

First, let's define a few circles we can use:

```
> (define c1 (make-circ 25 "red" 100 70))  
  
> (define c2 (make-circ 50 "blue" 90 30))  
  
> (define c3 (make-circ 10 "green" 50 80))
```

A `(make-circ R C X Y)` is interpreted as a circle of radius `R`, color `C`, and centered at position `(X,Y)` in graphics-coordinates.

The `to-image` function turns a circle into an image:

```
> (to-image c1)
```



```
> (to-image c2)
```




```
> (to-image c3)
```




While the `draw-on` function draws a circle onto a given scene:


```
> (draw-on c1 (empty-scene 200 200))
```




```
> (draw-on c2 (empty-scene 200 200))
```



```
> (draw-on c3 (empty-scene 200 200))
```



```
> (draw-on c1 (draw-on c2 (draw-on c3 (empty-scene 200 200))))
```




The `area` function computes the area of a circle:

```
> (area c1)
1963.4954084936207
> (area c2)
7853.981633974483
> (area c3)
314.1592653589793
```


The `move-to` function moves a circle to be centered at the given coordinates:

```
> (draw-on (move-to c1 100 100) (empty-scene 200 200))
```



While `move-by` moves a circle by the given change in coordinates:

```
> (draw-on (move-by c1 -30 20) (empty-scene 200 200))
```



The `within?` function tells us whether a given position is located within the circle; this includes any points on the edge of the circle:

```
> (within? c1 (make-posn 0 0))  
#f  
> (within? c1 (make-posn 110 80))  
#t
```

The `change-color` function produces a circle of the given color:

```
> (to-image (change-color c1 "purple"))
```




The `=?` function compares two circle for equality; two rectangles are equal if they have the same radius and center point—we ignore color for the purpose of equality:

```
> (=? c1 c2)
#f
> (=? c1 c1)
#t
> (=? c1 (change-color c1 "purple"))
#t
```

The `stretch` function scales a circle by a given factor:

```
> (draw-on (stretch c1 3/2) (empty-scene 200 200))
```



The `overlap?` function determines if two circles overlap at all:

```
> (overlap? c1 c2)
#t
> (overlap? c2 c1)
```




```
#t  
> (overlap? c1 c3)  
#f
```

2. The `circ%` class.

Develop a class-based data representation for Circle values. Develop the *methods* corresponding to all the functions above.

The methods should work similar to their functional counterparts:

```
> (define c1 (new circ% 25 "red" 100 70))  
  
> (define c2 (new circ% 50 "blue" 90 30))  
  
> (send c1 area)  
1963.4954084936207  
> (send c1 draw-on (empty-scene 200 200))
```



Chapter 2

Classes of Objects: Data Definitions

One of the most important lessons of *How to Design Programs* is that the structure of code follows the structure of the data it operates on, which means that the structure of your code can be derived *systematically* from your data definitions. In this chapter, we see how to apply the design recipe to design data represented using classes as well as operations implemented as methods in these classes.

We've seen various kinds of data definitions:

1. Atomic: numbers, images, strings, ...
2. Compound: structures, posns, ...
3. Enumerations: colors, key events, ...
4. Unions: atoms, ...
5. Recursive unions: trees, lists, matryoshka dolls, s-expressions, ...
6. Functions: infinite sets, sequences, ...

Each of these kinds of data definitions can be realized with objects. In this chapter, we'll examine how each the first five are implemented with a class-based design. We'll return to representing functions later.

2.1 Atomic and Compound Data

We already saw how to program with the object equivalent of atomic data in the chapter 2 chapter. If you worked through the section 2.4.1 exercise, you've already seen how to program with compound data, too.

Stepping back, we can see that the way to represent some fixed number N of data is with a class with N fields. For example, a position can be represented by a pair (x,y) of real numbers:

```
;; A Posn is (new posn% Real Real)
(define-class posn%
  (fields x y))
```

Methods can compute with any given arguments and the object that calling the method, thus the template for a `posn%` method is:

```
;; posn%-method : Z ... -> ???
(define (posn%-method z ...)
  (... (send this x) (send this y) z ...))
```

Here we see that our template lists the available parts of the `posn%` object, in particular the two fields `x` and `y`.

2.2 Enumerations

An *enumeration* is a data definition for a finite set of possibilities. For example, we can represent a traffic light like the ones on Huntington Avenue with a finite set of symbols, as we did in Fundies I:

```
;; A Light is one of:
;; - 'Red
;; - 'Green
;; - 'Yellow
```

Following the design recipe, we can construct the template for functions on Lights:

```
;; light-function : Light -> ???
(define (light-function l)
  (cond [(symbol=? 'Red l) ...]
        [(symbol=? 'Green l) ...]
        [(symbol=? 'Yellow l) ...]))
```

Finally, we can define functions over Lights, following the template.

```
;; next : Light -> Light
;; Next light after the given light
(check-expect (next 'Green) 'Yellow)
(check-expect (next 'Red) 'Green)
(check-expect (next 'Yellow) 'Red)
(define (next l)
  (cond [(symbol=? 'Red l) 'Green]
        [(symbol=? 'Green l) 'Yellow]
        [(symbol=? 'Yellow l) 'Red])))
```

That's all well and good for a function-oriented design, but we want to design this using classes, methods, and objects.

There are two obvious possibilities. First, we could create a `light%` class, with a field holding a `Light`. However, this fails to use classes and objects to their full potential. Instead, we will design a class for each state the traffic light can be in. Each of the three classes will have their own implementation of the `next` method, producing the appropriate `Light`.

```
#lang class/0
;; A Light is one of:
;; - (new red%)
;; - (new green%)
;; - (new yellow%)

(define-class red%
  ;; next : -> Light
  ;; Next light after red
  (check-expect (send (new red%) next) (new green%))
  (define (next)
    (new green%)))

(define-class green%
  ;; next : -> Light
  ;; Next light after green
  (check-expect (send (new green%) next) (new yellow%))
  (define (next)
    (new yellow%)))

(define-class yellow%
  ;; next : -> Light
  ;; Next light after yellow
  (check-expect (send (new yellow%) next) (new red%)))
```

```
(define (next)
  (new red%))
```


If you have a Light, L, how do you get the next light?

```
(send L next)
```

Note that there is no use of `cond` in this program, although the previous design using functions needed a `cond` because the `next` function has to determine *what kind of light is the given light*. However in the object-oriented version there's no use of a `cond` because we ask an object to call a method; each kind of light has a different `next` method that knows how to compute the appropriate next light. Notice how the purpose statements are revised to reflect knowledge based on the class the method is in; for example, the `next` method of `yellow%` knows that this light is yellow.

2.3 Unions and Recursive Unions

Unions are a generalization of enumerations to represent infinite families of data. One example is *binary trees*, which can contain arbitrary other data as elements. We'll now look at how to model binary trees of numbers, such as:



How would we represent this with classes and objects?

```
#lang class/0
;;  +- - - - - +
;;  | +- - - - - + |
;;  V V                               | |
;;  A BT is one of:                   | |
;;  - (new leaf% Number)               | |
;;  - (new node% Number BT BT)        | |
;;                                     | +- - - + |
;;                                     +- - - - - +
(define-class leaf%
  (fields number))

(define-class node%
  (fields number left right))
```

```

(define ex1 (new leaf% 7))
(define ex2 (new node% 6
                    (new leaf% 8)
                    (new node% 4
                        (new leaf% 3)
                        (new leaf% 2))))
(define ex3 (new node% 8
                    (new leaf% 2)
                    (new leaf% 1)))

```

We then want to design a method `count` which produces the number of numbers stored in a BT.

Here are our examples:

```

(check-expect (send ex1 count) 1)
(check-expect (send ex2 count) 5)
(check-expect (send ex3 count) 3)

```

Next, we write down the templates for methods of our two classes.

The template for `leaf%`:

leaf%

```

;; count : -> Number
;; count the number of numbers in this leaf
(define (count)
  (... (send this number) ...))

```

The template for `node%`:

node%

```

;; count : -> Number
;; count the number of numbers in this node
(define (count)
  (send this number) ...
  (send (send this left) count) ...
  (send (send this right) count) ...)

```

Now we provide a definition of the `count` method for each of our classes.

leaf%

```

;; count : -> Number
;; count the number of numbers in this leaf
(define (count)
  1)

```

node%

```
;; count : -> Number
;; count the number of numbers in this node
(define (count)
  (+ 1
     (send (send this left) count)
     (send (send this right) count)))
```

Next, we want to write the `double` function, which takes a number and produces two copies of the BT with the given number at the top. Here is a straightforward implementation for `leaf%`:

leaf%

```
;; double : Number -> BT
;; double this leaf and put the number on top
(define (double n)
  (new node%
    n
    (new leaf% (send this number))
    (new leaf% (send this number))))
```

Note that `(new leaf% (send this number))` is just constructing a new `leaf%` object just like the one we started with. Fortunately, we have a way of referring to ourselves, using the identifier `this`. We can thus write the method as:

leaf%

```
;; double : Number -> BT
;; double this leaf and put the number on top
(define (double n)
  (new node% n this this))
```

Since these two methods are so similar, you may wonder if they can be abstracted to avoid duplication. We will see how to do this in a subsequent class.

For `node%`, the method is very similar:

node%

```
;; double : Number -> BT
;; double this node and put the number on top
(define (double n)
  (new node% n this this))
```

The full BT code is now:

```
#lang class/0
;; +- - - - - +
```



```

;;      | +- - - - - - - - - - + |
;;      V V                        | |
;; A BT is one of:                  | |
;; - (new leaf% Number)             | |
;; - (new node% Number BT BT)       | |
;;                                  | +- - -+ |
;;                                  +- - - - -+
(define-class leaf%
  (fields number)
  ;; count : -> Number
  ;; count the number of numbers in this leaf
  (define (count)
    1)

  ;; double : Number -> BT
  ;; double the leaf and put the number on top
  (define (double n)
    (new node% n this this)))

(define-class node%
  (fields number left right)
  ;; count : -> Number
  ;; count the number of numbers in this node
  (define (count)
    (+ 1
      (send (send this left) count)
      (send (send this right) count)))

  ;; double : Number -> BT
  ;; double the node and put the number on top
  (define (double n)
    (new node% n this this)))

(define ex1 (new leaf% 7))
(define ex2 (new node% 6
  (new leaf% 8)
  (new node% 4
    (new leaf% 3)
    (new leaf% 2)))))
(define ex3 (new node% 8
  (new leaf% 2)
  (new leaf% 1)))

(check-expect (send ex1 count) 1)
(check-expect (send ex2 count) 5)
(check-expect (send ex3 count) 3)

```

```
(check-expect (send ex1 double 5)
              (new node% 5 ex1 ex1))
(check-expect (send ex3 double 0)
              (new node% 0 ex3 ex3))
```

2.4 Revisiting the Rocket

2.4.1 Landing and taking off

Let's now revise our section 2.2 program so that the rocket first descends toward the ground, lands, then lifts off again. Our current representation of a world is insufficient since it's ambiguous whether we are going up or down. For example, if the rocket is at 42, are we landing or taking off? There's no way to know. We can revise our data definition to include a representation of this missing information. As we hear this revised description, the idea of a union data definition should jump out: "a rocket is either landing or taking off." Let's re-develop our program with this new design.

Our revised class definition is then:

```
;; A World is a Rocket.

;; A Rocket is one of:
;; - (new takeoff% Number)
;; - (new landing% Number)

;; Interp: distance from the ground to base of rocket in AU,
;; either taking off or landing.

(define-class takeoff%
  (fields dist)
  ...)
(define-class landing%
  (fields dist)
  ...)
```

The signatures for our methods don't change, however we now have two sets of methods to implement: those for rockets taking off, and those for landing rockets.

First, let's make some test cases for the `next` method. We expect that a rocket taking off works just as before:

```
(check-expect (send (new takeoff% 10) next)
```

```
(new takeoff% (+ 10 DELTA-Y))
```

However, when landing we expect the rocket to be descending toward the ground. For simplicity, let's specify the rocket descends as fast as it ascends:

```
(check-expect (send (new landing% 100) next)
              (new takeoff% (- 100 DELTA-Y)))
```

There is an important addition case though. When the rocket is descending and gets close to the ground, we want it to land. So when the rocket is descending and less than `DELTA-Y` units from the ground, we want its next state to be on the ground, ready to lift off:

```
(check-expect (send (new landing% (sub1 DELTA-Y)) next)
              (new takeoff% 0))
```

Based on these examples, we can now define the `next` method in the `landing%` and `takeoff%` classes:

```
takeoff%
```

```
;; next : -> Rocket
;; Compute next position of this ascending rocket after one tick of time.
(define (next)
  (new takeoff% (+ (send this dist) DELTA-Y)))
```

```
landing%
```

```
;; next : -> Rocket
;; Compute next position of this descending rocket after one tick of time.
(define (next)
  (cond [(< (send this dist) DELTA-Y) (new takeoff% 0)]
        [else (new landing% (- (send this dist) DELTA-Y))]))
```

Now let's turn to the remaining methods such as `render`. When rendering a rocket, it's clear that it doesn't matter whether the rocket is landing or taking off; it will be drawn the same. This leads to having two *identical* definitions of the `render` method in both `takeoff%` and `landing%`. Since the method relies upon the helper method `draw-on`, we likewise have two identical definitions of `draw-on` in `takeoff%` and `landing%`.

```
landing% and takeoff%
```

```
;; render : -> Scene
;; Render this rocket as a scene.
(define (render)
```

```

(send this draw-on MT-SCENE))

; draw-on : Scene -> Scene
; Draw this rocket on to scene.
(define (draw-on scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 (send this dist))
    scn))

```

This duplication of code is unsettling, but for now let's just live with the duplication. We could abstract the code by defining a *function* and calling the function from both methods, but as we try to focus on object-oriented designs, let's instead recognize there's a need for an object-oriented abstraction mechanism here and revisit the issue later in the chapter on chapter 10.


We can experiment and see ascending rockets climb and descending rockets land:

Examples:

```

> (send (new landing% 5) next)
(new landing% 149/30)
> (send (new takeoff% 5) next)
(new takeoff% 151/30)
> (send (new landing% 0) next)
(new takeoff% 0)
> (send (new landing% (quotient HEIGHT 2)) render)

```



Implementing the needed methods for a **big-bang** animation is straightforward:

landing% and takeoff%

```
(define (on-tick) (send this next))
(define (to-draw) (send this render))
```

And to run the animation, just start `big-bang` with a landing rocket:

```
(big-bang (new landing% HEIGHT))
```

2.4.2 Adding a satellite

Let's now add an orbiting satellite. To do so, let's first forget about rockets and make an satellite animation. The satellite is represented by a class with a single field—a number giving the distance from the date line, which we'll draw at the left of the screen, to the center of the satellite. When the satellite gets to the edge of the screen, it will wrap around starting over again at the date line.

```
(define CLOCK-SPEED 1/30) ; SEC/TICK
(define SATELLITE-SPEED 1) ; AU/SEC
(define DELTA-X (* CLOCK-SPEED SATELLITE-SPEED)) ; AU/TICK

(define SATELLITE (circle 30 "solid" "red"))
(define WIDTH 100) ; PX
(define HEIGHT 200) ; PX
(define SATELLITE-Y (quotient HEIGHT 4))
(define MT-SCENE (empty-scene WIDTH HEIGHT))

;; A World is a Satellite.

;; A Satellite is a (new satellite% Number).
;; Interp: distance in AU from date line to center of satellite.

(define-class satellite%
  (fields dist)

  ;; next : -> Satellite
  ;; Move this satellite distance travelled in one tick.
  (define (next)
    (local [(define n (+ (send this dist) DELTA-X))]
      (new satellite% (cond [(> n WIDTH) (- n WIDTH)]
                            [else n])))))
```

Drawing the satellite is a little more tricky than the rocket because the satellite can appear on both the left and right side of the screen as it passes over the date line. A simple trick to manage this is to draw *three* satellites, each a full “day” behind and ahead of the current satellite, thus when the satellite is just past the date line, the

day ahead image appears on the right, and when the satellite approaches the end of the day, the day behind satellite appears on the left. To accomodate this, we define a helper method that draws the satellite at given day offsets.

satellite%


```
;; render : -> Scene
;; Render this satellite as a scene.
(define (render)
  (send this draw-on MT-SCENE))

;; draw-on : Scene -> Scene
;; Draw this satellite on scene.
(define (draw-on scn)
  (send this draw-on/offset -1
    (send this draw-on/offset 0
      (send this draw-on/offset 1
        MT-SCENE))))


;; draw-on/offset : Number Scene -> Scene
;; Draw this satellite on scene with given day offset.
(define (draw-on/offset d scn)
  (place-image SATELLITE
    (+ (send this dist) (* d WIDTH))
    SATELLITE-Y
    scn))
```

Examples:

```
> (send (new satellite% 0) next)
(new satellite% 1/30)
> (send (new satellite% (quotient WIDTH 2)) render)
```



```
> (send (new satellite% 0) render)
```



We can now add the needed methods to animate satellites with `big-bang`:

satellite%

```
(define (on-tick) (send this next))  
(define (to-draw) (send this render))
```

And then animate a satellite with:

```
(big-bang (new satellite% 0))
```

Now we have animations of rockets and of satellites, but putting the pieces together is simple. We need to revise our data definition. Let's make a new class of compound that *contains* a rocket and a satellite and implement the methods needed to make an animation:


```
;; A World is a (new space% Rocket Satellite).
(define-class space%
  (fields rocket satellite)

  (define (on-tick)
    (new space%
      (send (send this rocket) next)
      (send (send this satellite) next)))

  (define (to-draw)
    (send (send this rocket) draw-on
      (send (send this satellite) draw-on
        MT-SCENE))))
```

Example:

```
> (send (new space%
  (new landing% (quotient HEIGHT 2))
  (new satellite% (quotient WIDTH 3)))
  to-draw)
```



Finally, to animate the whole thing, we just call `big-bang` with an initial `space%` object:

```
(big-bang (new space%
  (new landing% HEIGHT)
  (new satellite% 0)))
```


2.5 Exercises

2.5.1 Lists of Numbers

Design classes to represent lists of numbers. Implement the methods `length`, `append`, `sum`, `prod`, `contains?`, `reverse`, `map`, and `max`. Note that `max` raises some interesting design decisions in the case of the empty list. One solution is to define the `max` of the empty list as negative infinity, `-inf.0`, a number smaller than every other number (except itself). Another solution is to only define `max` for non-empty lists of numbers.

2.5.2 Home on the Range

A *range* represents a set of numbers between two endpoints. To start with, you only need to consider ranges that *include* the smaller endpoint and *exclude* the larger endpoint—such ranges are called *half-open*. For example, the range `[3,4.7)` includes all of the numbers between 3 and 4.7, including 3 but *not* including 4.7. So 4 and 3.0000001 are both in the range, but 5 is not. In the notation used here, the “[” means include, and the “)” means exclude.

- Design a representation for ranges and implement the `in-range?` method, which determines if a number is in the range. For example, the range `[3,7.2)` includes the numbers 3 and 5.0137, but not the numbers -17 or 7.2.
- Extend the data definition and implementation of ranges to represent ranges that *exclude* the low end of the range and *include* the high end, written `(lo,hi]`.
- Add a `union` method to the interface for ranges and implement it in all range classes. This method should consume a range and produces a new range that includes all the numbers in this range *and* all the numbers in the given range. You may extend your data definition for ranges to support this method. Don't worry if your initial design duplicates code; you can abstract later.

Chapter 3

Classes of Objects: Interface Definitions

In this chapter, we take an alternative perspective on defining sets of objects; we can characterize objects not just by their construction, as done with a data definition, but also by the methods they support. We call this characterization an interface definition. As we'll see, designing to interfaces leads to generic and extensible programs.

3.1 Lights, revisited

Let's take another look at the `Light` data definition we developed in section 3.2. We came up with the following data definition:

```
;; A Light is one of:  
;; - (new red%)  
;; - (new green%)  
;; - (new yellow%)
```

We started with a `next` method that computes the successor for each light. Let's also add a `draw` method and then build a `big-bang` animation for a traffic light.

```
#lang class/0  
(require 2htdp/image)  
(define LIGHT-RADIUS 20)  
  
(define-class red%  
  ;; next : -> Light
```

```

;; Next light after red
(check-expect (send (new red%) next) (new green%))
(define (next)
  (new green%))

;; draw : -> Image
;; Draw this red light
(check-expect (send (new red%) draw)
  (circle LIGHT-RADIUS "solid" "red"))
(define (draw)
  (circle LIGHT-RADIUS "solid" "red"))

(define-class green%
  ;; next : -> Light
  ;; Next light after green
  (check-expect (send (new green%) next) (new yellow%))
  (define (next)
    (new yellow%))

  ;; draw : -> Image
  ;; Draw this green light
  (check-expect (send (new green%) draw)
    (circle LIGHT-RADIUS "solid" "green"))
  (define (draw)
    (circle LIGHT-RADIUS "solid" "green")))

(define-class yellow%
  ;; next : -> Light
  ;; Next light after yellow
  (check-expect (send (new yellow%) next) (new red%))
  (define (next)
    (new red%))

  ;; draw : -> Image
  ;; Draw this yellow light
  (check-expect (send (new yellow%) draw)
    (circle LIGHT-RADIUS "solid" "yellow"))
  (define (draw)
    (circle LIGHT-RADIUS "solid" "yellow")))


```

We can now create and view lights:

```
> (send (new green%) draw)
```



```
> (send (new yellow%) draw)
```



```
> (send (new red%) draw)
```



To create an animation we can make the following world:

```
(define-class world%
  (fields light)
  (define (tick-rate) 5)
  (define (to-draw)
    (send (send this light) draw))
  (define (on-tick)
    (new world% (send (send this light) next))))

(require class/universe)
(big-bang (new world% (new red%)))
```

At this point, let's take a step back and ask the question: *what is essential to being a light?* Our data definition gives us one perspective, which is that for a value to be a light, that value must have been constructed with either `(new red%)`, `(new yellow%)`, or `(new green%)`. But from the world's perspective, what matters is not how lights are constructed, but rather what can lights compute. All the world does is call methods on the light it contains, namely the `next` and `draw` methods. We can rest assured that the light object understands the `next` and `draw` messages because, by definition, a light must be one of `(new red%)`, `(new yellow%)`, or `(new green%)`, and each of these classes defines `next` and `draw` methods. But it's possible we could relax the definition of what it means to be a light by just saying what methods an object must implement in order to be considered a light. We can thus take a constructor-agnostic view of objects by defining a set of objects in terms of the methods they understand. We call a set of method signatures (i.e., name, contract, and purpose statement) an *interface*.

3.2 A light of a different color

So let's consider an alternative characterization of lights **not in terms of *what they are*, but rather *what they do***. Well a light does two things: it can render as an image and it can transition to the next light; hence our *interface definition* for a light is:

```
;; An ILight implements
;; next : -> ILight
;; Next light after this light.
;; draw : -> Image
;; Draw this light.
```

Now it's clear that every Light is an ILight because every Light implements the methods in the ILight interface, but we can imagine new kinds of implementations of the ILight interface that are not Lights. For example, here's a class that implements the ILight interface:

```
;; A ModLight is a (new mod-light% Natural)
;; Interp: 0 = green, 1 = yellow, otherwise red.
(define-class mod-light%
  (fields n)
  ;; next : -> ILight
  ;; Next light after this light.
  (define (next)
    (new mod-light% (modulo (add1 (send this n)) 3)))

  ;; draw : -> Image
  ;; Draw this light.
  (define (draw)
    (cond [(= (send this n) 0)
           (circle LIGHT-RADIUS "solid" "green")]
          [(= (send this n) 1)
           (circle LIGHT-RADIUS "solid" "yellow")]
          [else
           (circle LIGHT-RADIUS "solid" "red")]))))
```

Now clearly a ModLight is never a Light, but every ModLight is an ILight. Moreover, any program that is written for ILights will work *no matter what implementation we use*. So notice that the world program only assumes that its light field is an ILight; this is easy to inspect—the world never assumes the light is constructed in a particular way, it just calls `next` and `draw`. Which means that if we were to start our program off with

```
(big-bang (new world% (new mod-light% 2)))
```

it would work exactly as before.

3.3 Representation independence and extensibility

We've now developed a new concept, that of an *interface*, which is a collection of *method signatures*. We say that an object *is* an instance of an interface whenever it implements the methods of the interface.

The idea of an interface is already hinted at in the concept of a union of objects since a function over a union of data is naturally written as a method in each class variant of the union. In other words, to be an element of the union, an object must implement all the methods defined for the union—the object must implement the union's interface. But interfaces are about more than just unions. By focusing on interfaces, we can see there are two important engineering principles that can be distilled even from this small program:

1. Representation independence

As we've seen with the simple world program that contains a light, when a program is written to use only the methods specified in an interface, then the program is *representation independent* with respect to the interface; we can swap out any implementation of the interface without changing the behavior of the program.

2. Extensibility

When we write interface-oriented programs, it's easy to see that they are *extensible* since we can always design new implementations of an interface. Compare this to the construction-oriented view of programs, which defines a set of values once and for all.

These points become increasingly important as we design larger and larger programs. Real programs consist of multiple interacting components, often written by different people. *Representation independence* allows us to exchange and refine *components* with some confidence that the whole system will still work after the change. *Extensibility* allows us to add functionality to existing programs without having to change the code that's already been written; that's good since in a larger project, it may not even be possible to edit a component written by somebody else.

Let's look at the extensibility point in more detail. Imagine we had developed the Light data definition and its functionality along the lines of *HtDP*. We would have (we omit *draw* for now):

```
;; A Light is one of:  
;; - 'Red  
;; - 'Green  
;; - 'Yellow
```



```
;; next : Light -> Light
;; Next light after the given light
(check-expect (next 'Green) 'Yellow)
(check-expect (next 'Red) 'Green)
(check-expect (next 'Yellow) 'Red)
(define (next l)
  (cond [(symbol=? 'Red l) 'Green]
        [(symbol=? 'Green l) 'Yellow]
        [(symbol=? 'Yellow l) 'Red])))
```

Now imagine if we wanted to add a new kind of light—perhaps to represent a blinking yellow light. For such lights, let’s assume the next light is just a blinking yellow light:

```
(check-expect (next 'BlinkingYellow) 'BlinkingYellow)
```

That’s no big deal to implement *if we’re allowed to revise* `next`—we just add another clause to `next` to handle `'BlinkingYellow` lights. But what if we can’t? What if `next` were part of a module provided as a library? Well then life is more complicated; we’d have to write a new function, say `fancy-next`, that handled blinking lights and used `next` for all non-blinking lights. And while that gets us a new function with the desired behavior, that won’t do anything for all the places the `next` function is used. If we’re able to edit the code that uses `next`, then we can replace each use of `next` with `fancy-next`, but what if we can’t...? Well then we’re just stuck. If we cannot change the definition of `next` or all the places it is used, then it is not possible to extend the behavior of `next`.


Now let’s compare this situation to one in which the original program was developed with objects and interfaces. In this situation we have an interface for lights and several classes, namely `red%`, `yellow%`, and `green%` that implement the `next` method. Now what’s involved if we want to add a variant of lights that represents a blinking yellow light? We just need to write a class that implements `next`:

```
;; Interp: blinking yellow light
(define-class blinking-yellow%
  ;; next : -> ILight
  ;; Next light after this blinking yellow light.
  (check-expect (send (new blinking-yellow%) next)
                (new blinking-yellow%))
  (define (next) this))
```

Notice how we didn’t need to edit `red%`, `yellow%`, or `green%` at all! So if those things are set in stone, that’s no problem. Likewise, programs that were written to use the light interface will now work even for blinking lights. We don’t need to edit any uses of the `next` method in order to make it work for blinking lights. This program is truly extensible.

3.4 Case Study: Mobiles

Let's put our newfound interface skills to work by defining an interface and classes to represent mobiles, delicately balanced sculptures of shapes connected by branches.



To define mobiles, then, we'll first need to define simple shapes with weights:

```
;; An IShape implements:
;; to-draw: -> Image           -- renders the IShape
;; area: -> NonNeg             -- computes the IShape's area
;; scale: NonNeg -> IShape     -- returns a new IShape expanded
;;                             -- by the given factor

;; A circle% implements IShape and is
;; (new circle% NonNeg String NonNeg)
;;      -- radius, color, and weight
;; A rect% implements IShape and is
;; (new rect% NonNeg NonNeg String NonNeg)
;;      -- width, height, color and weight
```

From these definitions we can easily write the template for our classes:

```
#lang class/0
(require 2htdp/image)

(define-class circle%
  (fields radius color weight)
  (define (to-draw) ...)
  (define (area) ...)
  (define (scale factor) ...))
(define-class rect%
  (fields width height color weight)
  (define (to-draw) ...))
```

```
(define (area) ...)
(define (scale factor) ...))
```

Exercise

Complete the bodies of these methods.

Now we can move on to define mobiles.

Do Now!

Construct data definitions to describe a mobile. Can you think of a second definition?

One way to look at the picture above is to see a mobile as a tree structure, with either branch nodes or leaves. The leaves just contain `IShapes`, while the nodes contain a left mobile, a right mobile, and the lengths of the left and right branches:

```
;;
;;      +-----+
;;      | +-----+ |
;;      v v         | |
;; A Mobile is one of: | |
;; (new mobile-node% Mobile NonNeg Mobile NonNeg)
;;   -- left mobile, left branch length,
;;      right mobile, right branch length
;; (new mobile-leaf% IShape)
;;   -- shape hanging at this point
```

What methods might we want `Mobiles` to implement? Certainly we want to draw them. Supposing we are modelling a sculpture to hang in a room, we might want to know how much area it takes up, and then scale it larger or smaller to ensure it fits. In other words, we expect `Mobiles` to implement the `IShape` interface.

Do Now!

If a `Mobile` implements the `IShape` interface, then perhaps a simpler data definition will suffice: After all, we defined `mobile-node%` so that we could include an `IShape` into the mobile. But if a `Mobile` *implements* `IShape`, then we might not need this extra step. So perhaps the following would work:

```
;; A Mobile is  
;; (new mobile% IShape NonNeg IShape NonNeg)
```

Does this data definition seem better than the first one? Do you see any potential problems using it?



3.5 Sharing Interfaces

```
;; A Posn implements  
;; move-by : Real Real -> Posn  
;; move-to : Real Real -> Posn  
;; dist-to : Posn -> Real  
  
;; A Segment implements  
;; draw-on : Scene -> Scene  
;; move-by : Real Real -> Segment  
;; move-to : Real Real -> Segment  
;; dist-to : Posn -> Real
```

Notice that any object that is a Segment is also a Posn.

Chapter 4


PacMan

If interfaces and classes were all we needed to design object-oriented programs, our job would be done. We can successfully design small programs so far, but as our programs grow larger, we'll find that we need additional abstraction mechanisms to keep our code understandable. So let's build a larger system: we'll write a simplified version of PacMan, using the skills we currently have. And as we extend the game to include additional features, we'll introduce the new concepts that will keep the code clear.

4.1 Basic game mechanics

The first step in designing any program is to understand what information it must represent and manipulate.

Do Now!

 What information is necessary for PacMan?

Brainstorming the basic gameplay of PacMan, we probably need to represent:

- The playing area
- The player, which might include
 - The player's position

- The player’s score
 - The player’s remaining lives
- The ghosts, which can be
 - Angry ghosts, that chase the player, or
 - Scared ghosts, that run away from the player
- In-game items, which might include
 - The food dots
 - The powerups, that make ghosts scared
 - Bonus fruit
- The walls and obstacles
- Any extra features we want to represent

So our complete game will be structured something like

```
<pacman-game> ::=
  <data-definitions>
  <playing-area>
  <player>
  <ghosts>
  <in-game-items>
  <walls>
  <extras>
  <the-world>
```

4.2 Defining the playing space

For now, our playing area will just consist of an empty rectangular board: we’ll add walls and the ability to “loop around” from one side of the board to the other later. To represent this area, we’ll separate the *logical* board size from the physical size, just as we did with the rocket example in chapter 2:

```
<data-definitions> ::=

(define BOARD-WIDTH 20) ;; cells
(define BOARD-HEIGHT 30) ;; cells
(define SCALE-X 20) ;; px/cell
(define SCALE-Y 20) ;; px/cell
```

```
(define mtScene
  (empty-scene (* BOARD-WIDTH SCALE-X)
               (* BOARD-HEIGHT SCALE-Y)))
```

4.3 The player

What game state is relevant to the player? We need to keep track of its position, its score, how many lives it has, and so on. Should these pieces of information be included in the definition of the `player%` class, or the `world%` class?

Do Now!

Which do you think will be easier to work with as we develop the game?

If we keep the player's position in the `world%` class, then the world must be responsible for moving the player around. By similar reasoning, the world would need to be responsible for moving the ghosts around. But such movements are complicated (players, angry ghosts and scared ghosts all move differently), and placing all the code in the `world%` class would clutter it up. If we keep the player's coordinates stored as fields within the `player%` class, then when the world needs to move the player, it might be as simple as invoking `(the-player . move)`, and similarly for the ghosts. All the logic for how players and ghosts move would be placed in the most relevant classes.

This decision is one instance of a general pattern we will see many times: we are *delegating* responsibility over some computation (moving players and ghosts) from methods in one class (the `world%` object) to others (the `player%` class and the ghost classes).

Applying this reasoning again to the player's score and lives, we can represent them as fields in the player class as well.

So our initial design for the player class is

`<player-take-1> ::=`

```
;; A Player is (new player% Natural Natural Natural Natural)
(define-class player%
  (fields x y score lives)
  ...
)
```

4.4 Delegation

TODO

4.5 Ghosts

Let's start by just modeling angry ghosts, and come back to scared ghosts after we have the simplest possible skeleton of a game running. We don't yet know our full data definitions, but we can at least say

```
<ghosts> ::=

;; A Ghost is one of
;; -- (new angry-ghost% ...)
;; -- (new scared-ghost% ...)
```

We know there will be multiple ghosts, distinguishable by their colors and positions, so our first design for our `angry-ghost%` class is

```
<angry-ghost-take-1> ::=

(define-class angry-ghost%
  (fields x y color)
  ...
)
```

4.6 Drawing the game

So far, our world just contains a player and a bunch of ghosts, and we know eventually there will be a bunch of items as well. This means we need an interface for working with a list of ghosts:

```
<data-definitions> ::= (part 2)

;; A List of Ghosts is one of
;; -- (new consLoG% Ghost LoG)
;; -- (new mtLoG%)
(define-class consLoG%
  (fields first rest)
  ...)
(define-class mtLoG%
```



```
(fields)
...)
```

And an interface for working with a list of items:

<data-definitions> ::= (part 3)

```
;; A List of Items is one of
;; -- (new consLoI% Item LoG)
;; -- (new mtLoI%)
(define-class consLoI%
  (fields first rest)
  ...)
(define-class mtLoI%
  (fields)
  ...)
```

Now we can define our world:

<the-world> ::=

```
;; A World is (new world% Player LoG LoI)
(define-class world%
  (fields player ghosts items)
  <world-drawing>
  <world-tick>
  <world-key-handling>
  )
```

Let's get drawing! We know that the world must implement a method `to-draw`: -> [Scene](#), and that it must somehow draw all its subcomponents into a single [Scene](#).

4.7 Changing Places

4.7.1 Player in motion, or, I ain't afraid of no ghosts!

At the very least, we want our player to be able to outrun the ghosts:

<data-definitions> ::= (part 4)

```
(define PLAYER-SPEED 2)
(define GHOSTS-SPEED 1)
```

Our player object needs to respond to keypresses, so that it moves when the users

presses an arrow key. Following the design recipe, we start with examples of expected motion:

```
(check-expect ((new player% 3 5 40 1) . on-key "up") (new player% 3 (- 5 PLAYER-
SPEED) 40 1))
(check-expect ((new player% 3 5 40 1) . on-key "down") (new player% 3 (+ 5 PLAYER-
SPEED) 40 1))
(check-expect ((new player% 6 5 40 1) . on-key "left") (new player% (- 6 PLAYER-
SPEED) 5 40 1))
(check-expect ((new player% 6 5 40 1) . on-key "right") (new player% (+ 6 PLAYER-
SPEED) 5 40 1))
```

The template for the `on-key` method comes straight from our class definition: all we have available so far are our fields.

```
(define (on-key k)
  ... (this . x) ... (this . y) ... (this . score) ... (this . lives) ...)
```

We know that we have to handle four different keys, so we refine the template:

```
(define (on-key k)
  (cond
    [(string=? k "up")
     ... (this . x) ... (this . y) ... (this . score) ... (this . lives) ...]
    [(string=? k "down")
     ... (this . x) ... (this . y) ... (this . score) ... (this . lives) ...]
    [(string=? k "left")
     ... (this . x) ... (this . y) ... (this . score) ... (this . lives) ...]
    [(string=? k "right")
     ... (this . x) ... (this . y) ... (this . score) ... (this . lives) ...]
    [else
     ... ]))
```

Looking at the expected outputs guides our implementation:

```
(define (on-key k)
  (cond
    [(string=? k "up")
     (new player% (this . x) (- (this . y) PLAYER-SPEED) (this . score) (this . lives))]
    [(string=? k "down")
     (new player% (this . x) (+ (this . y) PLAYER-SPEED) (this . score) (this . lives))]
    [(string=? k "left")
     (new player% (- (this . x) PLAYER-SPEED) (this . y) (this . score) (this . lives))]
    [(string=? k "right")
     (new player% (+ (this . x) PLAYER-SPEED) (this . y) (this . score) (this . lives))]
    [else
     this ]))
```

Do Now!

Try running the game right now. Does it do what you expect?

Our code happily allows the player to exit the playing field. Our tests must be woefully incomplete, or they would have caught this behavior.

Do Now!

Add more tests to confirm that the player can never *move* outside the playing field

We clearly need to revise our implementation of *on-key*. We need to restrict the player's position so that it is always within bounds. But wait...

Our player has a position, and can change positions over time. Our ghosts also have positions, and also can change positions over time. Leaving aside *how* players and ghosts choose *where* to move, should we implement the position-manipulating logic in players and ghosts? No! Instead, as we just discussed, we should factor out our position-related code into a *posn%* class and delegate the details of representing motion to that class.

<data-definitions> ::= (part 5)

```
(define-class% posn%
  (fields x y)
  <posn-methods>
)
```

Now we can revise our representation of *player%* to use a *posn%* instead of *x* and *y* fields:

<player> ::=

```
;; A Player is (new player% Posn Natural Natural)
(define-class player%
  (fields p score lives)
  <player-methods>
)
```

How might we refactor *on-key*? Each of the branches in that method do nearly the same thing: create a new *player%* with a position that has moved by some increment in some direction. Following the design recipe for abstraction, this suggests

we might define a `velocity%` class, and add a `move` method to `posn%` that takes a `velocity` and produces a new `posn%`:

`<data-definitions> ::= (part 6)`

```
;; A Velocity is (new vel% Natural Natural)
(define-class velocity%
  (fields vx vy)
  ...
)
```

`<posn-methods> ::=`

```
;; move: Velocity -> Posn
;; move creates a new Posn by adding the given Velocity to the current Posn
(define (move v)
  (new posn% (+ (this #,dot x) (v #,dot vx)) (+ (this #,dot y) (v #,dot vy))))

(define (on-key k)
  (cond
    [(string=? k "up")
     (new player% (this . p . move (new vel% 0 (- PLAYER-SPEED)))
       (this . score) (this . lives))]
    [(string=? k "down")
     (new player% (this . p . move (new vel% 0 PLAYER-SPEED))
       (this . score) (this . lives))]
    [(string=? k "left")
     (new player% (this . p . move (new vel% (- PLAYER-SPEED) 0))
       (this . score) (this . lives))]
    [(string=? k "right")
     (new player% (this . p . move (new vel% PLAYER-SPEED 0))
       (this . score) (this . lives))]
    [else
     this ]))
```

This is better, but still has not solved the problem of moving off the board. But notice that now our positions are all `posn%`s — if we had a method on `posn%`s that “clamped” them to within a boundary, we would be done. We’ll also take this opportunity to clean up our repetitive code:

`<player-methods> ::=`

```
(define (on-key k)
  (local [(raw-moved-pos)
    (cond
      [(string=? k "up") (this . p . move (new vel% 0 (- PLAYER-SPEED)))]
      [(string=? k "down") (this . p . move (new vel% 0 PLAYER-SPEED))]
```

```

      [(string=? k "left") (this . p . move (new vel% (- PLAYER-SPEED) 0))]
      [(string=? k "right") (this . p . move (new vel% PLAYER-SPEED 0))]
      [else (this . p)]]]
    (new player% (raw-moved-pos . clamp 0 0 BOARD-WIDTH BOARD-HEIGHT)
      (this . score) (this . lives)))

```

Much better. What should `clamp` do?

```

(check-expect ((new posn% 40 10) . clamp 0 0 10 10) (new posn% 10 10))
(check-expect ((new posn% -5 10) . clamp 0 0 5 5) (new posn% 0 5))
(check-expect ((new posn% 5 5) . clamp 1 1 10 10) (new posn% 5 5))

```

<posn-methods> ::= (part 2)

```

;; clamp : Number Number Number Number -> Posn
;; relocates the current point to within the provided rectangle
(define (clamp min-x min-y max-x max-y)
  (new posn%
    (min (max (this . x) min-x) max-x)
    (min (max (this . y) min-y) max-y)))

```

4.8 World-building

<world-init> ::=

```

(big-bang the-world)

```


Chapter 5

Parameterized Data and Interfaces

5.1 Parametric data

Consider the parametric data definition for lists we studied last semester:

```
;; A [List X] is one of:  
;; - empty  
;; - (cons X [List X])
```

Recall this is really not just a single data definition, but a family of data definitions. We can obtain a different member of this family by plugging in some data definition for `X`; `List X` works just like a function—apply it to arguments results in its definition, but with `X` replaced by the argument. Unlike with a function, the arguments are not values, but *data definitions*. So for example, plugging in `Number` for `X`, written `[List Number]` results in:

```
;; A [List Number] is one of:  
;; - empty  
;; - (cons Number [List Number])
```

Applying `List` to `String` results in:

```
;; A [List String] is one of:  
;; - empty  
;; - (cons String [List String])
```

The process by which we obtained the parameterized data definition, `List X`, is one of abstraction. Looking back, we *started* with the non-parameterized data definitions:

```
;; A ListNumber is one of:
;; - empty
;; - (cons Number ListNumber)

;; A ListString is one of:
;; - empty
;; - (cons String ListString)
```

These definitions are so similar, it's natural to want to abstract them to avoid repeating this nearly identical code again and again. It's easy to see what is different: `Number` vs `String`, and that's exactly how we arrived at `[List X]`.

We can repeat the process for objects. Let's start with the non-parameterized data definitions we've been working with:

```
;; A ListNumber is one of:
;; - (new empty%)
;; - (new cons% Number ListNumber)

;; A ListString is one of:
;; - (new empty%)
;; - (new cons% String ListString)

(define-class empty%)
(define-class cons%
  (fields first rest))
```

Abstracting out the data definition of elements results in:

```
;; A [List X] is one of:
;; - (new empty%)
;; - (new cons% X [List X])
```

5.2 Parametric interfaces

We have now developed a parametric data definition for lists, focusing on the representation of lists. If instead we focused the behaviors of lists, we would arrive at a *parametric interface*:

```
;; A [List X] implements:
```



```
;; - cons : X -> [List X]
;;   Cons given value on to this list.
;; - first : -> X
;;   Get first element of this non-empty list.
;; - rest : -> [List X]
;;   Get the rest of this non-empty list.
```

We can implement the interface as follows:

```
;; A (new empty%) implements [List X]
(define-class empty%
  (define (cons x) (new cons% x this)))

;; A (new cons% X [List X]) implements [List X]
(define-class cons%
  (fields first rest)
  (define (cons x) (new cons% x this)))
```

5.3 Parameteric methods

We can design further methods:

```
;; A [List X] implements:
;; - cons : X -> [List X]
;;   Cons given value on to this list.
;; - first : -> X
;;   Get first element of this non-empty list.
;; - rest : -> [List X]
;;   Get the rest of this non-empty list.
;; - length : -> Natural
;;   Get the length of this list.
;; - append : [List X] -> [List X]
;;   Append this to given list.
;; - reverse : -> [List X]
;;   Reverse this list of elements.
;; - map : [X -> Y] -> [List Y]
;;   Map given function over this list.
;; - filter : [X -> Boolean] -> [List X]
;;   Select elements that satisfy the given predicate.
;; - foldr : [X Y -> Y] Y -> Y
;;   Fold right over this list.
;; - foldl : [X Y -> Y] Y -> Y
;;   Fold left over this list.
```

Again, the model of a parametric interface is as a *function* of classes of data. In this interface, X is the parameter, bound at the point of “A [List X] implements”. That variable occurs several times within the definition and is replaced by the argument of List. But on closer inspection, there are other variables that are not bound, e.g. Y in:

```
;; - map : [X -> Y] -> [List Y]
```

To be precise, there’s really another, implicit, parameter that doesn’t range over the whole interface, but just the `map` method. We can make this implicit parameter in the contract notation by adding a class variable at the level of `map`:

```
;; - map [Y] : [X -> Y] -> [List Y]
```

So for example, we might have a [List Number], in which case, the object has the `map` method:

```
;; - map [Y] : [Number -> Y] -> [List Y]
```

Even though the X has been replaced by Number, the Y parameter remains and only takes on a specific meaning from the function argument of `map`. So if `ns` is a List Number, we apply `map` to a [Number -> String] function, its contract is interpreted with Y replaced by String:

Examples:

```
> (define ns (new cons% 3 (new cons% 4 (new empty%))))

> (ns . map number->string)
(new cons% "3" (new cons% "4" (new empty%)))
```

5.4 Exercises

5.4.1 Parametric Lists

Design an implementation of the [List X] interface given in this chapter.

Chapter 6

Solidifying what we've done

So far, we've seen a number of different ways of specifying the creation and behavior of the data we work with. At this point, it's valuable to take a step back and consider all of the concepts we've seen.

6.1 Data Definitions

A data definition defines a *class*¹ of values by describing how instances of that data are constructed. In this book, we focus on particular new kind of value: *objects*, and data definitions will primarily define a class of objects. New kinds of objects are made with `define-class`, while instances of a class are made with a class constructor, written `(new class-name% arg ...)`.

Data definitions can be built out of primitive data, such as Numbers, Strings, Images, etc., but also compound data can be represented with objects containing data. Data definitions can also be formed as the (possibly recursive) union of other data definitions. For example:

```
;; A ListofImage is one of:  
;; - (new empty%)  
;; - (new cons% Image ListofImage)
```

Data definitions may be parameterized, meaning a family of similar data definitions is simultaneously defined by use of variable parameters that range over other classes of values. For example:

¹Here we mean “class” in the general sense of “a set of values,” not to be confused with the concept of a “class” as defined by the `define-class` form.

```
;; A [Pair X Y] is one of:  
;; - (new pair% X Y)
```

Here the Pair family of data definition is parameterized over classes of values X and Y.

6.2 Interface Definitions

Another way to define a class of values is by way of an *interface definition*. Unlike a data definition, which focuses on how data is represented, an interface defines a set of values by the operations that set of values support. Interfaces provide a means for defining a set of values independent of representation.

For example:

```
;; A [List X] implements  
  
;; - empty : -> [List X]  
;;   Produce an empty list  
;; - cons : X -> [List X]  
;;   Produce a list with the given element at the front.  
;; - empty? : -> Boolean  
;;   Determine if this list is empty.  
;; - length : -> Number  
;;   Count the elements in this list  
;; ... and other methods ...
```

There are several important aspects of this interface definition to note. First, it lists all of the methods that can be used on an [List X], along with their contracts and purpose statements. Mere method names are not enough—with just a method name you have no idea how to use a method, or what to use it for. Second, interface definitions can have parameters (here X), just like data definitions. Third, there is no description of how to *construct* an [List X]. That's the job of data definitions that implement this interface.

Of course, just like data definitions don't have to be named, interface definitions don't have to be named either. If you need to describe an interface just once, it's fine to write the interface right there where you need it.

6.3 Contracts

Contracts describe the appropriate inputs and outputs of functions and methods. In the past, we've seen many contracts that refer to data definitions. In this class, we've also seen contracts that refer to interface definitions, like so:

```
;; [IList Number] -> [IList Number]
```

When describing the contract of a function or method, it's almost always preferable to refer to an interface definition instead of a data definition that commits to a specific representation.

6.4 Design Recipe

Interfaces change the design recipe in one important way. In the Template step, we take an inventory of what is available in the body of a function or method. When designing a method, we have the following available to us:

- The fields of this object, accessed with accessor methods,
- The methods of this object, accessed by calling them,
- And the operations of the arguments, which are given by their *interfaces*.

For example, if a method takes an input `a-list` which is specified in the contract to be an `IList`, then we know that `(send a-list empty?)`, `(send a-list length)`, and so on.

6.5 Design Choices

Q: Which is considered a better design: a union with two variants, or a single variant with a Boolean field that indicates "which part of the union this data belongs to"? For example, is it better to have a `live-zombie%` and `dead-zombie%` class or a single `zombie%` class with a `dead?` field.

A: One of the themes of this and last semester is that *once you have settled on choice for the representation of information in your program, the structure of your program follows the structure of that data*. We've trained you to systematically derive code structure from data structure; there's a recipe—you don't even have to think. That's great because it frees up significant quantities of the most precious and limited resource

in computing: your brain. But unfortunately, that recipe kicks in only after you've chosen how information will be represented, i.e. after you've written down data definitions. Much of the hard work in writing programs, and where your creative energy and brain power is really needed, is going from information to representation. There is no recipe for this part of program design. You need to develop the ability to analyze problems, take an account of what knowledge needs to be represented to solve a problem, and practice making decisions about how that knowledge can be represented computationally. The good news is you don't have to be struck by divine inspiration to manage this step. Program design is a process of iterative refinement: make some choices, follow the recipe. You will discover ways to improve your initial choices, so go back and revise, then carry out the changes in code structure those design decisions entail. Rinse and repeat.

This is a long winded way of saying: there is no universal "right" answer to this question. It will depend on the larger context. That said, there are some important things to take into account for this particular question. It is much easier to add new variants to a union than it is to create Boolean values other than `true` and `false`. Good program design is often based on anticipating future requirements. If you think it's possible that there might be some other kind of zombie down the road, the union design will be less painful to extend and maintain.

6.6 Exercises

6.6.1 JSON, Jr.

JSON (JavaScript Object Notation) is a lightweight data-interchange format. It is easy for humans to read and write; it is based on a subset of the JavaScript Programming Language. There are a large number of huge corpora of data available on the internet. In order to write programs that can make use of this data, you'll need to design a data representation of JSON values.

A JSON value can take on the following forms:²

- An *object* is an unordered set of name/value pairs.
- An *array* is an ordered collection of values.
- A *value* can be a string, or a number, or true or false or null, or an object or an array. These structures can be nested.

This definition is written in the terminology of JSON, so don't confuse a JSON object

²This definition is drawn directly from <http://json.org/>

with an object in the class language. Likewise, don't confuse JSON strings with class strings.

To start things off simply, let's focus on a subset of JSON we'll call "JSON, Jr." which consists solely of strings and arrays. Using the notation of JSON, the following are examples of JSON, Jr. values:

- "this is JSON, Jr."
- ["this is JSON", "Jr."]
- [[], ["So", "is"], "this"]

The first example is just a string. The second is an array of two JSON elements, which are both strings. The third is another array, but of three elements: the first is an array of zero elements, the second is an array of two strings, and the third is a string.

1. Design an object-based data representation for JSON, Jr values.
2. Design a method for counting the number of strings in a JSON, Jr. value.

6.6.2 JSON

Revise the program you developed in the previous problem to handle all of JSON.

Design the following methods for JSON values:

- A method for counting the number of numbers in a JSON value.
- A method for summing all the numbers in a JSON value.
- A method for finding the length of the longest array in a JSON value.
- A method for computing the nesting depth of a JSON value.

Design the following methods for JSON objects:

- A method that works on JSON object values that takes a string and produces the JSON value associated with that string in the object, or `false` if no such value exists.
- A method that counts the number of name/value pairs in an object.
- A method that extends an object by adding a given name/value pair to an object.

- A method that restricts an object by subtracting a given name/value pair from an object.

Design the following methods for JSON arrays:

- A method of computing the length of the array.
- A method for indexing the *i*th element of an array.
- A method for reversing an array.

Design the following functions for randomly generating JSON values:

- Given a nesting depth, compute a random JSON value of at most that nesting depth. (It must be the case that if called repeatedly, eventual this function will produce a JSON value of *exactly* the given nesting depth, but it may not always produce a value nested so deep.)

Design an alternative data representation of JSON values that uses a subset of S-Expressions to model JSON values, which we'll call JSEN (JSON S-Expression Notation).

Design a function for converting from a JSEN representation to the object representation of that value. Design a method for JSON values that produces their JSEN representation.

Part II

Abstraction with Objects

Chapter 7

Abstraction via Delegation

7.1 Constructor design issue in modulo zombie (Assignment 3, Problem 3)

Course staff solution for regular zombie game:

world%

```
(define (teleport)
  (new world%
    (new player%
      (random WIDTH)
      (random HEIGHT))
    (this . zombies)
    (this . mouse)))
```

This has a significant bug: it always produces a plain `player%`, not a `modulo-player%`.

Bug (pair0MN):

modulo-player%

```
(define (teleport)
  (new player%
    (* -1 (random WORLD-SIZE))
    (* -1 (random WORLD-SIZE))))
```

This has a similar bug: it always produces a plain `player%`, not a `modulo-player%`. However, it's in the `modulo-player%` file, so there's an easy fix.

Lack of abstraction (pair0PQ):

modulo-player%

```
;; warp : Real Real -> ModuloPlayer
;; change the location of this player to the given location
(define (warp x y)
  (new modulo-player%
    (this . dest-x)
    (this . dest-y)
    x y))
```

player%

```
;; warp : Real Real -> Player
;; change the location of this player to the given location
(define (warp x y)
  (new player%
    (this . dest-x)
    (this . dest-y)
    x y))
```

This works correctly (this is the fix for the bug in Pair0MN's solution), but it duplicates code.

We want to fix these bugs without duplicating code.

Possible solutions (suggested in class):

- Parameterize the `teleport` method with a class name. Unfortunately, this doesn't work because the class name in `new` is not an expression.
- Use `this` as the class name. This doesn't work because `this` is an *instance*, not a *class*.

The solution is to add a new method to the interface, which constructs a new instance of the appropriate class. So, we add this method to the `player%` class:

```
(define (move x y)
  (new player% x y))
```

And this method to the `modulo-player%` class:

```
(define (move x y)
  (new modulo-player% x y))
```

Here's an example of the technique in full. We start with these classes:

7.1. CONSTRUCTOR DESIGN ISSUE IN MODULO ZOMBIE (ASSIGNMENT 3, PROBLEM 3)93

```
#lang class/1
(define-class s%
  (fields x y))

;; A Foo is one of:
;; - (new c% Number Number)
;; - (new d% Number Number)

(define-class c%
  (super s%)
  (define (make x y) (new c% x y))
  (define (origin) (new c% 0 0)))
(define-class d%
  (super s%)
  (define (make x y) (new d% x y))
  (define (origin) (new d% 0 0)))
```

Now we abstract the `origin` method to use `make`, and we can abstract `origin` to the superclass `s%`, since it becomes identical in both classes, avoiding the code duplication.

```
#lang class/1
(define-class s%
  (fields x y)
  (this . make 0 0))

;; A Foo is one of:
;; - (new c% Number Number)
;; - (new d% Number Number)

(define-class c%
  (super s%)
  (define (make x y)
    (new c% x y)))
(define-class d%
  (super s%)
  (define (make x y)
    (new d% x y)))

(new c% 50 100)
((new c% 50 100) . origin)
```

7.2 Abstracting list methods with different representations

Here is a parametric list interface definition:

```
;; =====
;; Parametric lists

;; A [Listof X] implements
;;
;; cons : X -> [Listof X]
;; Add the given element to the front of the list.
;;
;; empty : -> [Listof X]
;; Produce the empty list.
;;
;; length : -> Nat
;; Count the number of elements in this list.
;;
;; append : [Listof X] -> [Listof X]
;; Append the given list to the end of this list.
;;
;; reverse : -> [Listof X]
;; Reverse the order of elements in this list.
;;
;; map : [X -> Y] -> [Listof Y]
;; Construct the list of results of applying the function
;; to elements of this list.
;;
;; filter : [X -> Boolean] -> [Listof X]
;; Construct the list of elements in this list that
;; satisfy the predicate.
;;
;; foldr : [X Y -> Y] Y -> Y
;; For elements x_0...x_n, (f x_0 ... (f x_n b)).
;;
;; foldl : [X Y -> Y] Y -> Y
;; For elements x_0...x_n, (f x_n ... (f x_0 b)).
```

Here's the usual implementation of a small subset of this interface, first for the recursive union implementation:

```
(define-class cons%
  (fields first rest))
```

```

(define (cons x)
  (new cons% x this))

(define (empty)
  (new empty%))

(define (length)
  (add1 (this . rest . length)))

(define (foldr c b)
  (c (this . first)
     (this . rest . foldr c b)))

(define-class empty%

  (define (cons x)
    (new cons% x this))

  (define (empty)
    this)

  (define (length)
    0)

  (define (foldr c b)
    b))

```

And for the wrapper list implementation:

```

(define-class wlist%
  (fields ls)

  (define (cons x)
    (new wlist% (ls:cons x (this . ls))))

  (define (empty)
    (new wlist% ls:empty))

  (define (length)
    (ls:length (this . ls)))

  (define (foldr c b)
    (ls:foldr c b (this . ls))))

```

None of these look the same, so how can we abstract? Our abstraction design recipe for using inheritance requires that methods look identical in order to abstract them

into a common super class. But, for example, the `length` method looks like this for `wlist%`:

```
(define (length)
  (ls:length (this . ls)))
```

Like this for `empty%`:

```
(define (length)
  0)
```

And like this for `cons%`:

```
(define (length)
  (add1 (this . rest . length)))
```

In fact, all of them—but that’s a topic for another day.

Before we can abstract this method, we must make them all look the same. Fortunately, many list operations can be expressed using just a few simple operations, of which the most important is `foldr`. Here’s an implementation of `length` which just uses `foldr` and simple arithmetic.

```
(define (length)
  (this . foldr (λ (a b) (add1 b)) 0))
```

Note that this isn’t specific to any one implementation of lists—in fact, we can use it for any of them. This means that we can now abstract the method, creating a new `list%` class to share all of our common code:

```
(define-class list%
  (define (length)
    (this . foldr (λ (a b) (add1 b)) 0))
  ;; other methods here
)
```

The only methods that need to be implemented differently for different list versions are `empty` and `cons`, because they construct new lists, and `foldr`, because it’s the fundamental operation we use to build the other operations out of. It’s also helpful to implement `foldl`, since it’s fairly complex to factor out.

7.3 Delegation

So far, we’ve seen multiple ways to abstract repeated code. First, in Fundies 1, we saw functional abstraction, where we take parts of functions that differ and make them

parameters to the abstracted function. Second, in this class we've seen abstraction by using inheritance, where if methods in two related classes are identical, they can be lifted into one method in a common superclass.

However, can we still abstract common code without *either* of these mechanisms? Yes.

Consider the `class/0` language, *without* helper functions. We can write a binary tree class like this:

```
#lang class/0
;; A BT is one of:
;; - (new leaf% Number)
;; - (new node% Number BT BT)

;; double : Number -> BT
;; Double this tree and put the number on top.

(define-class leaf%
  (fields number)

  (define (double n)
    (new node% n this this)))

(define-class node%
  (fields number left right)

  (define (double n)
    (new node% n this this)))
```

Unfortunately, the `double` method is identical in both the `leaf%` and `node%` classes. How can we abstract this without using inheritance or a helper function?

One solution is to create a new class, and *delegate* the responsibility of doing the doubling to it. Below is an example of this:

```
(define-class helper%
  ;; Number BT -> BT
  ;; Double the given tree and puts the number on top.
  (define (double-helper number bt)
    (new node% number bt bt)))

(define tutor (new helper%))

(define-class leaf%
  (fields number)
```

```
(define (double n)
  (tutor . double-helper n this))

(define-class node%
  (fields number left right)

  (define (double n)
    (tutor . double-helper n this)))
```

The `helper%` class has just one method, although we could add as many as we wanted. We also need only one instance of `helper%`, called `tutor`, although we could create new instances when we needed them as well. Now the body of `double-helper` contains all of the doubling logic in our program, which might become much larger without needing duplicate code.

Chapter 8

Abstraction via Inheritance

8.1 Method inheritance with binary trees

We developed classes for representing binary trees and wrote a couple methods for binary trees, but one of the troubling aspects of this code is the fact that the two implementations of `double` are identical:

```
#lang class/0

;; A BT is one of:
;; - (new leaf% Number)
;; - (new node% Number BT BT)

(define-class leaf%
  (fields number)
  ...
  ;; Number -> BT
  ;; double the leaf and put the number on top
  (define (double n)
    (new node% n this this)))

(define-class node%
  (fields number left right)
  ...
  ;; Number -> BT
  ;; double the node and put the number on top
  (define (double n)
    (new node% n this this)))
```

If we think by analogy to the structural version of this code, we have something like this:

```
#lang class/0

;; A BT is one of:
;; - (make-leaf Number)
;; - (make-node Number BT BT)
(define-struct leaf (number))
(define-struct node (number left right))

;; BT Number -> BT
;; Double the given tree and put the number on top.
(define (double bt n)
  (cond [(leaf? bt) (make-node n bt bt)]
        [(node? bt) (make-node n bt bt)]))
```

We would arrive at this code by developing the `double` function according to the design recipe; in particular, this code properly instantiates the template for binary trees. However, after noticing the duplicated code, it is straightforward to rewrite this structure-oriented function into an equivalent one that duplicates no code. All cases of the `cond` clause produce the same result, hence the `cond` can be eliminated, replaced by a single occurrence of the duplicated answer expressions:

```
;; BT Number -> BT
;; Double the given tree and put the number on top.
(define (double bt n)
  (make-node n bt bt))
```

But switching back to the object-oriented version of this code, it is not so simple to “eliminate the `cond`”—there is no `cond`! We would like to write this code just once, but the real question is *where*? The solution, in this context, is to *lift* the identical method definitions to a common *super* class. That is, we define a third class that contains the method shared among `leaf%` and `node%`:

```
(define-class bt%
  ;; -> BT
  ;; Double this tree and put the number on top.
  (define (double n)
    (new node% n this this)))
```

The `double` method can be removed from the `leaf%` and `node%` classes and instead these class can rely on the `bt%` definition of `double`, but to do this we must establish a relationship between `leaf%`, `node%` and `bt%`: we declare that `leaf%` and `node%` are *subclasses* of `bt%`, and therefore they *inherit* the `double` method; it is as if the code were duplicated without actually writing it twice:

```

(define-class leaf%
  (super bt%)
  (fields number)
  ;; -> Number
  ;; count the number of numbers in this leaf
  (define (count)
    1))

(define-class node%
  (super bt%)
  (fields number left right)
  ;; -> Number
  ;; count the number of numbers in this node
  (define (count)
    (+ 1
       (send (send this left) count)
       (send (send this right) count))))

```

8.2 The class/1 language

To accommodate this new feature—*inheritance*—we need to adjust our programming language. We'll now program in `class/1`, which is a superset of `class/0`—all `class/0` programs are `class/1` programs, but not vice versa. The key difference is the addition of the `(super class-name)` form.

At this point we can construct binary trees just as before, and all binary trees understand the `double` method even though it is only defined in `bt%`:

```

> (new leaf% 7)
(new leaf% 7)
> (send (new leaf% 7) double 8)
(new node% 8 (new leaf% 7) (new leaf% 7))
> (send (send (new leaf% 7) double 8) double 9)
(new node% 9 (new node% 8 (new leaf% 7) (new leaf% 7)) (new node% 8 (new
leaf% 7) (new leaf% 7)))

```

There are a couple other features of the `class/1` language that are worth knowing about. One is a trivial, but very handy shorthand form for writing `send`. The shorthand form is to write `(o . m)` to call method `m` on object `o`, that is, `(o . m)` is equivalent to `(send o m)`. Another nice feature of the “dot notation” is that it makes it easy to stack up a bunch of method calls, so for example `(o . m . n x y . p)` is shorthand for

```
(send (send (send o m) n x y) p)
```

From here on out, the book will use the dot notation since it's so nice.

8.3 “Abstract” classes

At this point, it is worth considering the question: what does a `bt%` value represent? We have arrived at the `bt%` class as a means of abstracting identical methods in `leaf%` and `node%`, but if we say `(new bt%)`, as we surely can, what does that *mean*? The answer is: *nothing*.

Going back to our data definition for BTs, it's clear that the value `(new bt%)` is **not** a BT since a BT is either a `(new leaf% Number)` or a `(new node% Number BT BT)`. In other words, a `(new bt%)` makes no more sense as a representation of a binary tree than does `(new node% "Hello Fred" 'y-is-not-a-number add1)`. With that in mind, it doesn't make sense for our program to ever construct `bt%` objects—they exist purely as an abstraction mechanism. Some languages, such as Java, allow you to enforce this property by declaring a class as “abstract”; a class that is declared abstract cannot be constructed. Our language will not enforce this property, much as it does not enforce contracts. Again we rely on data definitions to make sense of data, and `(new bt%)` doesn't make sense.

8.4 Data inheritance with binary trees

Inheritance allows us to share methods amongst classes, but it also possible to share data. Just as we observed that `double` was the same in both `leaf%` and `node%`, we can also observe that there are data similarities between `leaf%` and `node%`. In particular, both `leaf%` and `node%` contain a `number` field. This field can be abstracted just like `double` was—we can lift the field to the `bt%` super class and eliminate the duplicated field in the subclasses:

```
(define-class bt%
  (fields number)
  ;; -> BT
  ;; Double this tree and put the number on top.
  (define (double n)
    (new node% n this this)))

(define-class leaf%
  (super bt%)
  ;; -> Number
  ;; count the number of numbers in this leaf
  (define (count)
    1))
```

```

(define-class node%
  (super bt%)
  (fields left right)
  ;; -> Number
  ;; count the number of numbers in this node
  (define (count)
    (+ 1
      (this . left . count)
      (this . right . count))))

```

The `leaf%` and `node%` class now inherit both the `number` field and the `double` method from `bt%`. This has a consequence for constructing new instances. Previously it was straightforward to construct an object: you write down `new`, the class name, and as many expressions as there are fields in the class. But now that a class may inherit fields, you must write down as many expressions as there are fields in the class definition itself and in all of the super classes. What's more, the order of arguments is important. The fields defined in the class come first, followed by the fields in the immediate super class, followed by the super class's super classes, and so on. Hence, we still construct `leaf%`s as before, but the arguments to `new` for `node%` are changed: it takes the left subtree, the right subtree, and *then* the number at that node:

```

;; A BT is one of:
;; - (new leaf% Number)
;; - (new node% BT BT Number)

> (new leaf% 7)
(new leaf% 7)
> (new node% (new leaf% 7) (new leaf% 13) 8)
(new node% (new leaf% 7) (new leaf% 13) 8)

```

Although none of our method so far have needed to access the `number` field, it is possible to access `number` in `leaf%` and `node%` (and `bt%`) methods *as if* they had their own `number` field. Let's write a `sum` method to make it clear:

```

(define-class bt%
  (fields number)
  ;; -> BT
  ;; Double this tree and put the number on top.
  (define (double n)
    (new node% this this n)))

(define-class leaf%
  (super bt%)
  ;; -> Number

```

Notice that the `double` method swapped the order of arguments when constructing a new `node%` to reflect the fact that the node constructor now takes values for its fields first, then values for its inherited fields.

```

;; count the number of numbers in this leaf
(define (count)
  1)

;; -> Number
;; sum all the numbers in this leaf
(define (sum)
  (this . number)))

(define-class node%
  (super bt%)
  (fields left right)
  ;; -> Number
  ;; count the number of numbers in this node
  (define (count)
    (+ 1
      (this . left . count)
      (this . right . count)))

  ;; -> Number
  ;; sum all the numbers in this node
  (define (sum)
    (+ (this . number)
      (this . left . sum)
      (this . right . sum))))

```

As you can see, both of the `sum` methods refer to the `number` field, which is inherited from `bt%`.

8.5 Inheritance with shapes

Let's consider another example and see how data and method inheritance manifests. This example will raise some interesting issues for how super classes can invoke the methods of its subclasses. Suppose we are writing a program that deals with shapes that have position. To keep the example succinct, we'll consider two kinds of shapes: circles and rectangles. This leads us to a union data definition (and class definitions) of the following form:

```

;; A Shape is one of:
;; - (new circ% +Real Real Real)
;; - (new rect% +Real +Real Real Real)
;; A +Real is a positive, real number.
(define-class circ%

```

We are using `+Real` to be really precise about the kinds of numbers that are allowable to make for a sensible notion of a shape. A circle with radius `-5` or `3+2i` doesn't make a whole lot of sense.


```

(fields radius x y))
(define-class rect%
  (fields width height x y))

```

Already we can see an opportunity for data abstraction since `circ%` and `rect%` both have `x` and `y` fields. Let's define a super class and inherit these fields:

```

;; A Shape is one of:
;; - (new circ% +Real Real Real)
;; - (new rect% +Real +Real Real Real)
;; A +Real is a positive, real number.
(define-class shape%
  (fields x y))
(define-class circ%
  (super shape%)
  (fields radius))
(define-class rect%
  (super shape%)
  (fields width height))

```

Now let's add a couple of methods: `area` will compute the area of the shape, and `draw-on` will take a scene and draw the shape on the scene at the appropriate position:

```

(define-class circ%
  (super shape%)
  (fields radius)

  ;; -> +Real
  (define (area)
    (* pi (sqr (this . radius)))))

;; Scene -> Scene
;; Draw this circle on the scene.
(define (draw-on scn)
  (place-image (circle (this . radius) "solid" "black")
    (this . x)
    (this . y)
    scn)))

(define-class rect%
  (super shape%)
  (fields width height)

  ;; -> +Real
  ;; Compute the area of this rectangle.


```

```
(define (area)
  (* (this . width)
     (this . height)))


;; Scene -> Scene
;; Draw this rectangle on the scene.
(define (draw-on scn)
  (place-image (rectangle (this . width) (this . height) "solid" "black")
               (this . x)
               (this . y)
               scn)))
```

Examples:

```
> (send (new circ% 30 75 75) area)
2827.4333882308138
> (send (new circ% 30 75 75) draw-on (empty-scene 150 150))
```



```
> (send (new rect% 30 50 75 75) area)
1500
> (send (new rect% 30 50 75 75) draw-on (empty-scene 150 150))
```



The `area` method is truly different in both variants of the shape union, so we shouldn't attempt to abstract it by moving it to the super class. However, the two definitions of the `draw-on` method are largely the same. If they were *identical*, it would be easy to abstract the method, but until the two methods are identical, we cannot lift the definition to the super class. One way forward is to rewrite the methods by pulling out the parts that differ and making them separate methods. What differs between these two methods is the expression constructing the image of the shape, which suggests defining a new method `img` that constructs the image. The `draw-on` method can now call `img` and rewriting it this way makes both `draw-on` methods identical; the method can now be lifted to the super class:

```
(define-class shape%
  (fields x y)

  ;; Scene -> Scene
  ;; Draw this shape on the scene.
  (define (draw-on scn)
    (place-image (img)
                  (send this x)
                  (send this y)
                  scn)))
```

But there is a problem with this code. While this code makes sense when it occurs inside of `rect%` and `circ%`, it doesn't make sense inside of `shape%`. In particular, what does `img` mean here? The `img` method is a method of `rect%` and `circ%`, but not of `shape%`, and therefore the name `img` is unbound in this context.

On the other hand, observe that all shapes are either `rect%`s or `circ%`s. We therefore *know* that the object invoking the `draw-on` method understands the `img` message, since both `rect%` and `circ%` implement the `img` method. Therefore we can use `send` to invoke the `img` method on `this` object and thanks to our data definitions for shapes, it's guaranteed to succeed. (The message send would fail if `this` referred to a `shape%`, but remember that `shape%`s don't make sense as objects in their own right and should never be constructed).

We arrive at the following final code:

```
#lang class/1
(require 2htdp/image)

;; A Shape is one of:
;; - (new circ% +Real Real Real)
;; - (new rect% +Real +Real Real Real)
;; A +Real is a positive, real number.
(define-class shape%
  (fields x y)
```

```

;; Scene -> Scene
;; Draw this shape on the scene.
(define (draw-on scn)
  (place-image (send this img)
               (send this x)
               (send this y)
               scn)))

(define-class circ%
  (super shape%)
  (fields radius)

  ;; -> +Real
  ;; Compute the area of this circle.
  (define (area)
    (* pi (sqr (send this radius))))

  ;; -> Image
  ;; Render this circle as an image.
  (define (img)
    (circle (send this radius) "solid" "black")))

(define-class rect%
  (super shape%)
  (fields width height)

  ;; -> +Real
  ;; Compute the area of this rectangle.
  (define (area)
    (* (send this width)
       (send this height)))

  ;; -> Image
  ;; Render this rectangle as an image.
  (define (img)
    (rectangle (send this width) (send this height) "solid" "black")))

(check-expect (send (new rect% 10 20 0 0) area)
              200)
(check-within (send (new circ% 10 0 0) area)
              (* pi 100)
              0.0001)
(check-expect (send (new rect% 5 10 10 20) draw-on
                    (empty-scene 40 40))
              (place-image (rectangle 5 10 "solid" "black")

```

```

10 20
    (empty-scene 40 40)))
(check-expect (send (new circ% 4 10 20) draw-on
    (empty-scene 40 40))
    (place-image (circle 4 "solid" "black")
        10 20
        (empty-scene 40 40)))

```

8.6 Revisiting the Rocket with Inheritance

At this point, you may recall that unsettling feeling you had in the discussion of section 3.4, in which we wrote duplicate, identical methods in both the `landing%` and `takeoff%` variants of `Rocket` objects:

`landing% and takeoff%`

```

;; render : -> Scene
;; Render this rocket as a scene.
(define (render)
  (send this draw-on MT-SCENE))

; draw-on : Scene -> Scene
; Draw this rocket on to scene.
(define (draw-on scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 (send this dist))
    scn))

```

We now have the mechanism to eliminate this duplication. We can define a super class of `landing%` and `takeoff%` called `rocket%` and lift the methods to this class. Moreover, since there is duplication in the data of these classes, we can likewise lift the `dist` field to `rocket%`:

```

(define-class rocket%
  (fields dist)

  ;; render : -> Scene
  ;; Render this rocket as a scene.
  (define (render)
    (this . draw-on MT-SCENE))

  ;; draw-on : Scene -> Scene
  ;; Draw this rocket on to scene.

```

```

(define (draw-on scn)
  (overlay/align/offset "center" "bottom"
    ROCKET
    0 (add1 (this . dist))
    scn)))

```

To complete the revised program, the `landing%` and `takeoff%` classes should declare `rocket%` as a super class and remove the `dist` field and `render` and `draw-on` methods:

```

(define-class landing%
  (super rocket%)
  ...)
(define-class takeoff%
  (super rocket%)
  ...)

```

8.7 Exercises

8.7.1 Abstract Lists

Revisit your solution to the section 6.4.1 exercise.

Use inheritance to lift method definitions to a super class to the full extent possible. (*Hint:* it will help if you realize that many of these methods may be expressed in terms of a few “core” methods.) If possible, have both the recursive union representation and the wrapper representation share a common super class.

The `cons` and `empty` methods have been added to facilitate opportunities for abstraction. You might find them useful to use when you lift methods to a common super class so that the right kind of list (either a wrapped or a recursive union list) is constructed.

8.7.2 Shapes

Here is the signature for a method to compute the area of a shape’s bounding box—the smallest rectangle that can contain the shape.

```

;; bba : -> Number      (short for "bounding-box-area")
;; Compute the area of the smallest bounding box for this shape.

```

Here are some examples of how `bba` should work:

section ???

section ???

```
(check-expect ((new rect% 3 4) . bba) 12)
(check-expect ((new circ% 1.5) . bba) 9)
```

1. Design the `bba` method for the `rect%` and `circ%` class.
2. Design a super class of `rect%` and `circ%` and lift the `bba` method to the super class. Extend the shape interface as needed, but implement any methods you add.
3. Design a new variant of a Shape, Square, which should support all of the methods of the interface.

Part III

Invariants

Chapter 9

Invariants, Testing, and Abstraction Barriers

9.1 Invariants of Data Structures

Here's an interface for a sorted list of numbers.

```
#lang class/1
;; An ISorted implements
;; insert : Number -> Sorted
;; contains? : Number -> Boolean
;; ->list : -> [Listof Number]
;; empty? : -> Boolean

;; Invariant: The list is sorted in ascending order.

;; Precondition: the list must not be empty
;; max : -> Number
;; min : -> Number
```

How would we implement this interface?

We can simply adopt the recursive union style that we've already seen for implementing lists. Here we see the basic definition as well as the implementation of the `contains?` method.

```
#lang class/1
```

```

;; A Sorted is one of
;; - (new smt%)
;; - (new scon% Number Sorted)
(define-class smt%
  (check-expect ((new smt%) . contains? 5) false)
  (define (contains? n)
    false))

(define-class scon%
  (fields first rest)
  (check-expect ((new scon% 5 (new smt%)) . contains? 5) true)
  (check-expect ((new scon% 5 (new smt%)) . contains? 7) false)
  (check-expect ((new scon% 5 (new scon% 7 (new smt%))) . contains? 3)
    false)
  (check-expect ((new scon% 5 (new scon% 7 (new smt%))) . contains? 9)
    false)
  (define (contains? n)
    (or (= n (field first))
        ((field rest) . contains? n))))

```

However, we can write a new implementation that uses our invariant to avoid checking the rest of the list when it isn't necessary.

```

(define (contains? n)
  (cond [(= n (field first)) true]
        [(< n (field first)) false]
        [else ((field rest) . contains? n)]))

```

Because the list is always sorted in ascending order, if `n` is less than the first element, it must be less than every other element, and therefore can't possibly be equal to any element in the list.

Now we can implement the remaining methods from the interface. First, `insert`

smt%

```

(check-expect ((new smt%) . insert 5)
  (new scon% 5 (new smt%)))
(define (insert n)
  (new scon% n (new smt%)))

```

scon%

```

(check-expect ((new scon% 5 (new smt%)) . insert 7)
  (new scon% 5 (new scon% 7 (new smt%))))
(check-expect ((new scon% 7 (new smt%)) . insert 5)

```

```

                (new scons% 5 (new scons% 7 (new smt%)))
(define (insert n)
  (cond [(< n (field first))
        (new scons% n this)]
        [else
         (new scons%
              (field first)
              ((field rest) . insert n))]))

```

Note that we don't have to look at the whole list to insert the elements. This is again a benefit of programming using the invariant that we have a sorted list.

Next, the `max` method. We don't have to do anything for the empty list, because we have a precondition that we can only call `max` when the list is non-empty.

scons%

```

(define real-max max)
(check-expect ((new scons% 5 (new smt%)) . max) 5)
(check-expect ((new scons% 5 (new scons% 7 (new smt%))) . max) 7)
(define (max)
  (cond [(field rest) . empty?] (field first)]
        [else ((field rest) . max)]))

```

Again, this implementation relies on our data structure invariant. To make this work, though, we need to implement `empty?`.

smt%

```

(check-expect ((new smt%) . empty?) true)
(define (empty?) true)

```

scons%

```

(check-expect ((new scons% 1 (new smt%)) . empty?) false)
(define (empty?) false)

```

The final two methods are similar. Again, we don't implement `min` in `smt%`, because of the precondition in the interface.

smt%

```

;; no min method
(define (->list) empty)

```

scons%

```

(define (min) (field first))
(define (->list)
  (cons (field first) ((field rest) . ->list)))

```

9.2 Properties of Programs and Randomized Testing

A *property* is a claim about the behavior of a program. Unit tests check particular, very specific properties, but often there are more general properties that we can state and check about programs.

Here's a property about our sorted list library, which we would like to be true:

$$\forall \text{sls} : \text{ISorted} . \forall n : \text{Number} . ((\text{sls} . \text{insert } n) . \text{contains? } n)$$

How would we check this? We can check a few instances with unit tests, but this property makes a very strong claim. If we were working in ACL2, as in the Logic and Computation class, we could provide a machine-checked proof of the property, verifying that it is true for every single Sorted and Number.

For something in between these two extremes, we can use *randomized testing*. This allows us to gain confidence that our property is true, with just a bit of programming effort.

First, we want to write a program that asks the question associated with this property.

```
;; Property: forall sorted lists and numbers, this predicate holds
;; insert-contains? : ISorted Number -> Boolean
(define (insert-contains? sls n)
  ((sls . insert n) . contains? n))
```

Now we make lots of randomly generated tests, and see if the predicate holds. First, let's build a random sorted list generator.

```
;; build-sorted : Nat (Nat -> Number) -> Sorted
(define (build-sorted i f)
  (cond [(zero? i) (new smt%)]
        [else
         (new scon%
          (f i)
          (build-sorted (sub1 i) f))]))
(build-sorted 5 (lambda (x) x))
```

Oh no! We broke the invariant. The `scon%` constructor allows you to break the invariant, and now all of our methods don't work. Fortunately, we can implement a fixed version that uses the `insert` method to maintain the sorted list invariant:

```
;; build-sorted : Nat (Nat -> Number) -> Sorted
(define (build-sorted i f)
```

```

(cond [(zero? i) (new smt%)]
      [else
       ((build-sorted (sub1 i) f) . insert (f i))]))
(check-expect (build-sorted 3 (lambda (x) x))
              (new scon% 1 (new scon% 2 (new scon% 3 (new smt%)))))

```

Now `build-sorted` produces the correct answer, which we can easily verify at the Interactions window.

Using `build-sorted`, we can develop `random-sorted`, which generates a sorted list of random numbers.:

```

;; Nat -> Sorted
(define (random-sorted i)
  (build-sorted i (lambda (_) (random 100))))

```

Given these building blocks, we can write a test that checks our property.

```

(check-expect (insert-contains? (random-sorted 30) (random 50))
              true)

```

Every time we hit the Run button, we generate a random sorted list of numbers, and check if a particular random integer behaves appropriately when `inserted` into it. But if we could repeatedly check this property hundreds or thousands of times, it would be even more unlikely that our program violates the property. After all, we could have just gotten lucky.

First, we write a function to perform some action many times:

```

;; Nat (Any -> Any) -> 'done
;; run the function f i times
(define (do i f)
  (cond [(zero? i) 'done]
        [else (f (do (sub1 i) f))]))

```

Then we can run our test many times:

```

(do 1000
  (lambda (_)
    (check-expect (insert-contains? (random-sorted 30) (random 50))
                  true)))

```

When this says that we've passed 1000 tests, we'll be more sure that we've gotten our function right.

What if we change our property to this untrue statement?

```
;; Property: forall sorted lists and numbers, this predicate holds
;; insert-contains? : ISorted Number -> Boolean
(define (insert-contains? sls n)
  (sls . contains? n))
```

Now we get lots of test failures, but the problem is *not* in our implementation of sorted lists, it's in our property definition. If we had instead had a bug in our implementation, we would have similarly seen many failures. Thus, it isn't always possible to tell from a test failure, or even many failures, whether it's the code or the specification is wrong—you have to look at the test failure to check.

This is why it's extremely important to get your specifications (like contracts, data definitions, and interface definitions) correct. Your program can only be correct if they are.

9.3 Abstraction Barriers and Modules

Recall that in our original version of `build-sorted`, we saw that the `scons%` constructor allowed us to violate the invariant—it didn't check that the value provided for `first` was at least as small as the elements of `rest`. We would like to prevent clients of our sorted list implementation from having access to this capability, so that we can be sure that our invariant is maintained.

To address this, we set up an *abstraction barrier*, preventing other people from seeing the `scons%` constructor. To create these barriers, we use a *module system*. We will consider our implementation of sorted lists to be one module, and we can add a simple specification to allow other modules to see parts of the implementation (but not all of it).

Modules in our languages are very simple—you've already written them. They start with `#lang class/N` and cover a whole file.

Here's the module implementing our sorted list, which we save in a file called "sorted-list.rkt".

sorted-list.rkt

```
#lang class/1

;; ... all of the rest of the code ...
```



```
(define smt (new smt%))
(provide smt)
```

We've added two new pieces to this file. First, we define `smt` to be an instance of the empty sorted list. Then, we use `provide` to make `smt`, but not *any* other definition from our module, available to other modules.

Therefore, the *only* part of our code that the rest of the world can see is the `smt` value. To add new elements, the rest of the world has to use the `insert` method.

```
#lang class/1
(require "sorted-list.rkt")

(smt . insert 4)
```

Here, we've used `require`, which we've used to refer to libraries that come with DrRacket. However, we can specify the name of a file, and we get everything that the module in that file `provides`, which here is just the `smt` definition. Everything else, such as the dangerous `scons%` constructor, is hidden, and our implementation of sorted lists can rely on its invariant.

9.4 Exercises

9.4.1 Quick Lists

Van Horn has always been underwhelmed by the fact that `list-ref` is such a slow operation when you're accessing elements deep down in a big list. Why should it take a million `rests` just to get the millionth element?

To combat this drawback of an otherwise lovely data structure, the list, Van Horn has devised an idea for a new implementation of lists that would let you get the millionth element in about 20 operations. If his idea works, the `list-ref` operation will take roughly $\log(i)$ steps to get the i th element. The other list operations, on the other hand, would remain more or less just as efficient as before; taking the rest of a list, for example, might take a few more steps to compute, but it would be some small constant number of extra steps. In the end, we'd have something that behaves just like a list, but with a much better `list-ref` operation.

Your task is to take Van Horn's idea and implement it. Since you'll be building a new kind of list data structure, let's first agree on the list interface we want:

```
;; A [List X] implements
;; - cons : X -> [List X]
;;   Cons given element on to this list
```

```
;; - first : -> X
;;   Get the first element of this list
;;   (only defined on non-empty lists)
;; - rest : -> [List X]
;;   Get the rest of this
;;   (only defined on non-empty lists)
;; - list-ref : Natural -> X
;;   Get the ith element of this list
;;   (only defined for lists of i+1 or more elements)
;; - length : -> Natural
;;   Compute the number of elements in this list

;; empty is a [List X] for any X.
```

In other words, you have to make an object named `empty` that implements the list interface above. Lists should work just like we’re used to, so for example, these tests should all pass if `empty` is appropriately defined:

```
#lang class/1
(require "your-implementation-of-lists.rkt") ; provides empty

(define ls (empty . cons 'a . cons 'b . cons 'c . cons 'd . cons 'e))

(check-expect (empty . length) 0)
(check-expect (ls . length) 5)
(check-expect (ls . first) 'e)
(check-expect (ls . rest . first) 'd)
(check-expect (ls . rest . rest . first) 'c)
(check-expect (ls . rest . rest . rest . first) 'b)
(check-expect (ls . rest . rest . rest . rest . first) 'a)

(check-expect (ls . list-ref 0) 'e)
(check-expect (ls . list-ref 1) 'd)
(check-expect (ls . list-ref 2) 'c)
(check-expect (ls . list-ref 3) 'b)
(check-expect (ls . list-ref 4) 'a)
```

So now let’s talk about Van Horn’s idea.

Van Horn thinks if instead of representing a list as a “list of elements” you could do better by representing a list as a “forest of trees of elements”. (A *forest* is just an arbitrarily long sequence of trees.) Moreover, the trees will get bigger and bigger as you go deeper into the forest, and every tree is full. (A *full tree* is a binary in which every node has a left and right subtree that are full and of the same.) For the moment, don’t worry about *why* this makes `list-ref` fast—think about that after you’ve implemented Van Horn’s idea.

So here are the key invariants of a *quick list*:

- A quick list is a forest of increasingly large full binary trees.
- With the possible exception of the first two trees, every successive tree is *strictly* larger.

Now that we have the invariant, let's talk about the operations and how they both can use and maintain the invariant.

First, `first`. Since the list must be non-empty, we know the forest has at least one tree, so we can get the first element of the list by getting "the first" element of the tree, which for quick lists, will be the top element.

Now, `length`. If the forest is empty, the list has length 0. If a forest has a tree, the length of the list is the size of the tree plus the size of the rest of the forest. (It's useful to store the size of a tree separately from a tree so that you don't have to compute it every time you need it.)


The `list-ref` method works as follows: if the index is 0, the list must be non-empty, so take the first element, i.e. the top element of the first tree in the forest. If the index is non-zero, there are two cases: if it's less than the size of the first tree, the element is in that tree, so fetch it from the first tree. If it's larger, adjust the index, and look in the remaining trees of the forest.

To fetch an element from a tree: if the index is zero, the element is the top element. Otherwise, if the index is less than half the size, it's on the left side; if the index is greater than half, it's on the right. (You might do yourself a favor and develop `tree-ref` for full binary trees and get it working and thoroughly tested before attempting `list-ref`.)


These element-producing operations considered so far have used the invariant. Now let's turn to the list-producing operations which must maintain it.

When an element is `consed`, there are two cases to consider:

- If there are at least two trees in the forest and the first two trees are the same size, then make a new tree out of these two and with the given element on top. Here it is pictorially; we are given a forest of full binary trees where the first two trees have the same height:




To `cons` on the new element, we make a node that contains the element and the first two trees as its left and right subtree:




Notice how this first tree is necessarily full, since the first two trees were full and the same height; notice how this new first tree in the forest is at most as large as the second tree (previously the third tree). These two observations demonstrate that the invariant holds on the resulting forest, so `cons` really makes a quick list in this case.

- Otherwise, we know that the size of the trees in the forest is strictly increasing:




Therefore, we can just make a new tree with one element and make it the first tree in the forest:




Notice how the one element tree is obviously full and that it is no larger than the (now) second tree in the forest, so the invariant holds in this case too.

To take the [rest](#) of a list, there must be at least one tree in the forest (since the list is non-empty):



We want to split this tree into its left and right and make these the first two trees in the forest. The element that was on top is dropped on the floor and we're left with a representation of the rest of the list:



And that's that. When writing your code you want to make sure the invariants are always true. Good code should make this fact obvious; bad code, not so much.

This is a nice little exercise in data structure design and implementation, and although Van Horn *wishes* this were really his idea, he actually got it from reading a book by Chris Okasaki, who has designed a bunch of these kinds of data structures. Go forth, and may your [list-ref](#) never be slow again.

Part IV

Schemes of a Larger Design

Chapter 10

Larger system design: Snakes on a plane

So far, we have introduced a lot of important new concepts such as interfaces and data and method inheritance for class-based abstraction. In this chapter, we are going to see these concepts being applied in the context of a larger program design. It's a game we've all seen and designed before; we're going to develop a class-based version of the Snake Game.

10.1 Information in the Snake Game

Our first task in designing the Snake Game is to take an account of the information that our program will need to represent. This includes:

- a system of coordinates
- a snake, which has
 - direction
 - segments
- food
- a world

10.2 The world

Let's start by designing a minimal `world%` class. We will iteratively refine it later to add more and more features. For now, let's just have the snake move.

```
;; A World is a (new world% Snake Food).
(define-class world%
  (fields snake food)

  (define (on-tick)
    (new world%
      (send (send this snake) move)
      (send this food)))

  (define (tick-rate) 1/8)

  (define (to-draw)
    (send (send this food) draw)
    (send (send this snake) draw MT-SCENE))))
```

10.3 Coordinate interface

Let's focus on the system of coordinates. There are really two coordinate systems we will need to represent; one is necessitated by the animation system we are using, the other by the logic of the Snake Game. We are *consumers* of the `big-bang` animation system, which uses a pixel-based, graphics-coordinate system (meaning the origin is at the Northwest corner). This is part of `big-bang`'s interface, which we don't have the power to change, and therefore we have to communicate to `big-bang` using pixels in graphics-coordinates. In general, when we design programs, the interfaces of libraries we use impose obligations on our code.

On the other hand, using pixel-based graphics-coordinates is probably not the best representation choice for the information of the Snake Game. We'll be better off if we design our own representation and have our program translate between representations at the communication boundaries between our code and `big-bang`. Let's use a grid-based coordinate system where the origin is the Southwest corner. We can define a mapping between the coordinate systems by defining some constants such as the grid size and the size of the screen:

```
#lang class/0
(define WIDTH 32) ; in grid units
(define HEIGHT 32) ; in grid units
(define SIZE 16) ; in pixels / grid unit
```

```
(define WIDTH-PX (* SIZE WIDTH)) ; in pixels  
(define HEIGHT-PX (* SIZE HEIGHT)) ; in pixels
```

This defines that our game is logically played on a 32x32 grid, which we will render visually as a 512x512 pixel image. These are the values we cooked up in class, which results in the following grid for our game:

As an exercise, try to write an expression that produces this image.

0,31	1,31	2,31	3,31	4,31	5,31	6,31	7,31	8,31	9,31	10,31	11,31	12,31	13,31	14,31	15,31	16,31	17,31	18,31	19,31	20,31	21,31	22,31	23,31	24,31	25,31	26,31	27,31	28,31
0,30	1,30	2,30	3,30	4,30	5,30	6,30	7,30	8,30	9,30	10,30	11,30	12,30	13,30	14,30	15,30	16,30	17,30	18,30	19,30	20,30	21,30	22,30	23,30	24,30	25,30	26,30	27,30	28,30
0,29	1,29	2,29	3,29	4,29	5,29	6,29	7,29	8,29	9,29	10,29	11,29	12,29	13,29	14,29	15,29	16,29	17,29	18,29	19,29	20,29	21,29	22,29	23,29	24,29	25,29	26,29	27,29	28,29
0,28	1,28	2,28	3,28	4,28	5,28	6,28	7,28	8,28	9,28	10,28	11,28	12,28	13,28	14,28	15,28	16,28	17,28	18,28	19,28	20,28	21,28	22,28	23,28	24,28	25,28	26,28	27,28	28,28
0,27	1,27	2,27	3,27	4,27	5,27	6,27	7,27	8,27	9,27	10,27	11,27	12,27	13,27	14,27	15,27	16,27	17,27	18,27	19,27	20,27	21,27	22,27	23,27	24,27	25,27	26,27	27,27	28,27
0,26	1,26	2,26	3,26	4,26	5,26	6,26	7,26	8,26	9,26	10,26	11,26	12,26	13,26	14,26	15,26	16,26	17,26	18,26	19,26	20,26	21,26	22,26	23,26	24,26	25,26	26,26	27,26	28,26
0,25	1,25	2,25	3,25	4,25	5,25	6,25	7,25	8,25	9,25	10,25	11,25	12,25	13,25	14,25	15,25	16,25	17,25	18,25	19,25	20,25	21,25	22,25	23,25	24,25	25,25	26,25	27,25	28,25
0,24	1,24	2,24	3,24	4,24	5,24	6,24	7,24	8,24	9,24	10,24	11,24	12,24	13,24	14,24	15,24	16,24	17,24	18,24	19,24	20,24	21,24	22,24	23,24	24,24	25,24	26,24	27,24	28,24
0,23	1,23	2,23	3,23	4,23	5,23	6,23	7,23	8,23	9,23	10,23	11,23	12,23	13,23	14,23	15,23	16,23	17,23	18,23	19,23	20,23	21,23	22,23	23,23	24,23	25,23	26,23	27,23	28,23
0,22	1,22	2,22	3,22	4,22	5,22	6,22	7,22	8,22	9,22	10,22	11,22	12,22	13,22	14,22	15,22	16,22	17,22	18,22	19,22	20,22	21,22	22,22	23,22	24,22	25,22	26,22	27,22	28,22
0,21	1,21	2,21	3,21	4,21	5,21	6,21	7,21	8,21	9,21	10,21	11,21	12,21	13,21	14,21	15,21	16,21	17,21	18,21	19,21	20,21	21,21	22,21	23,21	24,21	25,21	26,21	27,21	28,21
0,20	1,20	2,20	3,20	4,20	5,20	6,20	7,20	8,20	9,20	10,20	11,20	12,20	13,20	14,20	15,20	16,20	17,20	18,20	19,20	20,20	21,20	22,20	23,20	24,20	25,20	26,20	27,20	28,20
0,19	1,19	2,19	3,19	4,19	5,19	6,19	7,19	8,19	9,19	10,19	11,19	12,19	13,19	14,19	15,19	16,19	17,19	18,19	19,19	20,19	21,19	22,19	23,19	24,19	25,19	26,19	27,19	28,19
0,18	1,18	2,18	3,18	4,18	5,18	6,18	7,18	8,18	9,18	10,18	11,18	12,18	13,18	14,18	15,18	16,18	17,18	18,18	19,18	20,18	21,18	22,18	23,18	24,18	25,18	26,18	27,18	28,18
0,17	1,17	2,17	3,17	4,17	5,17	6,17	7,17	8,17	9,17	10,17	11,17	12,17	13,17	14,17	15,17	16,17	17,17	18,17	19,17	20,17	21,17	22,17	23,17	24,17	25,17	26,17	27,17	28,17
0,16	1,16	2,16	3,16	4,16	5,16	6,16	7,16	8,16	9,16	10,16	11,16	12,16	13,16	14,16	15,16	16,16	17,16	18,16	19,16	20,16	21,16	22,16	23,16	24,16	25,16	26,16	27,16	28,16
0,15	1,15	2,15	3,15	4,15	5,15	6,15	7,15	8,15	9,15	10,15	11,15	12,15	13,15	14,15	15,15	16,15	17,15	18,15	19,15	20,15	21,15	22,15	23,15	24,15	25,15	26,15	27,15	28,15
0,14	1,14	2,14	3,14	4,14	5,14	6,14	7,14	8,14	9,14	10,14	11,14	12,14	13,14	14,14	15,14	16,14	17,14	18,14	19,14	20,14	21,14	22,14	23,14	24,14	25,14	26,14	27,14	28,14
0,13	1,13	2,13	3,13	4,13	5,13	6,13	7,13	8,13	9,13	10,13	11,13	12,13	13,13	14,13	15,13	16,13	17,13	18,13	19,13	20,13	21,13	22,13	23,13	24,13	25,13	26,13	27,13	28,13
0,12	1,12	2,12	3,12	4,12	5,12	6,12	7,12	8,12	9,12	10,12	11,12	12,12	13,12	14,12	15,12	16,12	17,12	18,12	19,12	20,12	21,12	22,12	23,12	24,12	25,12	26,12	27,12	28,12
0,11	1,11	2,11	3,11	4,11	5,11	6,11	7,11	8,11	9,11	10,11	11,11	12,11	13,11	14,11	15,11	16,11	17,11	18,11	19,11	20,11	21,11	22,11	23,11	24,11	25,11	26,11	27,11	28,11
0,10	1,10	2,10	3,10	4,10	5,10	6,10	7,10	8,10	9,10	10,10	11,10	12,10	13,10	14,10	15,10	16,10	17,10	18,10	19,10	20,10	21,10	22,10	23,10	24,10	25,10	26,10	27,10	28,10
0,9	1,9	2,9	3,9	4,9	5,9	6,9	7,9	8,9	9,9	10,9	11,9	12,9	13,9	14,9	15,9	16,9	17,9	18,9	19,9	20,9	21,9	22,9	23,9	24,9	25,9	26,9	27,9	28,9
0,8	1,8	2,8	3,8	4,8	5,8	6,8	7,8	8,8	9,8	10,8	11,8	12,8	13,8	14,8	15,8	16,8	17,8	18,8	19,8	20,8	21,8	22,8	23,8	24,8	25,8	26,8	27,8	28,8
0,7	1,7	2,7	3,7	4,7	5,7	6,7	7,7	8,7	9,7	10,7	11,7	12,7	13,7	14,7	15,7	16,7	17,7	18,7	19,7	20,7	21,7	22,7	23,7	24,7	25,7	26,7	27,7	28,7
0,6	1,6	2,6	3,6	4,6	5,6	6,6	7,6	8,6	9,6	10,6	11,6	12,6	13,6	14,6	15,6	16,6	17,6	18,6	19,6	20,6	21,6	22,6	23,6	24,6	25,6	26,6	27,6	28,6
0,5	1,5	2,5	3,5	4,5	5,5	6,5	7,5	8,5	9,5	10,5	11,5	12,5	13,5	14,5	15,5	16,5	17,5	18,5	19,5	20,5	21,5	22,5	23,5	24,5	25,5	26,5	27,5	28,5
0,4	1,4	2,4	3,4	4,4	5,4	6,4	7,4	8,4	9,4	10,4	11,4	12,4	13,4	14,4	15,4	16,4	17,4	18,4	19,4	20,4	21,4	22,4	23,4	24,4	25,4	26,4	27,4	28,4
0,3	1,3	2,3	3,3	4,3	5,3	6,3	7,3	8,3	9,3	10,3	11,3	12,3	13,3	14,3	15,3	16,3	17,3	18,3	19,3	20,3	21,3	22,3	23,3	24,3	25,3	26,3	27,3	28,3
0,2	1,2	2,2	3,2	4,2	5,2	6,2	7,2	8,2	9,2	10,2	11,2	12,2	13,2	14,2	15,2	16,2	17,2	18,2	19,2	20,2	21,2	22,2	23,2	24,2	25,2	26,2	27,2	28,2
0,1	1,1	2,1	3,1	4,1	5,1	6,1	7,1	8,1	9,1	10,1	11,1	12,1	13,1	14,1	15,1	16,1	17,1	18,1	19,1	20,1	21,1	22,1	23,1	24,1	25,1	26,1	27,1	28,1
0,0	1,0	2,0	3,0	4,0	5,0	6,0	7,0	8,0	9,0	10,0	11,0	12,0	13,0	14,0	15,0	16,0	17,0	18,0	19,0	20,0	21,0	22,0	23,0	24,0	25,0	26,0	27,0	28,0

But for the sake of our notes, let's develop the game for a much smaller grid that is rendered on a larger scale. It will be easy to change our definitions at the end in

order to recover the original design. If done properly, all of test cases will remain unaffected by the change.

```
(define WIDTH 8) ; in grid units
(define HEIGHT 8) ; in grid units
(define SIZE 32) ; in pixels / grid unit
```

This defines that our game is logically played on a 8x8 grid, which we will render visually as a 256x256 pixel image. The grid for our game now looks like:

0,7	1,7	2,7	3,7	4,7	5,7	6,7	7,7
0,6	1,6	2,6	3,6	4,6	5,6	6,6	7,6
0,5	1,5	2,5	3,5	4,5	5,5	6,5	7,5
0,4	1,4	2,4	3,4	4,4	5,4	6,4	7,4
0,3	1,3	2,3	3,3	4,3	5,3	6,3	7,3
0,2	1,2	2,2	3,2	4,2	5,2	6,2	7,2
0,1	1,1	2,1	3,1	4,1	5,1	6,1	7,1
0,0	1,0	2,0	3,0	4,0	5,0	6,0	7,0

Now we need to consider the interface for coordinates (henceforth, the term “coordinate” refers to *our* representation of coordinates in the Snake Game, not the graphics-coordinates of [big-bang](#)). What do we need to do with coordinates? Here’s a coarse first approximation:

- Compare two coordinates for equality.
- Draw something at a coordinate on to a scene.
- Move a coordinate.
- Determine whether a coordinate is on the board.

This list suggest the following interface for coordinates:

```

;; A Coord implements

;; same-pos? : Coord -> Boolean
;; Is this coordinate at the same position as the given one?

;; draw : Scene -> Scene
;; Draw this coordinate on the scene.

;; move : -> Coord
;; Move this coordinate.

;; on-board? : -> Boolean
;; Is this coordinate on the board?

```

This is a good place to start, but as we start thinking about *what* these methods should do and *how* we might write them, some issues should come to mind. For example, in thinking about the *what*: what should be drawn when the `draw` method is invoked? Perhaps we want the coordinate “to just know” what should be drawn, which suggests that when we implement coordinates they should contain data representing what to draw. Perhaps we want to tell the `draw` method what to draw, which suggests we should revise the contract to include an argument or arguments that represent what to draw. For the time-being, let’s decide that the coordinate will know what to draw.

In the thinking about the *how*: how do you imagine the `draw` coordinate will be written? Assuming we know what to draw, the next question is how will the method know where to draw it? We have a coordinate, which is a grid coordinate, but will need to use `place-image` to actually draw that image on the given scene. But `place-image` works in the pixel-based graphics-coordinate system. We need to be able to convert a coordinate to a pixel-based graphics-coordinate in order to write the `draw` method, but there is nothing in the interface that gives us that capability, *and the interface is all we will have to work with*. This suggests we should revise the interface to include this needed behavior.

Similarly, if we consider how to write the `same-pos?` method, we will want to compare the *x*- and *y*-components of the given coordinate with the *x*- and *y*-components of this coordinate. Again, there is nothing in the interface as given that allows this, so we need to revise.

Now consider the `move` method. How can we write it? What do we expect to happen? There’s not enough information to know—should the coordinate move up? Down? Right three and down seven? Hard to say. While we might expect a coordinate to know how to draw itself, we cannot expect a coordinate to know which way to move itself. This suggests we need to add inputs to the method that represent this needed information. For the purposes of our game, a position needs to be able to move one grid unit in one of four directions. Let’s design the representation of a direction and

a directional input to the `move` method.

Interface design is incredibly important, especially when, unlike in our current situation, it is not easy to revise in the future. Modern computer systems are littered with detritus of past interface design choices because interfaces are difficult and expensive, if not impossible, to change. As an example, the developers of the UNIX operating system, which was developed in the 1970s and is now the basis of both Linux and Mac OS, made the choice to save characters and call the operation that creates a file "CREAT". Forty years later, I'm writing these notes on a portable computer while flying from Boston to Houston. My machine, which weighs far less than any computer that ran UNIX in the 70's, has not one, but two 2.66 GHz processors and 8 gigs of RAM: unimaginable computing resources in the 70s. And yet, when my OS wants to make a new file, it calls the "CREAT" function—not because that missing "E" is a computational extravagance I cannot afford, far from it, but because it is simply too difficult a task to realize a redesign of the interface between my computer and its operating system. The unfortunate thing about interfaces is that when they change, all parties that have agreed to that interface must change as well. Too many people, programs, and devices have agreed to the UNIX interface to make changing "CREAT" to "CREATE" worthwhile.

You won't be able to make the perfect interface on the first try, but the closer you get, the better your life will be in the future.

Revising our interface as described above, we arrive at the following:

```
;; A Coord implements:

;; same-pos? : Coord -> Boolean
;; Is this coordinate at the same position as the given one?

;; draw : Scene -> Scene
;; Draw this coordinate on the scene.

;; move : Dir -> Coord
;; Move this coordinate in the given direction.

;; on-board? : -> Boolean
;; Is this coordinate on the board?

;; {x,y} : -> Nat
;; The {x,y}-component of grid-coordinate.

;; {x-px,y-px} : -> Nat
;; The {x,y}-component of pixel-graphics-coordinate.
```

We haven't designed `Dir` data definition for representing direction; let's take care of

that quickly. In our game, a direction is one of four possibilities, i.e. it is an enumeration. We could use a class-based enumeration, but for the sake of simplicity, let's just use strings and say that:

```
;; A Dir is one of:
;; - "left"
;; - "right"
;; - "up"
;; - "down"
```

This representation has the nice property of being a subset of [big-bang](#)'s KeyEvent representation, so we can rely on the coincidence and handle the "up" key event by moving in the "up" direction without need to convert between representations.

10.4 An implementation of coordinates: segments

At this point, we've flushed out enough of the initial design of the coordinate interface we can now start working on an implementation of it. There are two components that will implement the coordinate interface: segments and food. Let's start with segments.

```
;; A (new seg% Int Int) is a Coord
;; Interp: represents a segment grid-coordinate.
(define-class seg%
  (fields x y)
  ...)
```

Our template for `seg%` methods is:

```
;; ? ... -> ?
(define (seg-template ...)
  (... (send this x) ... (send this y) ...))
```

We've now made a data definition for segments and committed ourselves to implementing the interface. This obligates us to implement all of the methods in `coord%`. We've decided to implement the `coord%` using a class with an `x` and `y` field. This satisfies part of our implementation right off the bat: we get an `x` and `y` method by definition. Let's now do `same-pos?`:

seg%

```
(check-expect (send (new seg% 0 0) same-pos? (new seg% 0 0)) true)
(check-expect (send (new seg% 0 0) same-pos? (new seg% 1 0)) false)
```



```
(define (same-pos? c)
  (and (= (send this x) (send c x))
        (= (send this y) (send c y))))
```

And now draw:

seg%

```
(check-expect (send (new seg% 0 0) draw MT-SCENE)
  (place-image (square SIZE "solid" "red")
    (* 1/2 SIZE)
    (- HEIGHT-PX (* 1/2 SIZE))
    MT-SCENE))

(define (draw scn)
  (place-image (square SIZE "solid" "red")
    (send this x-px)
    (send this y-px)
    scn))
```

And now move:

seg%

```
(check-expect (send (new seg% 0 0) move "up") (new seg% 0 1))
(check-expect (send (new seg% 0 0) move "down") (new seg% 0 -1))
(check-expect (send (new seg% 0 0) move "left") (new seg% -1 0))
(check-expect (send (new seg% 0 0) move "right") (new seg% 1 0))
(define (move d)
  (cond [(string=? d "up")
    (new seg% (send this x) (add1 (send this y)))]
    [(string=? d "down")
    (new seg% (send this x) (sub1 (send this y)))]
    [(string=? d "left")
    (new seg% (sub1 (send this x)) (send this y))]
    [(string=? d "right")
    (new seg% (add1 (send this x)) (send this y))]))
```

And now on-board?:

seg%

```
(check-expect (send (new seg% 0 0) on-board?) true)
(check-expect (send (new seg% 0 -1) on-board?) false)
(check-expect (send (new seg% 0 (sub1 HEIGHT)) on-board?) true)
(check-expect (send (new seg% 0 HEIGHT) on-board?) false)
(define (on-board?)
  (and (<= 0 (send this x) (sub1 WIDTH))
    (<= 0 (send this y) (sub1 HEIGHT))))
```

And finally, the `x-px` and `y-px` methods:

`seg%`

```
(check-expect (send (new seg% 0 0) x-px) (* 1/2 SIZE))
(check-expect (send (new seg% 0 0) y-px) (- HEIGHT-PX (* 1/2 SIZE)))
(define (x-px)
  (* (+ 1/2 (send this x)) SIZE))
(define (y-px)
  (- HEIGHT-PX (* (+ 1/2 (send this y)) SIZE)))
```

That completes all of the obligations of the `seg%` interface.

10.5 Another implementation of coordinates: food

Food is another implementation of the `coord<*>` interface, and it is largely similar to the `seg%` class, which suggests that `seg%` and `food%` may be good candidates for abstraction, but that's something to worry about later. For now, let's implement `food%`. Since we've already been through the design of `seg%`, we'll do `food%` quickly:

```
;; A Food is a (new food% Nat Nat) is a Coord
(define-class food%
  (fields x y)

  (define (same-pos? c)
    (and (= (send this x) (send c x))
         (= (send this y) (send c y))))

  (define (draw scn)
    (place-image (square SIZE "solid" "green")
                  (send this x-px)
                  (send this y-px)
                  scn))

  (define (move d)
    (cond [(string=? d "up")
            (new food% (send this x) (add1 (send this y)))]
          [(string=? d "down")
            (new food% (send this x) (sub1 (send this y)))]
          [(string=? d "left")
            (new food% (sub1 (send this x)) (send this y))]
          [(string=? d "right")
            (new food% (add1 (send this x)) (send this y))]))
```

```

(define (on-board?)
  (and (<= 0 (send this x) (sub1 WIDTH))
        (<= 0 (send this y) (sub1 HEIGHT))))

(define (x-px)
  (* (+ 1/2 (send this x)) SIZE))
(define (y-px)
  (- HEIGHT-PX (* (+ 1/2 (send this y)) SIZE)))

```

You'll notice that this class definition is nearly identical to the definition of `seg%`. The key differences are in `move` and `draw`. We'll hold off on abstracting for now.

10.6 Representing the snake

What information needs to be represented in a snake?

- Direction
- Segments

What are the operations we need to perform on snakes?

```

;; A Snake implements:

;; move : -> Snake
;; Move this snake in its current direction.

;; grow : -> Snake
;; Grow this snake in its current direction.

;; turn : Dir -> Snake
;; Turn this snake in the given direction.

;; draw : Scene -> Scene
;; Draw this snake on the scene.

```

Here's a possible data definition:

```

;; A (new snake% Dir [Listof Seg]) is a Snake
(define-class snake%
  (fields dir segs))

```

But after a moment of reflection, you will notice that a snake with no segments doesn't make sense—a snake should always have at least one segment. Moreover, we need to settle on an interpretation of the order of the list; either the front of the list is interpreted as the front of the snake or the rear of the list is interpreted as the front of the snake. Together, non-emptiness and order let us determine which element is the head of the snake.

Here's our revised data definition:

```
;; A (new snake% Dir (cons Seg [Listof Seg])) is a Snake
(define-class snake%
  (fields dir segs)
  ...)
```

An alternative data definition that might be worth considering is:

```
;; A Snake is a (new snake% Dir Seg [Listof Seg])
(define-class snake%
  (fields dir head segs))
```

But for the time being let's stick with the former one.

Now let's implement the interface. Here's the template:

```
;; ? ... -> ?
(define (snake-template ...)
  (send this dir) ... (send this segs) ...)
```

The `move` method works by moving the head of the snake and dropping the last element of the list of segments:

```
(check-expect (send (new snake% "right" (list (new seg% 0 0))) move)
  (new snake% "right" (list (new seg% 1 0))))
```

snake%

```
(define (move)
  (new snake%
    (send this dir)
    (cons (send (first (send this segs)) move (send this dir))
          (all-but-last (send this segs)))))
```

This relies on a helper function, `all-but-last`, which is straightforward to write (recall that `segs` is a non-empty list):

```
(check-expect (all-but-last (list "x")) empty)
```

```
(check-expect (all-but-last (list "y" "x")) (list "y"))

;; (cons X [Listof X]) -> [Listof X]
;; Drop the last element of the given list.
(define (all-but-last ls)
  (cond [(empty? (rest ls)) empty]
        [else (cons (first ls)
                     (all-but-last (rest ls)))]))
```

The `grow` method is much like `move`, except that no element is dropped from the segments list:

snake%

```
(check-expect (send (new snake% "right" (list (new seg% 0 0))) grow)
              (new snake% "right" (list (new seg% 1 0)
                                         (new seg% 0 0))))

(define (grow)
  (new snake%
    (send this dir)
    (cons (send (first (send this segs)) move (send this dir))
          (send this segs))))
```

Now let's write the `turn` method:

snake%

```
(check-expect (send (new snake% "left" (list (new seg% 0 0))) turn "up")
              (new snake% "up" (list (new seg% 0 0))))

(define (turn d)
  (new snake% d (send this segs)))
```

And finally, `draw`:

snake%

```
(check-expect (send (new snake% "left" (list (new seg% 0 0))) draw MT-SCENE)
              (send (new seg% 0 0) draw MT-SCENE))

(define (draw scn)
  (foldl (λ (s scn) (send s draw scn))
        scn
        (send this segs)))
```

As this method shows, functions and methods can co-exist nicely in a single language.


10.7 Seeing the world

At this point we have a working but incomplete system and we can interact with it in the interactions window:


Examples:

```
> (define w0 (new world%  
  (new snake%  
    "right"  
    (list (new seg% 5 1)  
          (new seg% 5 0)  
          (new seg% 4 0))))  
  (new food% 3 4)))
```


```
> (send w0 to-draw)
```



```
> (send (send w0 on-tick) to-draw)
```



```
> (send (send (send w0 on-tick) on-tick) to-draw)
```



We'll leave it at this point and further the refine the program in the future.

10.8 The whole ball of wax

```
#lang class/0
(require 2htdp/image)
(require class/universe)

(define WIDTH 8) ; in grid units
(define HEIGHT 8) ; in grid units
(define SIZE 32) ; in pixels / grid unit
(define WIDTH-PX (* SIZE WIDTH)) ; in pixels
(define HEIGHT-PX (* SIZE HEIGHT)) ; in pixels
(define MT-SCENE (empty-scene WIDTH-PX HEIGHT-PX))

;; A World is a (new world% Snake Food).
(define-class world%
  (fields snake food)

  (define (on-tick)
    (new world%
      (send (send this snake) move)
      (send this food)))

  (define (tick-rate) 1/8)

  (define (to-draw)
    (send (send this food) draw)
    (send (send this snake) draw MT-SCENE))))

;; A Coord implements:

;; same-pos? : Coord -> Boolean
;; Is this coordinate at the same position as the given one?

;; draw : Scene -> Scene
;; Draw this coordinate on the scene.

;; move : Dir -> Coord
;; Move this coordinate in the given direction.

;; on-board? : -> Boolean
;; Is this coordinate on the board?

;; {x,y} : -> Nat
;; The {x,y}-component of grid-coordinate.
```



```

;; {x-px,y-px} : -> Nat
;; The {x,y}-component of pixel-graphics-coordinate.

;; A Dir is one of:
;; - "left"
;; - "right"
;; - "up"
;; - "down"

;; A (new seg% Int Int) is a Coord
;; Interp: represents a segment grid-coordinate.
(define-class seg%
  (fields x y)
  (check-expect (send origin same-pos? (new seg% 0 0)) true)
  (check-expect (send origin same-pos? (new seg% 1 0)) false)
  (define (same-pos? c)
    (and (= (send this x) (send c x))
          (= (send this y) (send c y))))
  (define (draw scn)
    (place-image (square SIZE "solid" "red")
                  (send this x-px)
                  (send this y-px)
                  scn))
  (define (move d)
    (cond [(string=? d "up")
            (new seg% (send this x) (add1 (send this y)))]
          [(string=? d "down")
            (new seg% (send this x) (sub1 (send this y)))]
          [(string=? d "left")
            (new seg% (sub1 (send this x)) (send this y))]
          [(string=? d "right")
            (new seg% (add1 (send this x)) (send this y))]))
  (define (on-board?)
    (and (<= 0 (send this x) (sub1 WIDTH))
          (<= 0 (send this y) (sub1 HEIGHT))))
  (define (x-px)
    (* (+ 1/2 (send this x)) SIZE))
  (define (y-px)
    (- HEIGHT-PX (* (+ 1/2 (send this y)) SIZE))))

(check-expect (send (new seg% 0 0) draw MT-SCENE)
  (place-image (square SIZE "solid" "red")
                (* 1/2 SIZE)
                (- HEIGHT-PX (* 1/2 SIZE))
                MT-SCENE))
(check-expect (send (new seg% 0 0) move "up") (new seg% 0 1))

```

```

(check-expect (send (new seg% 0 0) move "down") (new seg% 0 -1))
(check-expect (send (new seg% 0 0) move "left") (new seg% -1 0))
(check-expect (send (new seg% 0 0) move "right") (new seg% 1 0))
(check-expect (send (new seg% 0 0) on-board?) true)
(check-expect (send (new seg% 0 -1) on-board?) false)
(check-expect (send (new seg% 0 (sub1 HEIGHT)) on-board?) true)
(check-expect (send (new seg% 0 HEIGHT) on-board?) false)
(check-expect (send (new seg% 0 0) x-px) (* 1/2 SIZE))
(check-expect (send (new seg% 0 0) y-px) (- HEIGHT-PX (* 1/2 SIZE)))

;; A Food is a (new food% Nat Nat) is a Coord
(define-class food%
  (fields x y)

  (define (same-pos? c)
    (and (= (send this x) (send c x))
          (= (send this y) (send c y))))

  (define (draw scn)
    (place-image (square SIZE "solid" "green")
                  (send this x-px)
                  (send this y-px)
                  scn))

  (define (move d)
    (cond [(string=? d "up")
            (new food% (send this x) (add1 (send this y)))]
          [(string=? d "down")
            (new food% (send this x) (sub1 (send this y)))]
          [(string=? d "left")
            (new food% (sub1 (send this x)) (send this y))]
          [(string=? d "right")
            (new food% (add1 (send this x)) (send this y))]))

  (define (on-board?)
    (and (<= 0 (send this x) (sub1 WIDTH))
          (<= 0 (send this y) (sub1 HEIGHT))))

  (define (x-px)
    (* (+ 1/2 (send this x)) SIZE))
  (define (y-px)
    (- HEIGHT-PX (* (+ 1/2 (send this y)) SIZE))))

;; A Snake implements:

;; move : -> Snake

```

```

;; Move this snake in its current direction.

;; grow : -> Snake
;; Grow this snake in its current direction.

;; turn : Dir -> Snake
;; Turn this snake in the given direction.

;; draw : Scene -> Scene
;; Draw this snake on the scene.

;; A (new snake% Dir Seg [Listof Seg]) is a Snake
(define-class snake%
  (fields dir segs)
  (define (move)
    (new snake%
      (send this dir)
      (cons (send (first (send this segs)) move (send this dir))
            (all-but-last (send this segs)))))

  (define (grow)
    (new snake%
      (send this dir)
      (cons (send (first (send this segs)) move (send this dir))
            (send this segs))))

  (define (turn d)
    (new snake% d (send this segs)))

  (define (draw scn)
    (foldl (λ (s scn) (send s draw scn))
          scn
          (send this segs))))

(define origin (new seg% 0 0))
(check-expect (send (new snake% "right" (list (new seg% 0 0))) move)
              (new snake% "right" (list (new seg% 1 0))))
(check-expect (send (new snake% "right" (list (new seg% 0 0))) grow)
              (new snake% "right" (list (new seg% 1 0)
                                         (new seg% 0 0))))
(check-expect (send (new snake% "left" (list (new seg% 0 0))) turn "up")
              (new snake% "up" (list (new seg% 0 0))))
(check-expect (send (new snake% "left" (list (new seg% 0 0))) draw MT-SCENE)
              (send (new seg% 0 0) draw MT-SCENE))

(check-expect (all-but-last (list "x")) empty)

```

```

(check-expect (all-but-last (list "y" "x")) (list "y"))

;; (cons X [Listof X]) -> [Listof X]
;; Drop the last element of the given list.
(define (all-but-last ls)
  (cond [(empty? (rest ls)) empty]
        [else (cons (first ls)
                      (all-but-last (rest ls)))]))

(big-bang (new world%
              (new snake%
                    "right"
                    (list (new seg% 5 1)
                          (new seg% 5 0)
                          (new seg% 4 0)))
              (new food% 3 4)))

```

10.9 Exercises

10.9.1 Different representation of Snakes

Consider the alternative data definition suggested for Snakes:

```

; A Snake is a (new snake% Dir Seg [Listof Seg])
(define-class snake%
  (fields dir head segs))

```

Revise the Snake Game to use this definition and carry out all the changes it implies.

10.9.2 Zombie!

Design and develop an interactive game called *Zombie!*. In this game, there are a number of zombies that are coming to eat your brains. The object is simple: stay alive. You can maneuver by moving the mouse. The player you control always moves toward the mouse position. The zombies, on the other hand, always move toward you. If the zombies ever come in contact with you, they eat your brains, and you die. If two zombies happen to come in to contact with each other, they will mistakenly eat each other's brain, which it turns out is fatal to the zombie species, and so they both die. When a zombie dies, the zombie flesh will permanently remain where it is and should any subsequent zombie touch the dead flesh, they will try to eat it and

therefore die on the spot. Survive longer than all the zombies, and you have won the game.

(This game is based on the *Attack of the Robots!* game described in Land of Lisp. Unlike the Land of Lisp version, this game is graphical and interactive rather than text-based. Hence, our game doesn't suck.)

Once you have a working version of the game, add the following feature: whenever the user does a mouse-click, the player should be instantly teleported to a *random* location on the screen.

10.9.3 Primum non copy-and-paste

A natural design for the Zombie game is to have a `Zombie` and `Player` class of data. But you may find your first iteration of the Zombie game duplicates a lot of code between these classes. In fact, the `Zombie` and `Player` classes have more in common than apart. It may even be tempting to pursue an unnatural design in which there is only a single class of data, which must consist of an additional *bit*, which is interpreted as signifying “zombieness” versus “playerness.” Down that path waits shame, defeat, and a brittle design that makes babies cry.

To recoil at the prospect of copy-and-paste is commendable, but we shouldn't throw the crying babies out with the bathwater. Let's step back and ask ourselves if this dilemma is really inescapable.

First, let's consider the information that needs to be represented in a game. When you look at the game, you see several things: live zombies, dead zombies, a player, and a mouse. That might lead you to a representation of each of these things as a separate class, in which case you may later find many of the methods in these classes are largely identical. You would like to abstract to avoid the code duplication, but thus far, we haven't seen any class-based abstraction mechanisms. So there are at least two solutions to this problem:

1. Re-consider your data definitions.

Program design is an iterative process; you are continually revising your data definitions, which induces program changes, which may cause you to redesign your data definitions, and so on. So when you find yourself wanting to copy and paste lots of code, you might want to reconsider how you're representing information in your program. In the case of zombie, you might step back and see that although the game consists of a player, dead zombies, live zombies, and a mouse, these things have much in common. What *changes over time* about each of them is their position. Otherwise, what makes them different is how they are rendered visually. But it's important to note that way any of these things are rendered *does not change over the course of the game*—a dead zombie

is *always* drawn as a gray dot; a live zombie is *always* drawn as a green dot; etc. Taking this view, we can represent the position of each entity uniformly using a single class. This avoids duplicating method definitions since there is only a single class to represent each of these entities.

2. Abstract using the functional abstraction recipe of last semester.

Just because we are in a new semester and studying a new paradigm of programming, we should not throw out the lessons and techniques we've previously learned. In particular, since we are writing programs in a multi-paradigm language—one that accommodates both structural and functional programming *and* object-oriented programming—we can mix and match as our designs necessitate. In this case, we can apply the recipe for *functional abstraction* to the design of identical or similar methods, i.e. two methods with similar implementations can be abstracted to a single point of control by writing a helper function that encapsulates the common code. The method can then call the helper function, supplying as arguments values that encapsulate the differences of the original methods.

Revise your Zombie! program.

10.9.4 Space Invaders!

For this exercise, you will design and develop (a pared down version of) the classic game of *Space Invaders*. In this game, there are a number of space aliens that are descending from the top of the screen. They move left to right and then down at uniform speed. The player controls a laser canon that can be moved left or right along the bottom of the screen. The player can fire the laser, which shoots straight up. If the laser hits an alien, the alien dies. If any alien makes it to the bottom of the screen (or hits the cannon), the player loses. If the player destroys all the invaders, the player wins.

To get a sense of the game, you can play this online version of the game. Your version doesn't need to have all of the features of the online game; at a minimum, your game should:

- Allow players to shoot (using the spacebar)
- Allow players to move (using left and right arrow keys)
- Allow aliens to shoot
- Make aliens drop down and reverse direction when they reach the edge of the screen
- End the game when all aliens are destroyed or some alien reaches the ground.

If you want to embellish your game, you can add additional features:

- Keep score
- Add the flying saucer for bonus points
- Have several rows of aliens
- Have several kinds of aliens
- Give the player multiple lives
- Support bunkers to shield the player

But remember, you're graded for your program design, not making a cool video game. So whatever you add, make sure it's well designed.

Chapter 11

Universe

In this chapter, we're going to start looking at the design of multiple, concurrently running programs that communicate with each other. We will use the `universe` system as our library for communicating programs.

11.1 A look at the Universe API

The basic universe concept is that there is a "universe" program that is the administrator of a set of world programs. The universe and the world programs can communicate with each other by sending messages, which are represented as S-Expressions.

So far we have focused on the design of single programs; we are now going to start looking at the design of communicating systems of programs.

In addition to these notes, be sure to read the documentation on section ???.

%%
TODO: FIX THIS?

11.2 Messages

A message is represented as an S-Expression. Here is their data definition:

An S-expression is roughly a nested list of basic data; to be precise an S-expression is one of:

- a string,

- a symbol,
- a number,
- a boolean,
- a char, or
- a list of S-expressions.

The way that a world program sends a message to the universe is by constructing a package:

```
;; A Package is a (make-package World SExp).
```

The world component is the new world just like the event handler's produced for single world programs. The s-expression component is a message that is sent to the universe.

11.3 Simple world

As a simple example, let's look at a world program that counts up and sends messages to a universe server as it counts. In this simple example there is only one world that communicates with the server, and the server does nothing but receive the count message (it sends no messages back to the world).

Let's start with the counting world program, which does not communicate with any server, it just counts:

```
#lang class/0
(require class/universe)
(require 2htdp/image)

;; Scene is 300x100 pixels
(define WIDTH 300)
(define HEIGHT 100)

;; A CounterWorld is a (new cw% Natural)
;; and implements
;; - tick-rate : -> Number
;;   Tick rate for counting.
;; - on-tick : -> CounterWorld
;;   Increment counter world state.
;; - to-draw : -> Scene
```

```
;; View counter world state as a scene.
(define-class cw%
  (fields n)
  (define (tick-rate) 1)
  (define (on-tick)
    (new cw% (add1 (send this n))))
  (define (to-draw)
    (overlay (text (number->string (send this n)) 40 "red")
      (empty-scene WIDTH HEIGHT))))

;; Run, program, run!
(big-bang (new cw% 0))
```

When you run this program, you see the world counting up from zero.

11.4 Simple world, broadcasting to server

Now let's modify our program so that it does some simple communication with a server. As an initial design, we'll make a program that simply notifies a server as it counts. In other words, our program will only engage in one-way communication by broadcasting data to the server.

So what exactly do we want to broadcast? If we want to communicate the current state of the world, we may be tempted to try to communicate the current *CounterWorld* value; but remember that the current world state is an *object*, which isn't included in the data definition of messages. Moreover, the current state includes not only data, but functionality: functionality for handling tick events, viewing the current count as an image, etc.—all of which are things the server doesn't really need. To communicate the essence of what state the *CounterWorld* is in, all we really need to send is the current count: a number, which fortunately does fall under the set of message values.

We can now revise the *on-tick* method to not only produce the new state of the world, but additionally a message to be sent to the server:

CW%

```
(define (on-tick)
  (make-package (new cw% (add1 (send this n)))
    (add1 (send this n))))
```

Thinking about the message protocol for this simple scenario, the client and server communications will look like this:

Client			Server
-----			-----
Event	State	Message	
Bang	(new cw% 0)		=====>
Tick	(new cw% 1)	1	----->
Tick	(new cw% 2)	2	----->
Tick	(new cw% 3)	3	----->
...

Here, the ==> arrow indicates the world registering with the server. This is the one time communication that establishes a dialogue (really, in this case, a monologue) between the client and server. After registering with the server, the client will send its current count, indicated with -> arrows, as it ticks along.

In this table we show the state of the client and the events that occur. With each event, which potentially changes the state, we show what message is sent to the server. At this point the server is opaque—we don't know or really care what state the server is in and we assume that the server doesn't send any message back to the client.

Now to register this program with a universe server, we need to implement a `register` method that produces a string that is the IP address of the server. (Since we're going to run the universe and world on the same computer, we will use `LOCALHOST` which is bound to the address of our computer.)

CW%

```
;; register : -> String
;; IP address of server
(define (register) LOCALHOST)
```

Now when you run this program you will see the world program try to connect to the universe, but since we have not written—much less run—the server, it cannot find the universe. After a few tries, it gives up and continues running without communicating with the universe.

Our complete client is:

```
#lang class/0
(require class/universe)
(require 2htdp/image)

;; Scene is 300x100 pixels
(define WIDTH 300)
(define HEIGHT 100)

;; A CounterWorld is a (new cw% Natural)
;; and implements
```

```

;; - register : -> String
;;   IP address of server
;; - tick-rate : -> Number
;;   Tick rate for counting.
;; - on-tick : -> (make-package CounterWorld Number)
;;   Increment counter world state, broadcast to server.
;; - to-draw : -> Scene
;;   View counter world state as a scene.
(define-class cw%
  (fields n)
  (define (register) LOCALHOST)
  (define (tick-rate) 1)
  (define (on-tick)
    (make-package (new cw% (add1 (send this n)))
      (add1 (send this n))))
  (define (to-draw)
    (overlay (text (number->string (send this n)) 40 "red")
      (empty-scene WIDTH HEIGHT))))

;; Run, program, run!
(big-bang (new cw% 0))

```

11.5 Simple universe, receiving broadcasts

Now let's write a simple server that receives the message from the counter world client. We could write the server in the same file as the client, but since these are really two separate programs that talk to each other, let's emphasize that by writing the server in its own tab.

A universe program is similar to a world program: it's a program that responds to events. The difference is in the kind of events that can occur. The most important events are already shown in our protocol diagram:

- a new world starts communicating with the server,
- a world sends a message.

When these events occur, the server reacts by calling the appropriate method, in this case `on-new` and `on-msg`.

We'll be working with the OO-style universe, but you should read the documentation for `2http/universe` and translate over the concepts to our setting as you've done for `big-bang`.

As it turns out, if you leave these methods off, the universe library will do something sensible, namely nothing. So for our simple counter program, the following works:

```
#lang class/0
(require class/universe)
(define-class cu%)

;; Run, server, run!
(universe (new cu%))
```

This is equivalent to doing the following:

```
#lang class/0
(require class/universe)

(define-class cu%
  ;; IWorld -> Bundle
  (define (on-new iw) this)

  ;; IWorld S-Expr -> Bundle
  (define (on-msg iw m) this))
```

When a world registers, the `on-new` method is called with an `IWorld` value. An `IWorld` value opaquely represents a world, that is you do not have the ability to examine the contents of the value, but you can compare it for equality with other `IWorld` values using `iworlde=?`.

When a world sends message, the `on-msg` method is called with the `IWorld` representing the world that sent the message and the S-Exp message that was sent.

In both cases, the method must either produce a universe or a bundle:

```
;; A Bundle is a (make-bundle Universe [Listof Mail] [Listof IWorld]).
;; A Mail is a (make-mail IWorld S-Exp).
```

A bundle signals communication to some set of worlds. Within a bundle, the universe component is the new state of the universe; the list of mail is a list of messages that will be sent back to the worlds (more on this in a moment), and the list of worlds are worlds that the server has chosen to disconnect from.

For the purposes of our example, the universe maintains no state (the class has no data). When a new world registers, we do nothing, and when a world sends a message, we also do nothing, send nothing in response, and disconnect no worlds.

Running this program launches the universe server, making it ready to receive registrations from worlds. After starting the universe server, if we switch back to the

world program tab and run it, we'll see that it successfully registers with the universe and the universe console reports that the world signed up with it.

If we examine the server side of the diagram considered above, we see:

Client		Server		
		Message	Event	State
			Bang	(new cu%)
	=====>		Join	(new cu%)
	----->	1	OnMsg	(new cu%)
	----->	2	OnMsg	(new cu%)
	----->	3	OnMsg	(new cu%)
	

From the perspective of the server, the client is opaque—all we can observe is the messages sent from the client.

The server, as written, will actually work if more than one client connect to the server. To try it out, just run two clients at the same time. When multiple clients connect, we may see interaction like this:

A	B		Server		
			Message	Event	State
				Bang	(new cu%)
		=====>		Join	(new cu%)
		----->	1	OnMsg	(new cu%)
		----->	2	OnMsg	(new cu%)
		=====>		Join	(new cu%)
		----->	1	OnMsg	(new cu%)
		----->	2	OnMsg	(new cu%)
		----->	3	OnMsg	(new cu%)
		----->	3	OnMsg	(new cu%)
		

If we would like to rule out this kind of interaction and have the server listen only to one client, that's easy to do. The idea is that we can develop a server that accepts no new clients. Such a server would have the following `on-new` method:

```
(define (on-new iw)
  (make-bundle this empty (list iw)))
```

Now we can have the server start in the accepting universe state that will allow a client to join, but as soon as one does, it transitions to the unaccepting state that rejects all new clients:

```

#lang class/0
(require class/universe)

;; A Server is one of:
;; - (new accept%)
;; - (new reject%)

(define-class accept%
  (define (on-new iw)
    (new reject%)))

(define-class reject%
  (define (on-new iw)
    (make-bundle this empty (list iw))))

;; Run, server, run!
(universe (new accept%))

```

The interaction described above would now look like:

A	B	Server		
		Message	Event	State
			Bang	(new accept%)
	=====>		Join	(new reject%)
	-----> 1		OnMsg	(new reject%)
	-----> 2		OnMsg	(new reject%)
	=====>		Join	(new reject%)
	<=====		Disconn	(new reject%)
	-----> 3		OnMsg	(new reject%)
	

We can imagine more sophisticated scenerios such as one where we want to enable peer-to-peer communication. As a simple example, let's build a system where one client can broadcast messages to another. This server will have to involve some data since it needs to remember who to broadcast mail to when it receives a message. The initial state of the server is waiting for the broadcaster to join, after which it waits for the broadcastee. Once both parties have joined, every time the broadcaster sends a message, the server relays it to the broadcastee:

```

#lang class/0
(require class/universe)

;; A Server is one of:
;; - (new wait-caster%)
;; - (new wait-castee%)

```



```
;; - (new relay% IWorld)

(define-class wait-caster%
  (define (on-new iw)
    (new wait-castee%)))

(define-class wait-castee%
  (define (on-new iw)
    (new relay% iw)))

(define-class relay%
  (fields castee)
  (define (on-new iw)
    (make-bundle this empty (list iw)))
  (define (on-msg iw msg)
    (make-bundle this (make-mail (send this castee) msg) empty)))

;; Run, server, run!
(universe (new wait-caster%))
```

Notice how the `relay%` class contains an `IWorld` that it broadcasts to when it receives a message. It also rejects any new clients that try to connect.

An example interaction for this serve is:

A	B	Server		
		Message	Event	State
			Bang	(new wait-caster%)
	=====>		Join	(new wait-castee%)
	-----> 1		OnMsg	(new wait-castee%)
	-----> 2		OnMsg	(new wait-castee%)
	=====>		Join	(new relay% A)
	-----> 3		OnMsg	(new relay% A)
<-----	3			
	-----> 4		OnMsg	(new relay% A)
<-----	4			
	

Notice how in this example, B, the broadcaster, starts sending messages before another client has joined. Those messages are simply ignored. After a second client joins and the server goes into the relay state, subsequent messages will be sent to A, the broadcastee.

11.6 Simple world, receiving messages from the server

This should be re-written to use the state pattern.

In the example considered so far, the server sends messages to a client, but that requires the client is willing to receive such messages. Let's look at how to adapt our simple broadcasting client to one that listens for message from the universe server.

This client is in one of two states: it either hasn't received a message yet, so it doesn't know what the current "count" is yet, or it has received at least one message and so it knows the most recent count, relayed from the counting to world to the server to this client. The listener client doesn't need to generate any of its own events so we drop the `on-tick` handler and instead add a `on-receive` method that handles incoming messages from the server.

```
#lang class/0
(require class/universe)
(require 2htdp/image)

;; Scene is 300x100 pixels
(define WIDTH 300)
(define HEIGHT 100)

;; A ListenerWorld is one of:
;; - (new wait-first%)      Interp: waiting for first msg.
;; - (new wait-next% Number) Interp: waiting for next msg.
;; and implements
;; - to-draw : -> Scene
;;   View counter world state as a scene.
;; - on-receive : Number -> ListenerWorld
;;   Receive new counter state from the universe.

(define-class wait-first%
  (define (register) LOCALHOST)
  (define (on-receive msg) (new wait-next% msg))
  (define (to-draw)
    (overlay (text "Waiting" 40 "red")
      (empty-scene WIDTH HEIGHT))))

(define-class wait-next%
  (fields n)
  (define (on-receive msg) (new wait-next% msg))
  (define (to-draw)
    (overlay (text (number->string (send this n)) 40 "red")
      (empty-scene WIDTH HEIGHT))))

;; Run, program, run!
(big-bang (new wait-first%))
```

We can now run the clients and server and observe the communication, but to do so, you have to be careful about the order in which you start things. Our server *assumes* the first client to connect is the broadcaster. This is really a flaw in our design, and we can look at eliminating it, but as a work around just be sure to first start the server, then the counter world, and then the listener world.

11.7 Rules of engagement: protocols and enforcement

There are a couple subtle points worth noting about the server and the possible interactions it allows.

First, if the broadcaster and broadcastee don't join in that order, nothing works as intended. While it may be reasonable to think we can control the order for simple experiments, when our programs are released into the wild and we have *no control* over when and where the clients run, this isn't a reasonable assumption to make. We should refine our program to allow the clients to join in any order.

If the clients can join in any order, a client must identify their role by sending a message after joining. But now the protocol, even for this simple scenario, becomes fairly complicated. You can imagine a client joins but waits a long time to identify their role. Another client may join in the interim and immediately identify their role. What if two clients want to broadcast? What if two clients want to listen?

Second, if the broadcastee ever decides to send a message, rather than just passively listening to what's sent its way, we will treat this message as though it came from the broadcaster (and thus send it back to the broadcastee). That is, the following scenario is possible:

A	B	Server		
-----	-----	-----	-----	-----
		Message	Event	State
			Bang	(new wait-caster%)
	=====>		Join	(new wait-castee%)
=====>			Join	(new relay% A)
	----->	1	Msg	(new relay% A)
<-----		1		
	----->	2	Msg	(new relay% A)
<-----		2		
----->		7	Msg	(new relay% A)
<-----		7		
	

This brings up the interesting issue of *protocols*, that is, what are the proper rules of conduct for the clients' and server's interaction. First and foremost, it's important

to establish and document the rules of interaction. After doing so, we may want to *enforce* protocols, either on the client or server side, or both.

Let's try to develop broadcast and listener clients and a server that address these two issues. In this design, the clients' changes are relatively straightforward. They start by sending an initial message identifying their role, either `'broadcast` or `'listen`.

The server is more involved. Here we model *connections* with worlds as either being an unknown connection, a broadcast connection, or a broadcastee connection. The server starts in a state with no connections. When a world joins, it transitions to a single, unknown connection state. When a second world joins, it transitions to a two-connection state. Any time after a connection has been made the server accepts role messages and sends them to the connections. If the connection is in an unknown state and the role came from that connections' underlying world, the connection is assigned the requested role. The only issue remaining with this server is handling the case of two clients want to both broadcast or both listen, which we just ignore for now.

```
#lang class/0
;; A Server is
;; - (new wait%)
;; - (new one% Conn)
;; - (new two% Conn Conn)

(define-class wait%
  (define (on-new iw)
    (new one% (new unknown% iw))))

(define-class one%
  (fields conn)
  (define (on-new iw)
    (new two% (send this conn) (new unknown% iw)))
  (define (on-msg iw msg)
    (cond [(role? msg)
            (new one% (send (send this conn) ident msg))]
          [else this])))

(define-class two%
  (fields conn1 conn2)
  (define (on-msg iw msg)
    (cond [(role? msg)
            (new two%
              (send (send this conn1) ident msg)
              (send (send this conn2) ident msg))]
          [else
            (make-bundle this
```

```

        (append (send (send this conn1) mail msg)
                 (send (send this conn2) mail msg))
        empty)))))

;; A Conn is one of:
;; - (new unknown% IWorld)
;; - (new caster% IWorld)
;; - (new castee% IWorld)
;; implements
;; - mail : Msg -> [Maybe Mail]
;;   Maybe send mail to this connection.
;; - ident : IWorld Role -> Conn
;;   Assign given role for this connection.

;; A [Maybe X] is one of:
;; - empty
;; - (list X)

;; An Role is one of 'broadcast or 'listen
(define (role? x) (or (eq? 'broadcast) (eq? 'listen)))

(define-class unknown%
  (fields iw)
  (define (mail msg) empty)
  (define (ident iw role)
    (cond [(iworld=? (send this iw) iw)
           (cond [(eq? role 'broadcast)
                  (new caster% (send this iw))]
                 [else
                  (new castee% (send this iw))])]
          [else this])))

(define-class caster%
  (fields iw)
  (define (mail msg) empty)
  (define (ident iw role) this))

(define-class castee%
  (fields iw)
  (define (mail msg)
    (list (make-mail (send this iw) msg)))
  (define (ident iw role) this))

```

Now that we've seen the basics of communicating programs, let's build something a little more substantial.

11.8 Exercises

11.8.1 Tron

For this assignment, you will design and develop the Tron Lightcycle game. The basic idea of this game, if you haven't seen the movie, is that two players move around the screen, leaving a trail of where each of the players has been. If a player runs into the trail that they or the other player has left behind, or into a wall, they lose. If the two players simultaneously hit each other, or both hit a wall or trail at the same time, the game is a tie.

Here's a flash game where you can play the game yourself online.

1. Distributed Tron

The first version of Tron you will develop is a distributed one, using the `class/universe` library. You will need to design two separate parts of the program:

The server: this will accept connections from two clients, communicate with the clients via messages indicating the directions the clients want to move, and then send back updated information about the positions of both players and the trails on the board.

The client: you should only need to *implement* one client, but you will *run* two of them, one for each player. The client will draw the world to the screen, receive messages from the server and update the world state in response, allow the user to input their desired direction (probably via the arrow keys) and communicate this direction to the server.

Once you've implemented both the server and client, you'll be able to play against your friends and classmates over the network.

2. Computer Tron

In this part of the assignment, you'll implement a new kind of client—a computer player. This player will, like the regular client, display the world as well as send and receive messages to and from the server. However, it won't take input from the user; instead it will make decisions itself based on the state of the board.

There is no requirement for any particular behavior for your computer player—you can have it behave randomly, behave dumbly, or be the world's best tron player. We won't grade your assignment based on its playing choices, but we encourage you to go wild with your choices of how the computer player behaves.

Chapter 12

Guess my number

In this chapter, we'll take an in-depth look at small, but interesting distributed game: the "Guess my Number" game.

12.1 One Player Guess my Number

Let's start by considering a slimmed-down version of guess my number in which there is just one player, the client, who tries to guess the number the server is thinking of.

12.1.1 The GmN server

In this simplified version of the game, there is not much the server needs to do:

- it should remember what number it is thinking of,
- and it should respond to guesses made by the player.

From the server's point of view, the interactions look like the following, supposing the server is thinking of 5:

```
Client          Server
-----
                -----
                Message
                =====>
```

```

----->
<----- "too small"
-----> 9
<----- "too big"
-----> 6
<----- "too big"
-----> 5
<----- "just right"

```

In order to respond with “too big”, “too small”, or “just right”, the state of the server will need to include the number that the server has in mind. Thus a natural representation of the state of the server is an object with a single field that contains the number, and an `on-msg` method that will respond to a guess made by the player:

```

(define-class thinking-of%
  (fields n)
  ;; on-msg : IWorld SExp -> Universe
  ;; Mail response to guess from given world
  (define (on-msg iw msg) ...))

```

In support of `on-msg`, let’s design a method that consumes a guess (a real number) and produces either the string “too small”, “too big”, or “just right” depending on whether the guess is smaller, bigger, or equal to the number the server contains.

thinking-of%

```

;; A Response is one of:
;; - "too big"
;; - "too small"
;; - "just right"

;; guess : Real -> Response
;; Respond to a given guess
(check-expect ((new thinking-of% 7) . guess 5) "too small")
(check-expect ((new thinking-of% 7) . guess 9) "too big")
(check-expect ((new thinking-of% 7) . guess 7) "just right")
(define (guess m) ...)

```

The final step of writing the code is trivial at this point, so we can move on to the `on-msg` method:

thinking-of%

```

;; on-msg : IWorld SExp -> Universe
;; Mail response to guess from given world
(check-expect ((new thinking-of% 7) . on-msg iworld1 "Bogus")
  (new thinking-of% 7))

```



```

(check-expect ((new thinking-of% 7) . on-msg iworld1 5)
  (make-bundle (new thinking-of% 7)
    (list (make-mail iworld1 "too small"))
    empty))
(define (on-msg iw msg) ...)

```

Again the code is trivial once the initial design work is complete.

12.1.2 The GmN Client

The client program will register with the server and allow the user to propose guesses which are sent to the server. The response of “too small”, “too larg”, or “just right” is shown to the user and they can propose more guesses if desired. For the moment, let’s just focus on guessing a single digit to make things simple. We’ll look at multi-digit guesses later. In this simplified setting the world can be in one of two states: the client is accepting guesses, or it has a guess and it is waiting for the server to respond to that guess.

So we arrive at the interface definitions:

```

;; A Client is one of:
;; - Waiting
;; - Accepting
;;
;; A Waiting implements:
;; - to-draw : -> Scene
;; - on-receive : SExp -> Client
;;
;; An Accepting implements:
;; - to-draw : -> Scene
;; - on-key : SExp -> Client

```

So from the client’s perspective, interactions with the server will look like the following:

Client		Server
Event	State	
Bang	Accepting	=====>
Key "3"	Waiting	-----> 3
Msg	Accepting	<----- "too small"
Key "9"	Waiting	-----> 9
Msg	Accepting	<----- "too big"
Key "6"	Waiting	-----> 6

```

Msg      Accepting <----- "too big"
Key "5"   Waiting  -----> 5
Msg      Accepting <----- "just right"

```

On further reflection, you should discover that there are in fact two different kinds of Accepting states the client could be in: one in which no guess has been made—so client is waiting to accept what will be the initial guess, and another in which a guess has been and a response has been received from the server about that guess. In this case, we want the client to display the guess and the server's response while waiting for the next guess. Based on this analysis, it's clear we will need two implementations of Accepting with different behaviour and data:

```

;; A (new no-guess%) implements Accepting
(define-class no-guess%
  (define (register) ...)
  (define (to-draw) ...)
  (define (on-key ke) ...))

;; A (new waiting% Number) implements Waiting
(define-class waiting%
  (fields n)
  (define (to-draw) ...)
  (define (on-receive msg) ...))

;; A (new inform% Number String) implements Accepting
(define-class inform%
  (fields n msg)
  (define (to-draw) ...)
  (define (on-key msg) ...))

```

Here is the interactions diagram, revised slightly to be more precise about the state of the client:

Client		Server
Event	State	
Bang	(new no-guess%)	=====>
Key "3"	(new waiting% 3)	-----> 3
Msg	(new inform% 3 ...)	<----- "too small"
Key "9"	(new waiting% 9)	-----> 9
Msg	(new inform% 9 ...)	<----- "too big"
...		

First, let's fix the dimensions of the background image and make a function for displaying strings:

```

(define MT-SCENE (empty-scene 400 400))

```

```
;; String -> Image
(define (txt str)
  (text str 40 'red))
```

The `no-guess%` class represents the initial state of the client and should display a message to the user to make a guess. When a key is pressed in this state, if it's numeric, that number becomes the new guess. Otherwise the key is ignored. Some examples:

no-guess%

```
(check-expect ((new no-guess%) . to-draw)
  (overlay (txt "Take a guess") MT-SCENE))
(check-expect ((new no-guess%) . on-key "h")
  (new no-guess%))
(check-expect ((new no-guess%) . on-key "7")
  (make-package (new waiting% 7) 7))
```

The remaining work of writing the code is easy:

no-guess%

```
(define (to-draw)
  (overlay (txt "Take a guess") MT-SCENE))

(define (on-key ke)
  (local [(define n (string->number ke))])
  (cond [(number? n)
        (make-package (new waiting% n) n)]
        [else this])))
```

The `string->number` function is being used to test for numeric key events—it works by producing `false` when given a string that cannot be converted to a number, otherwise it converts the string to a number.

The `waiting%` class represents the client waiting for a response from the server. To render this state, let's display the number that has been guessed. Since this class of objects doesn't have a `on-key` event, we are implicitly disallowing further guesses while waiting. If the server responds with a string message, the client transitions to a new accepting state. Some examples:

waiting%

```
(check-expect ((new waiting% 5) . to-draw)
  (overlay (txt "Guessed: 5") MT-SCENE))
(check-expect ((new waiting% 5) . on-receive "too small")
  (new inform% 5 "too small"))
```

```
(check-expect ((new waiting% 5) . on-receive 'something)
              (new waiting% 5))
```

All that's left is to write some code:

waiting%

```
(define (to-draw)
  (overlay (beside (txt "Guessed: ")
                  (txt (number->string (this . n))))
           MT-SCENE))

(define (on-receive msg)
  (cond [(string? msg)
        (new inform% (this . n) msg)]
        [else this]))
```

Finally, the `inform%` class represents clients that have guessed, received a response, and are now waiting for subsequent guesses.

By virtue of not having an `on-receive` method, a client in the accepting state will ignore message from the server (which should be considered an error on the server's part). Just like `no-guess%`, it should accept numeric key presses as a new guess and transition to the waiting state. To render the state, we should display the guess and the feedback from the server. For example:

```
(check-expect ((new inform% 7 "too small") . to-draw)
              (overlay (txt "Guessed: 7; too small") MT-SCENE))
(check-expect ((new inform% 7 "too small") . on-key "a")
              (new inform% 7 "too small"))
(check-expect ((new inform% 7 "too small") . on-key "9")
              (make-package (new waiting% 9) 9))
```

The code is just as easy as in the other classes:

inform%

```
(define (to-draw)
  (overlay (beside (txt "Guessed: ")
                  (txt (number->string (this . n)))
                  (txt "; ")
                  (txt (this . msg)))
           MT-SCENE))

(define (on-key ke)
  (local [(define n (string->number ke))])
  (cond [(number? n)
```

```
(make-package (new waiting% n) n)]
[else this]]))
```

Notice that the `on-key` method of `inform%` and `no-guess%` are *identical*. We'll discuss how to abstract such identical code in chapter 10. %% TODO: FIX THIS SO IT'S THE RIGHT NUMBER

Now we can play the game with:

```
(launch-many-worlds (big-bang (new no-guess%))
  (universe (new thinking-of% (random 10))))
```

12.1.3 Many Players, One Number

Although we've developed this program under the simplifying assumption that there's only one client, the server works just as well when there are multiple clients. Under this scenario, all of the clients are trying to guess the one number the server is thinking of in parallel. For example, try this out:

```
(launch-many-worlds (big-bang (new no-guess%))
  (big-bang (new no-guess%))
  (big-bang (new no-guess%))
  (universe (new thinking-of% (random 10))))
```

It would take more work and a redesign of the server if we wanted to have the server think of a number for each of the clients independently. We'll examine such a redesign later in the chapter, but first, let's look at how to implement a better client.

12.1.4 Guessing Big

It's not so fun to play guess my number when the numbers can only be between zero and nine. But note that this limitation exists only in the client. The server is perfectly capable of serving up any real number, but the client as currently designed will have a difficult time against `(new thinking-of% 11)`. The good news is that the hard part—dealing with the protocol of messages—is behind us. It's a small matter of iterative refinement to make the client capable of playing larger numbers.

Looking back at our initial design, it should be clear that some of the pieces we developed can still be used. In particular, the `waiting%` class is perfectly sufficient for dealing with numbers larger than 9. The problem is we have no good way of getting to that point from `no-guess%`. So let's reconsider the states of the client. It seems that if we want to accept multi-digit input, we need to have a new class of Accepting clients that has received some digits but is ready to accept more. We have

to settle on some input to signify the end of digits, at which point a complete number has been given and can be shipped off to the server as a guess.

```
;; A (new continue% NumberString) implements Accepting
(define-class continue%
  (fields digits)
  (define (to-draw) ...)
  (define (on-key ke) ...))
```

The `digits` field will hold a string containing all of the digits entered so far (it will always be non-empty and can be converted to a number with `string->number`). Let's say that when the user presses the "Enter" key, the input is complete. If the user presses any key other than "Enter" or a digit, let's ignore it. To render a continue state, let's display "Guessing:" and the digits entered so far followed by an underscore to indicate that the client is waiting for more input.

We can now formulate some examples:

continue%

```
(check-expect ((new continue% "123") . to-draw)
  (overlay "Guessing: 123_" MT-SCENE))
(check-expect ((new continue% "123") . on-key "4")
  (new continue% "1234"))
(check-expect ((new continue% "123") . on-key "a")
  (new continue% "123"))
(check-expect ((new continue% "123") . on-key "\r")
  (make-package (new waiting% 123) 123))
```

Now for the code:

continue%

```
(define (to-draw)
  (overlay (beside (txt "Guessing: ")
    (txt (this . digits))
    (txt "_"))
    MT-SCENE))

(define (on-key ke)
  (cond [(number? (string->number ke))
    (new continue% (string-append (this . digits) ke))]
    [(key=? "\r" ke)
    (local [(define n (string->number (this . digits)))]
      (make-package (new waiting% n) n))]
    [else this]))
```

We now need to go back and revise `no-guess%` and `inform%` to transition to `continue%` whenever a digit key is pressed:

```
(check-expect ((new no-guess%) . on-key "7")
              (new continue% "7"))
(check-expect ((new inform% 5 "too small") . on-key "7")
              (new continue% "7"))
```

The code in both cases is:

`no-guess% and inform%`

```
(define (on-key ke)
  (cond [(number? (string->number ke))
        (new continue% ke)]
        [else this]))
```

Now try this out:

```
(launch-many-worlds
  (big-bang (new no-guess%))
  (universe (new thinking-of% (random 1000))))
```

12.2 Two player guess my number

[FIXME this section is out of sync with previous sections and needs to be re-written.]

Now let's write a 2-player version of the game where one player thinks of a number and the other player guesses.

Here is the server:

```
#lang class/0
(require class/universe)

;; A Universe is a (new universe% [U #f Number] [U #f IWorld] [U #f IWorld]).
(define-class universe%
  (fields number
          picker
          guesser)

  ;; is the given world the picker?
  (define (picker? iw)
    (and (iworld? (send this picker))
```

```

        (iworld=? iw (send this picker))))

;; is the given world the guesser?
(define (guesser? iw)
  (and (iworld? (send this guesser))
        (iworld=? iw (send this guesser))))

(define (on-new iw)
  (cond [(false? (send this picker))
        (make-bundle
         (new universe% false iw false)
         (list (make-mail iw "pick a number"))
         empty)]
        [(false? (send this guesser))
        (make-bundle
         (new universe% (send this number) (send this picker) iw)
         empty
         empty)]
        [else
        (make-bundle this empty (list iw))]))

(define (on-msg iw m)
  (cond [(and (picker? iw)
              (false? (send this number)))
        (make-bundle
         (new universe% m (send this picker) (send this guesser))
         empty
         empty)]
        [(picker? iw) ;; already picked a number
        (make-bundle this empty empty)]
        [(and (guesser? iw)
              (number? (send this number)))
        (make-bundle this
                      (list (make-mail iw (respond m (send this number))))
                      empty)]
        [(guesser? iw)
        (make-bundle this
                      (list (make-mail iw "no number"))
                      empty))]))

;; Number Number -> String
(define (respond guess number)
  (cond [(< guess number) "too small"]
        [(> guess number) "too big"]
        [else "just right"])))

```



```
(universe (new universe% false false false))
```

The client stays the same! You can launch the two players with:

```
(launch-many-worlds  
  (big-bang (new guess-world% "guess a number"))  
  (big-bang (new guess-world% "guess a number")))
```


Chapter 13

Visitors and Folds

13.1 The Visitor Pattern

The visitor pattern is a general design pattern that allows you to separate data from functionality in an object-oriented style.

For instance, suppose you want to develop a library of ranges. Users of your library are going to want a bunch of different methods, and in principal, you can't possibly know or want to implement all of them. On the other hand, you may not want to expose the implementation details of *how* you chose to represent ranges.

The visitor pattern can help—it requires you to implement *one* method which accepts what we call a “visitor” that is then exposed to a view of the data. Any computation over shapes can be implemented as a visitor, so this one method is universal—no matter how people want to use your library, this one method is enough to ensure they can write whatever computation they want. What's better is that even if you change the representation of ranges, so long as you provide the same “view” of the data, everything will continue to work.

So here is our data definition for ranges:

```
; A Range is one of
;; - (new lo-range% Number Number)
;; Interp: represents the range between 'lo' and 'hi'
;;         including 'lo', but *not* including 'hi'

;; - (new hi-range% Number Number)
;; Interp: represents the range between 'lo' and 'hi'
;;         including 'hi', but *not* including 'lo'
```

```
;; - (new union-range% Range Range)
;; Interp: including all the numbers in both ranges

(define-class lo-range%
  (fields lo hi))

(define-class hi-range%
  (fields lo hi))

(define-class union-range%
  (fields left right))
```

We will add a single method to the interface for ranges:

```
;; The Range interface includes:
;; - visit : [RangeVisitor X] -> X
```

We haven't said what is in the `[RangeVisitor X]` interface, but the key idea is that something that implements a `[RangeVisitor X]` represents a computation over ranges that computes an `X`. So for example, if we wanted to compute where a number is included in a range, we would want to implement the `[RangeVisitor Boolean]` interface since that computation would produce a yes/no answer.

The idea, in general, of the visitor pattern is that the visitor will have a method for each variant of a union. And the method for a particular variant takes as many arguments as there are fields in that variant. In the case of a recursive union, the method takes the result of recursively visiting the data.

Under that guideline, the `[RangeVisitor X]` interface will contain 3 methods:

```
;; An [RangeVisitor X] implements:
;; lo-range : Number Number -> X
;; hi-range : Number Number -> X
;; union-range : X X -> X
```

Notice that the contracts for `lo` and `hi-range` match the contracts on the constructors for each variant, but rather than constructing a `Range`, we are computing an `X`. In the case of `union-range`, the method takes two inputs which are `Xs`, which represent the results of visiting the left and right ranges, respectively.

Now we need to implement the `visit` method in each of the `Range` classes, which will just invoke the appropriate method of the visitor on its data and recur where needed:

```
(define-class lo-range%
```

```

(fields lo hi)

(define (visit v)
  (v . lo-range (this . lo) (this . hi)))

(define-class hi-range%
  (fields lo hi)

  (define (visit v)
    (v . hi-range (this . lo) (this . hi)))

  (define-class union-range%
    (fields left right)

    (define (visit v)
      (v . union-range (this . left . visit v)
                        (this . right . visit v))))

```

We've now established the visitor pattern. Let's actually construct a visitor that does something.

We forgot to implement the `in-range?` method, but no worries – we don't need to edit our class definitions, we can just write a visitor that does the `in-range?` computation, which is an implementation of `[RangeVisitor Boolean]`:

```

;; An InRange? is an (in-range?% Number)
;; implements [RangeVisitor Boolean].

(define-class in-range%?
  (fields n)

  (define (lo-range lo hi)
    (and (>= (this . n) lo)
         (< (this . n) hi)))

  (define (hi-range lo hi)
    (and (> (this . n) lo)
         (<= (this . n) hi)))

  (define (union-range left right)
    (or left right)))

```

Now if we have our hands on a range and want to find out if a number is in the range, we just invoke the `visit` method with an instance of the `in-range?%` class:

```

(some-range-value . visit (in-range?% 5))    ;; is 5 in some-range-value ?

```

13.2 Folds

[FIXME]

13.3 Generators

generator-bad.rkt

```
#lang class/2
(require 2htdp/image class/universe)

;; A World is (world% Generator Number)
;; and implements IWorld
(define-class world%
  (fields generator num)
  ;; to-draw : -> Scene
  (define (to-draw)
    (overlay
      (text (number->string (this . num)) 20 "black")
      (empty-scene 500 500)))
  ;; on-key : Key -> World
  (define (on-key k)
    (world% (this . generator)
      ((this . generator) . pick))))

;; A Generator is a (generator% [Listof Number])
;; and implements
;; pick : -> Number
;; produce a number to show that isn't in bad
(define-class generator%
  (fields bad)
  (define (pick)
    (local [(define x (random 10))])
    (cond [(member x (this . bad)) (pick)]
          [else x])))
(check-expect (<= 0 ((generator% empty) . pick) 10) true)
(check-expect (= ((generator% (list 4)) . pick) 4) false)

(big-bang (world% (generator% empty) 0))
```

generator-register.rkt

```
#lang class/2
```

```

(require 2htdp/image class/universe)

;; A World is (world% Generator Number)
;; and implements IWorld
(define-class world%
  (fields generator num)
  ;; to-draw : -> Scene
  (define (to-draw)
    (overlay
      (text (number->string (this . num)) 20 "black")
      (empty-scene 500 500)))
  ;; on-key : Key -> World
  (define (on-key k)
    (cond [(key=? k "x")
      (local [(define g (this . generator . add-bad (this . num)))]
        (world% g (g . pick)))]
      [else
        (world% (this . generator)
          (this . generator . pick))]))))

;; A Generator is a (generator% [Listof Number])
;; and implements
;; pick : -> Number
;; produce a number to show that isn't in bad
;; add-bad : Number -> Generator
;; produce a generator like that with an additional bad number
(define-class generator%
  (fields bad)
  (define (add-bad n)
    (generator% (cons n (this . bad))))
  (define (pick)
    (local [(define x (random 10))])
    (cond [(member x (this . bad)) (pick)]
      [else x])))
(check-expect (<= 0 ((generator% empty) . pick) 10) true)
(check-expect (= ((generator% (list 4)) . pick) 4) false)
(check-expect (= (((generator% empty) . add-bad 4) . pick) 4) false)

(big-bang (world% (generator% empty) 0))

```

generator-initial.rkt

```

#lang class/3
(require 2htdp/image class/universe)

;; A World is (world% Generator Number)

```

```

;; and implements IWorld
(define-class world%
  (fields generator num)
  ;; to-draw : -> Scene
  (define (to-draw)
    (overlay
      (text (number->string (this . num)) 20 "black")
      (empty-scene 500 500)))
  ;; on-key : Key -> World
  (define (on-key k)
    (cond [(key=? "x" k)
      (begin (this . generator . tell-bad)
        (world% (this . generator)
          (this . generator . pick)))]
      [else
        (world% (this . generator)
          (this . generator . pick))]))

;; A Generator is a (generator% [Listof Number] Number)
;; interp: the list of bad numbers, and the last number picked
;; and implements
;; pick : -> Number
;; produce a number to show not in the list
;; tell-bad : ->
;; produces nothing
;; effect : changes the generator to add the last number picked to the bad list
(define-class generator%
  (fields bad last)
  (define (tell-bad)
    (set-field! bad (cons (this . last) (this . bad))))
  (define (pick)
    (local [(define rnd (random 10))])
    (cond [(member rnd (this . bad)) (this . pick)]
      [else (begin
        (set-field! last rnd)
        rnd)])))
(check-expect (<= 0 ((generator% (list 2 4 6) 0) . pick) 100) true)

(big-bang (world% (generator% (list 2 4 6) 0) 0))

(check-expect (member ((generator% (list 2 4 6) 0) . pick)
  (list 2 4 6))
  false)

(define (tell-bad-prop g)
  (local [(define picked (g . pick))])

```



```

(begin (g . tell-bad)
      (not (= picked (g . pick)))))

(check-expect (tell-bad-prop (generator% (list 1 2 3) 0)) true)

```

generator-mutate.rkt

```

#lang class/3
(require 2htdp/image class/universe)

;; A World is (world% Generator Number)
;; and implements IWorld
(define-class world%
  (fields generator num)
  ;; to-draw : -> Scene
  (define (to-draw)
    (overlay
      (text (number->string (this . num)) 20 "black")
      (empty-scene 500 500)))
  ;; on-key : Key -> World
  (define (on-key k)
    (cond [(key=? k "x")
           (world% (this . generator)
                   (this . generator . pick-bad))]
          [else
           (world% (this . generator)
                   (this . generator . pick))]))))

;; A Generator is a (generator% [Listof Number] Number)
;; interp: numbers not to pick, last number picked
;; and implements
;; pick : -> Number
;; produce a number to show that isn't in bad
;; pick-bad : -> Number
;; pick a number to show, and remember that the last one was bad
(define-class generator%
  (fields bad last)
  (define (pick-bad)
    (begin (set-field! bad (cons (this . last) (this . bad)))
           (pick)))
  (define (pick)
    (local [(define x (random 10))]
      (cond [(member x (this . bad)) (this . pick)]
            [else (begin (set-field! last x)
                          x])))))

```

```

(check-expect (<= 0 ((generator% empty 0) . pick) 10) true)
(check-expect (= ((generator% (list 4) 0) . pick) 4) false)

(big-bang (world% (generator% empty 0) 0))

```

13.4 Exercises

13.4.1 Quick visits

This problem builds on the *quick lists* problem.

Here was the interface you should have implemented for lists using the *quick list* data structure that supports a fast `list-ref` method:

```

;; A [List X] implements
;; - cons : X -> [List X]
;;   Cons given element on to this list.
;; - first : -> X
;;   Get the first element of this list
;;   (only defined on non-empty lists).
;; - rest : -> [List X]
;;   Get the rest of this
;;   (only defined on non-empty lists).
;; - list-ref : Natural -> X
;;   Get the ith element of this list
;;   (only defined for lists of i+1 or more elements).
;; - length : -> Natural
;;   Compute the number of elements in this list.

;; empty is a [List X] for any X.

```

Make sure your quick list implementation is working and place it into a file named `"quick-lists.rkt"`. That file should provide one name, `empty`, by including the following at the top of the file:

```
(provide empty)
```

In a file named `"slow-lists.rkt"` re-develop an implementation of the list interface, but in the usual way as a recursive union of `mt%` and `cons%` classes. That file should also provide `empty` by including the same line above at the top.

Finally, start a third file called `"use-lists.rkt"` that will make use of both kinds of lists by including the following at the top of the file:

```
(require (prefix-in q: "quick-lists.rkt"))
(require (prefix-in s: "slow-lists.rkt"))
```

You now have two lists: `q:empty` and `s:empty`; both are represented in very different ways, but so long as you use them according to the list interface, they should be indistinguishable.

Let's now revise the `[List X]` interface to include support for visitors:

```
;; A [List X] implements ...
;; - accept : [ListVisitor X Y] -> Y
;;   Accept given visitor and visit this list's data.

;; A [ListVisitor X Y] implements
;; - visit-mt : -> Y
;;   Visit an empty list.
;; - visit-cons : X [Listof X] -> Y
;;   Visit a cons lists.
```

Implement the revised `[List X]` interface in both `"quick-lists.rkt"` and `"slow-lists.rkt"`.

In `"use-lists.rkt"` you should be able to define particular visitors and have it work on *both* representations of lists. As an example, here is a list visitor that computes the length of a list:

```
;; A (new length%) implements [ListVisitor X Natural].
;; List visitor for computing the length of a list.
(define-class length%
  (define (visit-mt) 0)
  (define (visit-cons x r)
    (add1 (r . accept this))))

(define len (new length%))

(check-expect (q:empty . accept len) 0)
(check-expect (s:empty . accept len) 0)
(check-expect (q:empty . cons 'c . cons 'b . cons 'a . accept len) 3)
(check-expect (s:empty . cons 'c . cons 'b . cons 'a . accept len) 3)
```

And here's one for the sum of a list of numbers:

```
;; A (new sum%) implements [ListVisitor Number Number].
;; List visitor for computing the sum of a list of numbers.
(define-class sum%
```

```

(define (visit-mt) 0)
(define (visit-cons n r)
  (+ n (r . accept this)))

(define sum (new sum%))

(check-expect (q:empty . accept sum) 0)
(check-expect (s:empty . accept sum) 0)
(check-expect (q:empty . cons 3 . cons 4 . cons 7 . accept sum) 14)
(check-expect (s:empty . cons 3 . cons 4 . cons 7 . accept sum) 14)

```

Implement a `[ListVisitor X X]` named `reverse%` that reverses a list (note: you may need to implement a “helper” visitor that corresponds to the helper function you’d write for the `reverse` function). Note that this visitor will have to commit to produces either a quick list or a slow list, but it really doesn’t really matter which... well, except for testing. So for example, let’s say the reverse visitor produces slow lists. Then we would expect the following test to pass, assuming `reverse%` works as specified:

```

(define rev (new reverse%))

(check-expect (q:empty . accept rev) s:empty)
(check-expect (q:empty . cons 'c . cons 'b . cons 'a . accept rev)
  (s:empty . cons 'a . cons 'b . cons 'c))

```

Of course, this isn’t ideal since our *test* is testing more than is actually required of `reverse%`. In particular, it should be perfectly acceptable for `reverse%` to produce quick lists without tests failing.

What’s happening here is that `check-expect` is checking too much because it is not treating the objects it compares solely according to their interface. We will see how to fix this problem by defining an interface-respecting equality computation, but for now, just test as shown above.

Now to build some larger pieces with visitors. First, here’s an interface definition for functional objects that represent functions from `Xs` to `Ys`. Such an object has a single method called `apply` that consumes an `X` and produces a `Y`:

```

;; A [Fun X Y] implements
;; - apply : X -> Y
;;   Apply this function to given x.

```

Now implement the following two visitors:

```

;; A (new filter% [Predicate X]) implements [ListVisitor X [List X]].
;; Filters visited list to produce a list of elements satisfying predicate.

```

```
;; A (new map% [Fun X Y]) implements [ListVisitor X [List Y]].
;; Maps visited list to produce a list of results of applying the function.
```

Implement at least one `[Fun Natural String]` and one `[Predicate String]` to use for testing `filter%` and `map%`.

13.4.2 Folds vs Visitors

We can also implements *folds* over lists, in both for both kinds of lists. Extend your implementation of lists (both kinds) to support the `fold` method:

```
;; A [List X] implements ...
;; - fold : [ListFold X Y] -> Y
;;   Accept given fold and process this list's data.

;; A [ListFold X Y] implements
;; - fold-mt : -> Y
;;   Process an empty list.
;; - fold-cons : X Y -> Y
;;   Process a cons lists.
```

Now revise your implementations of the `filter%` and `map%` to implement folds as well as visitors. Be sure to specify what interfaces they implement.

Finally, implement the class `list-ref%`:

```
;; A (new list-ref% Number) implements [ListVisitor X X]
;; Retrieves the element at the specified index.
```

Could you implement this using the `ListFold` interface? Which was more elegant for `map%` and `filter%`?

13.4.3 JSON visitor

Develop the visitor pattern for JSON values.

Design an equality visitor for JSON values.

Part V

Mutation

Chapter 14

Ch-Ch-Ch-Ch-Changes

We want to design a class, `counter%`, with the following interface

```
; m : -> Number
;; Produce the number of times 'm' has been called
```

Now let's try to implement this class.

```
(define-class counter%
  (fields called)
  (define (m)
    hmmm))
```

Unfortunately, it's not immediately clear what to put in the body of `m`. We can understand our task better by writing examples.

```
(check-expect ((counter% 0) . m) 1)
(check-expect ((counter% 4) . m) 5)
```

This suggests the following implementation:

counter%

```
(define (m)
  (add1 (this . called)))
```

Now our all of our tests pass.

However, when we try our a few more examples, we see this:

```

> (define c (counter% 0))

> (send c m)
1
> (send c m)
1

```

Of course, this is the wrong answer. We shouldn't be surprised, since nothing has changed about `c`—in fact, nothing ever happens to `c`, and only one `counter%` instance is produced in this program. In order to give `m` the ability to remember things, we will need to do something to get a different `counter%`.

One possibility is to change `m` to produce both the desired result *and* a new counter.

```

> (define-struct r (n new-counter))

> (define-class counter%
  (fields called)
  (define (m)
    (make-r
     (add1 (send this called))
     (counter% (add1 (send this called))))))

> (define c (counter% 0))

> (send c m)
(make-r 1 (object:counter% 1))
> (define d (r-new-counter (send c m)))

> d
(object:counter% 1)
> (send d m)
(make-r 2 (object:counter% 2))

```

So far, so good—we can get a new `counter%`, and when we use that new value, we get the right answer. However, we haven't solved the problem yet:

```

> (send c m)
(make-r 1 (object:counter% 1))

```

This is the same answer that we had before, and not the desired one.

In fact, this behavior is the result of one of the important design principles of this class, and of Fundies 1, up until this point. If you call a function or method with the same inputs, you get the same result. Always!

Unfortunately, that make it impossible to implement `m`, because `ms` spec violates this assumption—if you call it, it is required to produce a *different* result from the last time it was called.

Previously, we’ve always been able to rely on this test passing, regardless of what you put in `E`

```
(check-expect E E)
```

Actually, it turns out that there have been a few exceptions to this rule:

- `(random 5)`
- User input, such as in `big-bang`

Now, however, we are proposing a much more fundament violation of this principle.

Before we violate the principle, though, let’s look at one more possible idea: accumulators.

We could add an accumulator to `m`, which is the previous number of times we’ve been called. We’ve used this solution before to create functions and methods that remember previous information. In this case, though, accumulators are a non-solution. If we add an accumulator to `m` to indicate what we’re remembering, we get this method:

```
counter%
```

```
(define (m accum) (add1 accum))
```

But that’s a pretty boring method—it’s just a wrapper around `add1`. And it’s not a solution to our problem: instead of the `counter%` class or the `m` method remembering the number of times we’ve called `m`, we have to remember it ourselves, and provide it as input every time we call `m`.

To truly solve our problem, and implement `m`, we need new language support. This support is provided in `class/3`.

The `class/3` language provides the new `set-field!` form, which is used like this:

```
(set-field! f new-value)
```

This *changes* the value of the field named `f` in `this` object to whatever `new-value` is.

We can now revise our definition of `m` to

```
counter%
```

```
(define (m)
  (begin (set-field! called (add1 (send this called)))
        (add1 (send this called))))
```

Note that `set-field!` *doesn't produce* a new version of the field, instead it *changes* the field named `called` to something new.

We've also introduced one more language feature in `class/3: begin`. The `begin` form works by evaluating each expression in turn, throwing away the result of every expression except that last one. Then it does the last part, and produces that result.

Unlike `set-field!`, `begin` doesn't add any new capability to the language. For example, we can simulate `begin` using `local`. For example:

```
(local [(define dummy (set-field! called (add1 (send this called))))]
  (add1 (send this called)))
```

This is very verbose, and requires creating new variables like `dummy` that are never used. Therefore, `begin` is a useful addition to our language, now that we work with expressions like `set-field!` that don't produce any useful results.

Now Expressions do two things: - produce a result (everything does this) - has some effect (some expressions do this)

Now we write effect statements. Have to write them for every method/function that has an effect.

counter%

```
;; m : -> Number
;; Produce the number of times m has been called
;; Effect : increment the called field
(define (m)
  (begin (set-field! called (add1 (send this called)))
        (add1 (send this called))))
```

We've lost a lot of reasoning power but gained expressiveness.

What have I really gained, though?

Imagine that you're modeling bank financial systems. You want to deposit money into the account, and then the money should be there afterwards.

```
;; An Account is (account% Number)
(define-class account%
  (fields amt)

  ;; Number -> Account
```

Question: How would we do something like this in a purely functional language?

Answer: We would do something similar to the `make-r` approach

Question: How would we have `begin`?
Answer: No, in racket, this approach is frequently used.

A brief discussion of `void`

What happens if we return the result of `set-field!`? It produces nothing—DrRacket doesn't print anything at all.

However, there's no way for DrRacket to truly have nothing at all, so it has an internal value called `void`. This value doesn't have any uses, though, and you shouldn't ever see it.

```
(define (deposit n)
  (account% (+ (this . amt) n)))
```

But this doesn't model bank accounts properly.

I deposit, my valentine deposits, I deposit – whoops!

New version:

```
;; An Account is (account% Number)
(define-class account%
  (fields amt)

  ;; Number -> Void
  ;; Effect: increases the field amt by n
  ;; Purpose: add money to this account
  (define (deposit n)
    (set-field! amt (+ (this . amt) n))))
```

Note that we don't need to produce any result at all.

```
;; A Person is (person% String Account Number)
(define-class person%
  (fields name bank paycheck)
  ;; -> Void
  ;; Deposit the appropriate amount
  ;; Effect: changes the the bank account amt
  (define (pay)
    (this . bank . deposit (this . paycheck))))
```

```
> (define dvh-acct (account% 0))

> (define dvh (person% "DVH" dvh-acct 150))

> (define sweetie (person% "Sweetie" dvh-acct 3000))

> (send dvh pay)

> dvh-acct
(object:account% 150)
> (send sweetie pay)

> dvh-acct
(object:account% 3150)
```

Note that we *cannot* replace `dvh-acct` with `(account% 0)` – we’d get totally different results.

Now equality is much more subtle – intensional equality vs extensional equality. Same fork example.

What if we do:

```
> (define new-acct dvh-acct)

> (define p (person% "Fred" new-acct 400))

> (send p pay)

; updated
> dvh-acct
(object:account% 3550)
```

What if we create new `account%` with 0? Then the effects are not shared.

What if we do:

```
> (define x (send dvh-acct amt))

> x
3550
> (send dvh pay)

; still the same
> x
3550
```

What if we do

```
> (define y (send dvh bank))

> y
(object:account% 3700)
> (send dvh pay)

; now different
> y
(object:account% 3850)
```

The difference is that `x` is the name of a number, and numbers don't change, but `y` is the name of an account, and accounts change over time.

Objects can change, but other things do not change. Structures and lists can contain objects that change, but the structures and lists themselves do not change, the object they point to are the same objects.

Testing is hard with mutation. Give an example in the notes.

Chapter 15

Circular Data

Books & Authors

Books have: title : String author : Author

Authors have: name : String books : [Listof Book]

As data def:

```
;; A Book is (book% String Author)
;; An Author is (author% String [Listof Book])
```

Can we make an Author?

```
(author% "Rose" empty)
(book% "Reign of Roquet" (author% "Rose" empty))
```

But this is wrong: the Rose has written a book, but the author object doesn't know about it.

Do we need books to know the author? Yes.

We've seen this before with graph structure. We represented graphs as association lists, using symbolic names.

```
;; A Book is (book% String Author)
(define-class book%
```

```

(fields title author))

;; An Author is (author% String [Listof Book])
(define-class author%
  (fields name books))

(define rose (author% "Rose" empty))
(define reign (book% "Reign of Roquet" rose))

```

But:

```

reign
rose

```

Question: Does `reign` contain a copy of `rose`, or are they all the same `rose`? Answer: always the same, because we use the name `rose`, we didn't construct a new one.

Let's add a new method for modifying the author after a book is written:

```

(define (add-book b)
  (set-field! books (cons b (this . books))))

```

Now we change our example:

```

(define rose (author% "Rose" empty))
(define reign (book% "Reign of Roquet" rose))
(rose . add-book reign)

```

How does it print?

```

> rose
#0=(object:author% "Rose" (list (object:book% "Reign of Roquet" #0#)))
> reign
#0=(object:book% "Reign of Roquet" (object:author% "Rose" '#0#))

```

See graph-style printing.

But every times we construct a book with an author, we want to use `add-book`. So, let's use the constructor.

```

(constructor (t a)
  (fields t a)
  (send a add-book this))

```

In the first expression, we *cannot* use `this`, and we must produce the result using `fields`. Later, we can use `this`, and we get the desired result.

Chapter 16

Back-channels

Up until this point, computations communicate by consuming arguments and producing values. Mutation enables more channels of communication between computations via *shared* mutable data.

As an example, recall our counter world program from section 15.3:

```
;; A Counter is a (counter-world% Natural)
(define-class counter-world%
  (fields n)
  ...
  ;; on-tick : -> Counter
  (define (on-tick)
    (new counter-world% (add1 (this . n)))))

(big-bang (new counter-world% 0))
```

To illustrate how mutable data provides alternative channels of communication, let's develop a variant of the program that communicates the state of the world through an implicit stateful back-channel.

```
;; A Counter is a (counter-world% Natural)
(define-class counter-world%
  (fields n)
  ...
  ;; on-tick : -> Counter
  (define (on-tick)
    (begin (set-field! n (add1 (send this n)))
           this)))
```

```
(big-bang (new counter-world% 0))
```

Notice how the new `on-tick` method doesn't produce a new `counter-world%` object with an incremented counter; instead it *mutates* its own counter and returns its (mutilated) self.

You'll find that this program appears to behave just like the old version, but backchannels open up new forms of interaction (and interference) that may not be intended. For example, can you predict what this program will do?

```
(launch-many-worlds (big-bang (new counter-world% 0))  
                    (big-bang (new counter-world% 0)))
```

How about this (seemingly equivalent) one?

```
(define w (new counter-world% 0))  
(launch-many-worlds (big-bang w)  
                    (big-bang w))
```

Chapter 17

Intensional equality

Mutation exposes yet another sense in which two things may be consider "*the same*": two things are the same if mutating one mutates the other.

How can we tell if two posns are two different names for the same thing, or if they're two different posns with the same contents?

For example:

```
(define p1 (posn% 3 4))
(define p2 (posn% 3 4))
```

or

```
(define p1 (posn% 3 4))
(define p2 p1)
```

These are very different in the presence of mutation. We have a way of testing this: `eq?`. Similarly, the `equal?` function checks structural equality.

But that didn't shed any light on the question. Is there a way we can check this *in* our language?

Answer: yes. Do some operation to one of them that changes the state of the object, and see if it *also* happens to the other one.

Drawback: you can't necessarily undo this operation.

```
(define (really-the-same? p1 p2)
  ....)
```

Now we need some operation to perform on `p1`.

posn%

```
;; -> (posn% 'black-eye Number)
(define (punch!)
  (begin
    (set-field! x 'black-eye)
    this))

(define sam (posn% 3 4))
(send sam punch!)
```

"I punched him so hard, I punched him right out of the data definition."

Now we can define `really-the-same?`.

```
(define (really-the-same? p1 p2)
  (begin
    (send p1 punch!)
    (symbol? (send p2 x))))

(really-the-same? p1 p2)
p1
```

Now `p1` is permanently broken, and can't be undone. So `really-the-same?` is a very problematic, and you should use `eq?`, which uses DrRacket's internal knowledge of where things came from to answer this question without changing the objects.

Question:

```
(eq? p1 (send p1 punch!))
```

Produces true.

```
(send p2 =? (send p2 punch!))
```

Produces true (or crashes).

```
(send (send p3 punch!) =? p3)
```

Produces true (or crashes).

Question: Does intensional equality imply structural equality? Yes.

Part VI

Java

Chapter 18

Java

18.1 Two Ideas: Java and Types

Types are a *mechanism* for enforcing data definitions and contracts.

Java is a programming language like the one we've seen, with a different syntax and with types.

18.2 Programming in Java

18.2.1 Java Syntax

```
;; A C is (C Number String String)
(define-class C
  (fields x y z)
  ;; Number Number -> C
  (define (m p q)
    ...))

class C {
  Number x;
  String y;
  String z;
  public C m(Number p, Number q) {
    ...
  }
}
```

```
        public C(Number x, String y, String z) {
            this.x = x;
            this.y = y;
            this.z = z;
        }
    }
}
```

Let's create a simple pair of numbers:

```
class Pair {
    Number left;
    Number right;

    public Pair(Number left, Number right) {
        this.left = left;
        this.right = right;
    }

    public Pair swap() {
        return new Pair(this.right, this.left);
    }
}
```

But really, this doesn't work, because Java doesn't have the type `Number`. So we'll choose `Integer` instead.

```
class Pair {
    Integer left;
    Integer right;

    public Pair(Integer left, Integer right) {
        this.left = left;
        this.right = right;
    }

    public Pair swap() {
        return new Pair(this.right, this.left);
    }
}
```

To test this, we'll import a testing library:

```
import tester.Tester;
```

and write some examples:

```
class Examples {
```

```
public Examples() {}  
public boolean testSwap(Tester t) {  
    return t.checkExpect(new Pair(3,4).swap(),  
                          new Pair(4,3));  
}  
}
```

18.3 Running Java Programs

We don't have a Run button in Java, so we need a different way to run our program. To do this, we first need to install our test library. This requires installing a JAR file from the NU Tester web site.

The we have to compile the program.

- javac
- The classpath and the -cp option
- Now we have class files, which are binary and all mashed up
- To run this, we use the java command, which also has a -cp option
- If we change things, we have to recompile and then rerun.

18.4 A More Complex Example

What if we want to represent a union?

```
class Square {  
    Integer size;  
    public Square(Integer size) {  
        this.size = size;  
    }  
}  
class Circ {  
    Integer radius;  
    public Circ(Integer radius) {  
        this.radius = radius;  
    }  
}
```

How do we declare that both of these are Shapes?

```
import tester.Tester;

interface IShape {}

class Square implements IShape {
    Integer size;
    public Square(Integer size) {
        this.size = size;
    }
    public IShape nothing(IShape i) {
        return i;
    }
}

class Circ implements IShape {
    Integer radius;
    public Circ(Integer radius) {
        this.radius = radius;
    }
}

class Examples {
    Examples () {}
    public boolean testNothing(Tester t) {
        Square s = new Square(5); // A local binding
        return t.checkExpect(s.nothing(new Circ(2)),
                               new Circ(2));
    }
}
```

18.5 Recursive Unions

```
import tester.Tester;
class Mt implements IList {
    public Mt() {}
}

class Cons implements IList {
    Integer first;
    IList rest;

    public Cons(Integer first, IList rest) {
```

```

        this.first = first;
        this.rest = rest;
    }
}

interface IList {}

class Examples {
    public Examples() {}

    public boolean testList(Tester t) {
        return t.checkExpect(new Mt(), new Mt())
            && t.checkExpect(new Cons(5, new Mt()), new Cons(5, new Mt()));
    }
}

```

18.6 Enumerations

In Fundies I, we might have written:

```

;; A Title is one of
;; - 'dr
;; - 'mr
;; - 'ms

```

In Java, we write:

```

interface ITitle {}
class Dr implements ITitle {
    Dr() {}
}
class Mr implements ITitle {
    Mr() {}
}
class Ms implements ITitle {
    Ms() {}
}

```

Why write these silly constructors?

```

interface ITitle {}
class Dr implements ITitle {
    Dr() {}
}

```

```
}
class Mr implements ITitle {
    Mr() {}
}
class Ms implements ITitle {
    Integer x;
    Ms() {}

    public Integer m() {
        return x+1;
    }
}

class Main {
    public static void main(String[] args) {
        new Ms().m();
        return;
    }
}
```

Now we get a `NullPointerException`. But what is that?

A discussion of the evils of `null`.

For example, this compiles:

```
interface ITitle {}
class Dr implements ITitle {
    Dr() {}
}
class Mr implements ITitle {
    Mr() {}
}
class Ms implements ITitle {
    Integer x;
    Ms(Integer x) {
        this.x = x;
    }

    public Integer m() {
        return null;
    }
}

class Main {
    public static void main(String[] args) {
```



```
        new Ms().m();
        return;
    }
}
```

Never write null in your program!

A long sermon on null.

18.7 Parameterized Data Definitions

Consider our Pair class:

```
class Pair {
    Integer left;
    Integer right;

    public Pair(Integer left, Integer right) {
        this.left = left;
        this.right = right;
    }

    public Pair swap() {
        return new Pair(this.right, this.left);
    }
}
```

Now if we want a Pair of Strings:

```
class PairString {
    String left;
    String right;

    public Pair(String left, String right) {
        this.left = left;
        this.right = right;
    }

    public Pair swap() {
        return new Pair(this.right, this.left);
    }
}
```

This is obviously bad—we had to copy and paste. So let's abstract:

```
class Pair<T,V> {
    T left;
    V right;

    public Pair(T left, V right) {
        this.left = left;
        this.right = right;
    }

    public Pair<V,T> swap() {
        return new Pair<V,T>(this.right, this.left);
    }
}

class Examples {
    public Examples() {}
    public boolean testSwap(Tester t) {
        return t.checkExpect(new Pair<Integer,Integer>(3,4).swap(),
                               new Pair<Integer,Integer>(4,3));
    }
}
```

18.8 Abstraction

```
class C {
    Integer x;
    Integer y;
    C(Integer x, Integer y) {
        this.x = x;
        this.y = y;
    }

    public Integer sq() {
        return this.x * this.x;
    }
}
```

```
class D {
    Integer x;
    String z;
```

```
D(Integer x, String z) {
    this.x = x;
    this.z = z;
}

public Integer sq() {
    return this.x * this.x;
}
}
```

Now to abstract:

```
class S {
    Integer x;
    public Integer sq() {
        return this.x * this.x;
    }
    S(Integer x) {
        this.x = x;
    }
}

class C extends S {
    Integer y;
    C(Integer x, Integer y) {
        super(x);
        this.y = y;
    }
}

class D {
    Integer x;
    String z;
    D(Integer x, String z) {
        this.x = x;
        this.z = z;
    }

    public Integer sq() {
        return this.x * this.x;
    }
}
```

18.9 Types

What sorts of things count as *types* in Java?

- Class names
- Interface names
- Other stuff: `int`, `boolean`, ...

What should be a part of the contract and purpose in Java? Well, we don't need to write down things that are already part of the type. If the contract corresponds to the type, you don't have to repeat it. However, some contracts can't be checked by the type system—you should still write those down.

Chapter 19

Extensional Equality in Java

19.1 Posn

19.2 Equality in Java

```
class Posn {
    Integer x;
    Integer y;
    Posn(Integer x, Integer y) {
        this.x = x;
        this.y = y;
    }

    public Boolean isEqual(Posn p) {
        return this.x == p.x
            && this.y == p.y;
    }
}

interface LoP {
    Boolean isEmpty();
    Posn getFirst();
    LoP getRest();
}

class MT implements LoP {
```

```
MT() {}

public Posn getFirst() {
    return null;
}

public LoP getRest() {
    return ???;
}

public Boolean isEmpty() {
    return true;
}

public Boolean isEqual(LoP lop) {
    return lop.isEmpty();
}
}

class Cons implements LoP {
    Posn first;
    LoP rest;

    Cons(Posn first, LoP rest) {
        this.first = first;
        this.rest = rest;
    }

    public Boolean isEmpty() {
        return false;
    }

    public Boolean isEqual(LoP lop) {
        return (!lop.isEmpty())
            && this.first.isEqual(lop.getFirst())
            && this.rest.isEqual(lop.getRest());
    }

    public Posn getFirst() {
        return this.first;
    }

    public LoP getRest() {
        return this.rest;
    }
}
```

```
}
```

19.3 List of Posn

```
interface LoP {
    Boolean isEmpty();
    Posn getFirst();
    LoP getRest();
    Boolean isEqual();
}

class MT implements LoP {
    MT() {}

    public Posn getFirst() {
        return ???;
    }

    public LoP getRest() {
        return ???;
    }

    public Boolean isEmpty() {
        return true;
    }

    public Boolean isEqual(LoP lop) {
        return lop.isEmpty();
    }
}

class Cons implements LoP {
    Posn first;
    LoP rest;

    Cons(Posn first, LoP rest) {
        this.first = first;
        this.rest = rest;
    }

    public Boolean isEmpty() {
```

```

        return false;
    }

    public Boolean isEqual(LoP lop) {
        return (!lop.isEmpty()
            && this.first.isEqual(lop.getFirst())
            && this.rest.isEqual(lop.getRest()));
    }

    public Posn getFirst() {
        return this.first;
    }

    public LoP getRest() {
        return this.rest;
    }
}

```

How did we get into this situation? Let's look at `isEqual` again:

```

public Boolean isEqual(LoP lop) {
    return (!lop.isEmpty()
        && this.first.isEqual(lop.getFirst())
        && this.rest.isEqual(lop.getRest()));
}

```

After the first conjunct, we know that `lop` is a `Cons`, and so we want to get the first and rest. But *Java* doesn't know that, and so we created the `getFirst` and `getRest` methods.

If we return `null`, then we create problems for all possible clients of the `getFirst` method. So instead we raise an error:

```

public Posn getFirst() {
    throw new RuntimeException("Can't take the first of an empty list.")
}

public Posn getRest() {
    throw new RuntimeException("Can't take the rest of an empty list.")
}

```

But if we think about `isEqual` again, can we help Java do something smarter? Let's try taking `getFirst` and `getRest` out of the interface. If we do that, we get an error message about not finding those methods in the interface. But can we persuade Java

that `lop` is really a `Cons`? Yes:

```
public Boolean isEqual(LoP lop) {
    return (!lop.isEmpty())
        && this.first.isEqual(((Cons)lop).getFirst())
        && this.rest.isEqual(((Cons)lop).getRest());
}
```

This is called *casting*, and it's crucial for (some kinds of) programming using the Java type system. However, it's important to remember that we're cheating the type system here. The cast is turned into a runtime check, and we could write anything down that we want. It's only when running the program that Java can check whether `lop` is really a `Cons`.

19.4 Equality and Parameterized Types

We try writing a parameterized version of `Listof<X>` with equality.

It doesn't work, because `X` doesn't have an `isEqual` method.

Fix: restrict `X` to implement the `IEqual` interface.

Question about structural vs nominal typing.

Discussion about polymorphism/bounds/extends/etc.

The class language version:

```
;; An [IEqual X] implements
;; =? : X -> Boolean

;; A [Listof X] implements
;; empty?: -> Boolean
;; and implements [IEqual [Listof X]]
;; where X implements [IEqual X]
```

If we unroll our definitions a little, we get:

```
;; An [IEqual X] implements
;; =? : X -> Boolean

;; A [Listof X] implements
;; empty? : -> Boolean
;; =? : [Listof X] -> Boolean
;; where X implements
```

```
;; =? : X -> Boolean
```

19.5 Comparing different kinds of things

This works:

```
(equal? 5 "fred")
```

This doesn't work:

```
new Posn(3,4).isEqual("fred")
```

The reason this doesn't work is that we're violating the contract on `isEqual`, which is enforced by the type system.

So let's create a different `isEqual` method:

```
public Boolean isEqualPosn(Posn p) {
    return this.x == p.x
        && this.y == p.y;
}

public Boolean isEqual(Object o) {
    return (o instanceof Posn)
        && this.x == p.x
        && this.y == p.y;
}
```

This doesn't typecheck, so we have to add casts again.

```
public Boolean isEqual(Object o) {
    return (o instanceof Posn)
        && this.x == ((Posn)p).x
        && this.y == ((Posn)p).y;
}
```

Or, we can use our way of comparing Posns.

```
public Boolean isEqual(Object o) {
    return (o instanceof Posn)
        && this.isEqualPosn((Posn)o);
}
```

But why would we want to compare things of different types? It turns out that Java has a built-in notion of equality, called the `equals` method. And this method expects

its input to be an `Object`.

We can reuse this implementation:

```
public Boolean equals(Object o) {  
    return (o instanceof Posn)  
        && this.isEqualPosn((Posn)o);  
}
```

There are two big problems with this:

- If we extend `Posn`, then `equals` will behave differently in different directions.
- We forgot to override `hashCode`.

19.5.1 equals and hashCode

The fundamental rules for overriding `equal`:

- If you override `equals`, you **must** override `hashCode`.
- `o.equals(p)` implies `o.hashCode() == p.hashCode()`

The rule of `hashCode`: if two objects have different hash codes, they are different. `hashCode` is a fast way to check if two objects are different.

Here's a quick and correct implementation of `hashCode`:

```
public int hashCode() {  
    return 0;  
}
```


Chapter 20

Iterating over Data

Part VII

A Class of Your Own

Chapter 21

Under the Hood: Implementing OO

21.1 Pulling back the veil from object-oriented programming

We will implement OO in ISL+lambda.

In Fundies 1 we saw a way to implement the language we were writing programs in. Today we're going to do something similar. How can we do this?

We could write a Java compiler, which would help us program Java, but the JLS is like 900 pages long.

Structs with functions in them. That would be similar to having objects because we'd have data with functionality bundled together.

We could also implement dictionary and use those to associate values and fields and values and methods.

We could use built-in Racket objects. That would be similar to what we've done, and would be similar to structure and functions.

We could use functions to represents objects.

We could write an interpreter for `class/1` in ISL+.

We are going to start out by using functions to represent objects. Ask yourself:

WHAT IS AN OBJECT?

- Data + functions - this is the "how they are made" or "what they are constructed out of".

21.2 Objects as Functions

Another view is: what do objects do?

- Objects respond to messages.

What are messages? Messages are names.

So let's write something that responds to things that are names. Let's make an example of something that responds to messages. As an example, we'll make a square object that responds to the messages:

- side
- area

If we make a square-10 object, what's its contract:

```
;; square-10 : Message -> Number
```

How should we represent messages? Symbols.

```
;; A Message is a Symbol
```

```
;; square-10 : Message -> Number
(define (square-10 msg)
  (cond [(symbol=? msg 'side) 10]
        [(symbol=? msg 'area) 100]
        [else "message not understood"])))
```

```
(check-expect (square-10 'side) 10)
(check-expect (square-10 'area) 100)
(check-error (square-10 'bad))
```

How would we write a square-5 object?

```
;; square-5 : Message -> Number
(define (square-5 msg)
  (cond [(symbol=? msg 'side) 5]
        [(symbol=? msg 'area) 25]
        [else "message not understood"])))
```

```
(check-expect (square-5 'side) 5)
(check-expect (square-5 'area) 25)
(check-error  (square-5 'bad))
```

Now we have two simple objects that look very similar. Let's abstract.

```
;; A Square is a Message -> Number
;; square% : Number -> Square
(define (square% side)
  (local [(define (the-square msg)
            (cond [(symbol=? msg 'side) side]
                  [(symbol=? msg 'area) (sqr side)]
                  [else (error "message not understood")]))])
    the-square))

(define square-10 (square% 10))
(define square-5  (square% 5))
```

Why is what've done a little weird considering we said that an object is data plus functions.

Depending on your perspective:

- We only have data.
- We only have functions.

Our messages are always just a symbol. But what about arguments?

But there's something else. Where did the data go?

The trick is that when we produce the square function, it remembers the values it can see, like side, when it was created. So the data is remember in the function.

Functions are really code plus data.

Basically a list of fields that map to values, plus code.

That's why we are able to implement objects so easily.

But how do we do inheritance?

Let's create another kind of shape – let's create circles.

```
;; A Circle is a Message -> Number
;; Number -> Circle
(define (circle% radius)
  (local [(define (the-circle msg))
```

```

      (cond [(symbol=? msg 'radius) radius]
            [(symbol=? msg 'area) (* radius radius pi)]
            [else (error "message not understood")]))
    the-circle))

```

```

(define circle-2 (circle% 2))
(check-expect (circle-2 'radius) 2)
(check-within (circle-2 'area) (* 2 2 pi) 0.0001)

```

There's a lot of repeated code here. How can we abstract something common to both of these definitions.

We have different code for handling message, but all of the objects have the same code for the message that is not understood.

```

(define (dumb-object msg)
  (error "message not understood"))

```

This is an object – not a constructor for an object.

Let's write a constructor for this:

```

(define (dumb%)
  (local [(define (the-dumb-object msg)
              (error "message not understood"))]
    the-dumb-object))

```

Now how could we use this?

```

(define (square% side super)
  (local [(define (the-square msg)
              (cond [(symbol=? msg 'side) side]
                    [(symbol=? msg 'area) (sqr side)]
                    [else ((super) msg)]))]
    the-square))

```

This is odd – we can make squares with different super classes. Let's fix that.

```

(define (square% side)
  (local [(define (the-square msg)
              (cond [(symbol=? msg 'side) side]
                    [(symbol=? msg 'area) (sqr side)]
                    [else ((dumb%) msg)]))]
    the-square))

```

dumb% is actually object%.

```

(define (object%)

```

```
(local [(define (the-dumb-object msg)
  (error "message not understood"))]
  the-dumb-object))
```

How many times are we going to create an object% object? How do we have it happen only once?

```
(define (square% side)
  (local [(define super (object%))
    (define (the-square msg)
      (cond [(symbol=? msg 'side) side]
            [(symbol=? msg 'area) (sqr side)]
            [else (super msg)]))]
    the-square))
```

And likewise for circle:

```
(define (circle% radius)
  (local [(define super (object%))
    (define (the-circle msg)
      (cond [(symbol=? msg 'radius) radius]
            [(symbol=? msg 'area) (* radius radius pi)]
            [else (super msg)]))]
    the-circle))
```

We've now abstracted out the behavior of the error message.

But, let's come back to the observation that we can add methods to object% that every object will now understand.

For example, we could add `=?`, but that sounds hard. Let's do something really simple:

```
(define (object%)
  (local [(define (the-dumb-object msg)
    (cond [(symbol=? msg 'hi) "Howdy"]
          [else (error "message not understood"])]))
    the-dumb-object))
```

But we've broken the contract. So we'll say instead that an object can respond to any message and produce anything.

OK, let's see how it works:

```
(check-expect ((square% 10) 'hi) "Howdy")
```

Look! We have inheritance! All instances of subclasses of object% understand the hi message!

This is almost everything that is going on under the hood in classN, Java, Ruby, etc.

But does this do overriding?

```
(define (square% side)
  (local [(define super (object%))
          (define (the-square msg)
            (cond [(symbol=? msg 'side) side]
                  [(symbol=? msg 'area) (sqr side)]
                  ;; Overriding the hi method.
                  [(symbol=? msg 'hi) "Good day, sir."]
                  [else (super msg)])])
    the-square))
```

Great, but none of our messages take arguments. How can we do that? We could change our representation of a messages to include arguments.

```
;; A Message is a (make-msg Symbol [Listof Anything]).
(define-struct msg (name args))

;; Symbol Message -> Boolean
(define (msg-is? sym msg)
  (symbol=? sym (msg-name msg)))

(define (object%)
  (local [(define (the-dumb-object msg)
            (cond [(msg-is? 'hi msg) "Howdy"]
                  [else (error "message not understood")]))]
    the-dumb-object))

(check-expect ((object%) (make-msg 'hi empty)) "Howdy")
```

Another approach, revert back to Message = Symbol.

```
;; An Object is a Message [Listof Symbol] -> Anything

(define (object%)
  (local [(define (the-dumb-object msg args)
            (cond [(symbol=? 'hi msg) "Howdy"]
                  [else (error "message not understood")]))]
    the-dumb-object))
```

Another approach, return a function that takes the arguments.

Suppose we want to add a multiply method to circles.

```
(define (circle% radius)
```

```
(local [(define super (object%))
        (define (the-circle msg)
          (cond [(symbol=? msg 'radius) radius]
                [(symbol=? msg 'area) (* radius radius pi)]
                [(symbol=? msg 'multiply)
                 (lambda (factor)
                   (circle% (* factor radius)))]
                [else (super msg)]))]
  the-circle))
```

These contracts suck. We really want to talk about the contract of each method that is supported by an object.

```
;; A Circle is a Object that implements
;; 'radius -> Number
;; 'area -> Number
;; 'multiply -> (Number -> Circle)
```

```
((circle% 10) 'multiply) 4 => circle with radius 40
```

```
(check-expect (((circle% 10) 'multiply) 4) 'radius) 40)
```

So we've got classes, objects, inheritance, overriding, and basically everything you'd want in a class system.

We might like to have a nice notation to make it more convenient to write programs in this style, but this is really all that is going on.

Suppose we add a field to `square%`, called `area`, which is computed at construction time and stored away in the field.

```
(define (square% side)
  (local [(define super (object%))
          (define area (sqr side))
          (define (the-square msg)
            (cond [(symbol=? msg 'side) side]
                  [(symbol=? msg 'area) area]
                  ;; Overriding the hi method.
                  [(symbol=? msg 'hi) "Good day, sir."]
                  [else (super msg)]))]
    the-square))
```

So we can write constructors that do computation.

What about this?

```
(define (square% side)
```

```

(local [(define super (object%))
        (define area (sqr side))
        (define (this msg)
          (cond [(symbol=? msg 'side) side]
                [(symbol=? msg 'area) area]
                ;; Overriding the hi method.
                [(symbol=? msg 'hi) "Good day, sir."]
                [else (super msg)]))]
  this))

```

Let's redefine area to use. So in `class/1` we might write `(sqr (this . side))`. We'll if we pick our names better, it should become obvious.

```

(define (square% side)
  (local [(define super (object%))
          (define (this msg)
            (cond [(symbol=? msg 'side) side]
                  [(symbol=? msg 'area) (sqr (this 'side))]
                  ;; Overriding the hi method.
                  [(symbol=? msg 'hi) "Good day, sir."]
                  [else (super msg)]))]
    this))

```

21.3 Objects as Structures

Now I want to step back and look at a different approach for doing this.

Guiding principle: Data + Functions

```

;; A Method is a Function.
;; An Object is a (make-obj [Listof Any] [Listof Method]).

```

```

(define-struct obj (fields methods))

```

But the "is a Function" contract is not very useful, but really we don't know what the contract on a method is until later.

Let's create our simple square-10 object:

```

(define square-10 (make-obj (list ...) (list ...)))

```

What should go in these lists?

```

(define square-10
  (make-obj (list 10)

```



```
(list
  ;; side : -> Number
  (lambda () 10)
  ;; area : -> Number
  (lambda () 100)))
```

```
(check-expect ((first (obj-methods square-10))) 10)
(check-expect ((second (obj-methods square-100))) 100)
```

What's wrong with this? Nothing is called by name.

Methods can't access the fields! What the hell is going on here?

Here's an idea: pass the object itself to the methods.

```
(define square-10
  (make-obj (list 10)
    (list
      ;; side : -> Number
      (lambda (itself)
        (first (object-fields themselves)))
      ;; area : -> Number
      (lambda (itself)
        (* (first (obj-fields itself))
           (first (obj-fields themselves)))))))

(check-expect ((first (obj-methods square-10)) square-10) 10)
(check-expect ((second (obj-methods square-100)) square-10) 100)
```

```
(define square-10
  (make-obj (list 10)
    (list
      ;; side : -> Number
      (lambda (itself)
        (first (object-fields themselves)))
      ;; area : -> Number
      (lambda (itself)
        (* ((first (obj-methods itself)) itself)
           ((first (obj-methods itself)) themselves)))))))
```

A better name for itself: this!

It's annoying to program like this, but we can abstract this

```
;; Object Name -> Anything
(define (send obj meth)
  ...)
```

Why would you ever do this? Every single object-oriented language you've programmed in works like this: it has a table of data and functions and those functions take as its first object the object itself.

Python makes you write `self` as the first argument, which is just exposing this implementation detail.

Why would you do one or the other?

The functional style is slow, but easy.

The structural style is fast, but hard.

Here's a question: where in the methods do we need to refer to `square-10`? Nowhere. Thus we can easily lift the methods out of the definition for `square-10`.

```
(define square-methods
  (list
    ;; side : -> Number
    (lambda (itself)
      (first (object-fields themselves)))
    ;; area : -> Number
    (lambda (itself)
      (* ((first (obj-methods themselves)) themselves)
         ((first (obj-methods themselves)) themselves))))))

(define square-10
  (make-obj (list 10) square-methods))

(define square-5
  (make-obj (list 5) square-methods))

(define (square% side)
  (make-obj (list side) square-methods))
```

Part VIII

Solutions

Chapter 22


Solutions

This appendix contains solutions to selected exercises.

Index

- “Abstract” classes, ??
- [2htdp/image](#), ??
- [2htdp/image](#), ??
- [2htdp/universe](#), ??
- A big-bang oriented to objects, ??
- A Brief History of Objects, ??
- A class of rockets, ??
- A Class of Your Own, ??
- A light of a different color, ??
- A look at the Universe API, ??
- A More Complex Example, ??
- Abstract Lists, ??
- Abstracting list methods with different representations, ??
- Abstraction, ??
- abstraction, ??
- Abstraction Barriers and Modules, ??
- Abstraction via Delegation, ??
- Abstraction via Inheritance, ??
- Abstraction with Objects, ??
- Acknowledgments, ??
- Adding a satellite, ??
- An implementation of coordinates: segments, ??
- animation, ??
- Another implementation of coordinates: food, ??
- astronomical units*, ??
- Atomic and Compound Data, ??
- AU, astronomical units, ??
- AU, ??
- AU, ??
- Back-channels, ??
- Basic Design with Objects, ??
- Basic game mechanics, ??
- [big-bang](#), ??
- [big-bang](#), ??
- [big-bang](#), ??
- [big-bang](#), ??
- [big-bang](#), ??
- [bitmap](#), ??
- Brown, Daniel, ??
- Case Study: Mobiles, ??
- Ch-Ch-Ch-Ch-Changes, ??
- Changing Places, ??
- Circles, ??
- Circular Data, ??
- class*, ??
- [class/0](#), ??
- [class/universe](#), ??
- Classes of Objects: Data Definitions, ??
- Classes of Objects: Interface Definitions, ??
- Cocoa framework, ??
- Comparing different kinds of things, ??
- complex number, mathematical notation, ??
- complex number, ??
- complex number*, ??
- complex numbers, ??
- Complex, with class, ??
- composition, ??
- compound data, ??
- Constructor design issue in modulo zombie (Assignment 3, Problem 3), ??
- Contracts, ??
- Coordinate interface, ??
- data, ??
- Data Definitions, ??

- Data inheritance with binary trees, ??
- `define` form, ??
- `define-class`, ??
- Defining the playing space, ??
- Delegation, ??
- Design Choices, ??
- Design Recipe, ??
- Design Recipes, ??
- Designing Programs with Class, ??
- Different representation of Snakes, ??
- Dijkstra, Edsger W., ??
- `dist`, ??
- `draw-on`, ??
- Drawing the game, ??
- DrRacket, ??
- `empty-scene`, ??
- enumeration*, ??
- Enumerations, ??
- Enumerations, ??
- Equality and Parameterized Types, ??
- Equality in Java, ??
- `equals` and `hashCode`, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Exercises, ??
- Extensional Equality in Java, ??
- Felleisen, Matthias, ??
- field*, ??
- Florence, Spencer, ??
- Folds, ??
- Folds vs Visitors, ??
- forest*, ??
- full tree*, ??
- `function`, ??
- Functional rocket, ??
- functions, ??
- Generators, ??
- Ghosts, ??
- Github, ??
- graphical user interface, GUI, ??
- graphics coordinates*, ??
- Guess my number, ??
- Guessing Big, ??
- Home on the Range, ??
- How to Design Programs*, ??
- images*, ??
- imaginary part*, ??
- Information in the Snake Game, ??
- Inheritance with shapes, ??
- Intensional equality, ??
- Interface Definitions, ??
- Invariants, ??
- Invariants of Data Structures, ??
- Invariants, Testing, and Abstraction Barriers, ??
- Iterating over Data, ??
- Java, ??
- Java, ??
- Java, ??
- Java Syntax, ??
- JSON, ??
- JSON visitor, ??
- JSON, Jr., ??
- Kay, Alan, ??
- Knauth, Geoffrey S., ??
- Labich, Nicholas, ??
- Landing and taking off, ??
- Laplante, Sarah, ??
- Larger system design: Snakes on a plane, ??
- Lee, Alex, ??
- Lift off, ??
- Lights, revisited, ??
- Lisp, ??
- List of Posn, ??
- Lists of Numbers, ??
- Mac OS X, ??
- MacKenzie, Becca, ??

- Many Players, One Number, ??
- Massachusetts Institute of Technology, ??
- Messages, ??
- method, headers, ??
- Method inheritance with binary trees, ??
- methods*, ??
- Mullins, Kathleen, ??
- Mutation, ??
- New York Times*, ??
- next*, ??
- next*, ??
- object*, ??
- object-oriented programming*, ??
- Object-oriented rocket, ??
- Objects = Data + Function, ??
- Objects as Functions, ??
- Objects as Structures, ??
- Okasaki, Chris, ??
- on-tick*, method, ??
- on-tick*, ??
- on-tick*, ??
- One Player Guess my Number, ??
- overlay/align/offset*, ??
- overlay/align/offset*, ??
- PacMan, ??
- Parameteric methods, ??
- Parameterized Data and Interfaces, ??
- Parameterized Data Definitions, ??
- Parametric data, ??
- Parametric interfaces, ??
- Parametric Lists, ??
- Patten, Nikko, ??
- pixels*, ??
- Player in motion, or, I ain't afraid of no ghosts!, ??
- Plessner, Ryan, ??
- Posn, ??
- Preface, ??
- Primum non copy-and-paste, ??
- Programming in Java, ??
- Properties of Programs and Randomized Testing, ??
- Pulling back the veil from object-oriented programming, ??
- PX, pixels, ??
- Pépin, Jacques, ??
- quick list*, ??
- Quick Lists, ??
- Quick visits, ??
- Racket, ??
- real part*, ??
- Recursive Unions, ??
- render*, ??
- render*, ??
- render*, ??
- Representation inpedendence and extensibility, ??
- Representing the snake, ??
- Revisiting the Rocket, ??
- Revisiting the Rocket with Inheritance, ??
- rocket, *ROCKET-SPEED*, ??
- rocket, *rocket%*, ??
- rocket, *next*, ??
- rocket, launch, ??
- rocket, *DELTA*, ??
- rocket, *CLOCK-SPEED*, ??
-  , ??
- ROCKET*, ??
- rocket, ??
- racket%*, *to-draw*, ??
- racket%*, *on-tick*, ??
- rocket%*, ??
- Rules of engagement: protocols and enforcement, ??
- Running Java Programs, ??
- Schemes of a Larger Design, ??
- Seeing the world, ??
- send*, ??
- Shapes, ??
- Shargo, Jim, ??
- Sharing Interfaces, ??
- Simple universe, receiving broadcasts, ??

Simple world, ??
 Simple world, broadcasting to server, ??
 Simple world, receiving messages from the server, ??
 Simula 67, ??
 Simula I, ??
 Smalltalk, ??
 Solidifying what we've done, ??
 Solutions, ??
 Solutions, ??
 Sontag, Trevor, ??
 Space Invaders!, ??
 structure, ??
 Takikawa, Asumu, ??
technique, ??
 The Choice of Language and Environment, ??
 The class/1 language, ??
 The GmN Client, ??
 The GmN server, ??
 The `next` and `render` methods, ??
 The `next` function, ??
 The Parts of the Book, ??
 The player, ??
 The `render` function, ??
 The Visitor Pattern, ??
 The whole ball of wax, ??
 The world, ??
`this` variable, ??
`tick-rate`, ??
`to-draw`, method, ??
`to-draw`, ??
`to-draw`, ??
 Tron, ??
 Turing Award, ??
 Two Ideas: Java and Types, ??
 Two player guess my number, ??
 Types, ??
 Under the Hood: Implementing OO, ??
Unions, ??
 Unions and Recursive Unions, ??
 Universe, ??
 Visitors and Folds, ??
world states, ??
 World-building, ??
 Xerox PARC, ??
 Zombie!, ??